



# **HAXE**

## **Haxe 3 Manual**

Haxe Foundation  
January 3, 2016

# Todo list

Could we have a big Haxe logo in the First Manual Page (Introduction) under the menu (a bit like a book cover ?) It looks a bit empty now and is a landing page for "Manual" .	8
This generates the following output: too many 'this'. You may like a passive sentence: the following output will be generated...though this is to be avoided, generally . . . . .	10
make sure the types are right for inc, dec, negate, and bitwise negate . . . . .	14
While introducing the different operations, we should include that information as well, including how they differ with the "C" standard, see <a href="http://haxe.org/manual/operators">http://haxe.org/manual/operators</a>	14
please review, doubled content . . . . .	16
review please, sounds weird . . . . .	16
for starters...please review . . . . .	17
Is there a difference between <code>y : Int</code> and <code>y : Null&lt;Int&gt;</code> or can you even do the latter? Some more explanation and examples with native optional and Haxe optional arguments and how they relate to nullability would be nice. . . . .	18
please review future tense . . . . .	18
Same as in 2.2, what is <code>Enum&lt;T&gt;</code> syntax? . . . . .	21
list arguments . . . . .	22
please reformat . . . . .	24
I don't really know how these work yet. . . . .	25
It seems a bit convoluted explanations. Should we maybe start by "decoding" the meaning of <code>Void -&gt; Void</code> , then <code>Int -&gt; Bool -&gt; Float</code> , then maybe have samples using <code>\$type</code> . . . . .	25
please review your use of "this" and try to vary somewhat to avoid too much word repetition	32
please review for correctness . . . . .	35
You have marked "Map" for some reason . . . . .	35
Mention <code>toString()</code> / <code>String</code> conversion somewhere in this chapter. . . . .	47
"parent class" should probably be used here, but I have no idea what it means, so I will refrain from changing it myself. . . . .	48
proper label and caption + code/identifier styling for diagram . . . . .	55
Figure out wtf our rules are now for when this is checked. . . . .	93
Comprehensions are only listing Arrays, not Maps . . . . .	94
what to use for listing of non-haxe code like <code>hxml</code> ? . . . . .	116
Check if we can build GADTs this way. . . . .	128
I think C++ can use integer operations, but I don't know about any other targets. Only saw this mentioned in an old discussion thread, still true? . . . . .	145

# Contents

<b>Todo list</b>	<b>1</b>
<b>1 Introduction</b>	<b>8</b>
1.1 What is Haxe?	8
1.2 About this Document	9
1.2.1 Authors and contributions	9
1.2.2 License	10
1.3 Hello World	10
1.4 History	10
<b>I Language Reference</b>	<b>12</b>
<b>2 Types</b>	<b>13</b>
2.1 Basic Types	14
2.1.1 Numeric types	14
2.1.2 Overflow	14
2.1.3 Numeric Operators	14
2.1.4 Bool	16
2.1.5 Void	16
2.2 Nullability	17
2.2.1 Optional Arguments and Nullability	18
2.3 Class Instance	18
2.3.1 Class constructor	19
2.3.2 Inheritance	19
2.3.3 Interfaces	20
2.4 Enum Instance	21
2.4.1 Enum Constructor	22
2.4.2 Using enums	23
2.5 Anonymous Structure	23
2.5.1 JSON for Structure Values	25
2.5.2 Class Notation for Structure Types	25
2.5.3 Optional Fields	25
2.5.4 Impact on Performance	25
2.6 Function Type	25
2.6.1 Optional Arguments	26
2.6.2 Default values	27
2.7 Dynamic	27
2.7.1 Dynamic with Type Parameter	28
2.7.2 Implementing Dynamic	29

2.8	Abstract	30
2.8.1	Implicit Casts	31
2.8.2	Operator Overloading	34
2.8.3	Array Access	35
2.8.4	Selective Functions	36
2.8.5	Enum abstracts	37
2.8.6	Forwarding abstract fields	38
2.8.7	Core-type abstracts	39
2.9	Monomorph	39
<b>3</b>	<b>Type System</b>	<b>40</b>
3.1	Typedef	40
3.1.1	Extensions	41
3.2	Type Parameters	42
3.2.1	Constraints	43
3.3	Generic	43
3.3.1	Construction of generic type parameters	45
3.4	Variance	46
3.5	Unification	47
3.5.1	Between Class/Interface	48
3.5.2	Structural Subtyping	49
3.5.3	Monomorphs	49
3.5.4	Function Return	49
3.5.5	Common Base Type	49
3.6	Type Inference	50
3.6.1	Top-down Inference	51
3.6.2	Limitations	52
3.7	Modules and Paths	52
3.7.1	Module Sub-Types	53
3.7.2	Import	53
3.7.3	Resolution Order	55
<b>4</b>	<b>Class Fields</b>	<b>58</b>
4.1	Variable	58
4.2	Property	59
4.2.1	Common accessor identifier combinations	61
4.2.2	Impact on the type system	62
4.2.3	Rules for getter and setter	63
4.3	Method	64
4.3.1	Overriding Methods	65
4.3.2	Effects of variance and access modifiers	66
4.4	Access Modifier	67
4.4.1	Visibility	67
4.4.2	Inline	68
4.4.3	Dynamic	69
4.4.4	Override	70
4.4.5	Static	70

<b>5</b>	<b>Expressions</b>	<b>71</b>
5.1	Blocks	73
5.2	Constants	73
5.3	Binary Operators	74
5.4	Unary Operators	74
5.5	Array Declaration	74
5.6	Object Declaration	74
5.7	Field Access	74
5.8	Array Access	75
5.9	Function Call	75
5.10	var	75
5.11	Local functions	75
5.12	new	76
5.13	for	76
5.14	while	77
5.15	do-while	77
5.16	if	78
5.17	switch	78
5.18	try/catch	78
5.19	return	79
5.20	break	79
5.21	continue	80
5.22	throw	80
5.23	cast	80
5.23.1	unsafe cast	80
5.23.2	safe cast	81
5.24	type check	81
<b>6</b>	<b>Language Features</b>	<b>82</b>
6.1	Conditional Compilation	84
6.2	Externs	85
6.3	Static Extension	87
6.3.1	In the Haxe Standard Library	87
6.4	Pattern Matching	88
6.4.1	Introduction	88
6.4.2	Enum matching	89
6.4.3	Variable capture	89
6.4.4	Structure matching	90
6.4.5	Array matching	90
6.4.6	Or patterns	90
6.4.7	Guards	91
6.4.8	Match on multiple values	91
6.4.9	Extractors	91
6.4.10	Exhaustiveness checks	93
6.4.11	Useless pattern checks	93
6.5	String Interpolation	94
6.6	Array Comprehension	94
6.7	Iterators	95
6.8	Function Bindings	96
6.9	Metadata	97
6.10	Access Control	98

6.11	Inline constructors	100
<b>II</b>	<b>Compiler Reference</b>	<b>102</b>
<b>7</b>	<b>Compiler Usage</b>	<b>103</b>
7.1	Global Compiler Flags	104
<b>8</b>	<b>Compiler Features</b>	<b>106</b>
8.1	Built-in Compiler Metadata	106
8.2	Dead Code Elimination	108
8.3	Compiler Services	108
8.3.1	Overview	108
8.3.2	Field access completion	110
8.3.3	Call argument completion	111
8.3.4	Type path completion	111
8.3.5	Usage completion	113
8.3.6	Position completion	114
8.3.7	Top-level completion	114
8.3.8	Completion server	115
8.4	Resources	116
8.4.1	Embedding resources	116
8.4.2	Retrieving text resources	117
8.4.3	Retrieving binary resources	117
8.4.4	Implementation details	117
8.5	Runtime Type Information	118
8.5.1	RTTI structure	118
<b>9</b>	<b>Macros</b>	<b>121</b>
9.1	Macro Context	122
9.2	Arguments	122
9.2.1	ExprOf	123
9.2.2	Constant Expressions	124
9.2.3	Rest Argument	124
9.3	Reification	124
9.3.1	Expression Reification	124
9.3.2	Type Reification	125
9.3.3	Class Reification	126
9.4	Tools	126
9.5	Type Building	127
9.5.1	Enum building	128
9.5.2	@:autoBuild	129
9.5.3	@:genericBuild	130
9.6	Limitations	132
9.6.1	Macro-in-Macro	132
9.6.2	Static extension	132
9.6.3	Build Order	132
9.6.4	Type Parameters	133
9.7	Initialization macros	133

<b>III</b>	<b>Standard Library</b>	<b>134</b>
<b>10</b>	<b>Standard Library</b>	<b>135</b>
10.1	String	135
10.2	Data Structures	135
10.2.1	Array	135
10.2.2	Vector	137
10.2.3	List	137
10.2.4	GenericStack	138
10.2.5	Map	138
10.2.6	Option	139
10.3	Regular Expressions	140
10.3.1	Matching	141
10.3.2	Groups	141
10.3.3	Replace	142
10.3.4	Split	143
10.3.5	Map	143
10.3.6	Implementation Details	143
10.4	Math	144
10.4.1	Special Numbers	144
10.4.2	Mathematical Errors	144
10.4.3	Integer Math	145
10.4.4	Extensions	145
10.5	Lambda	145
10.6	Template	147
10.7	Reflection	149
10.8	Serialization	151
10.8.1	Serialization format	153
10.9	Xml	154
10.9.1	Getting started with Xml	155
10.9.2	Parsing Xml	156
10.9.3	Encoding Xml	156
10.10	Json	156
10.10.1	Parsing JSON	156
10.10.2	Encoding JSON	157
10.10.3	Implementation details	157
10.11	Input/Output	157
10.12	Sys/sys	157
10.13	Remoting	157
10.13.1	Remoting Connection	157
10.13.2	Implementation details	159
10.14	Unit testing	160
<b>IV</b>	<b>Miscellaneous</b>	<b>162</b>
<b>11</b>	<b>Haxelib</b>	<b>163</b>
11.1	Using a Haxe library with the Haxe Compiler	163
11.2	haxelib.json	163
11.2.1	Versioning	164
11.2.2	Dependencies	165
11.3	extraParams.html	166

11.4 Using Haxelib . . . . .	166
<b>12 Target Details</b>	<b>168</b>
12.1 JavaScript . . . . .	168
12.1.1 Getting started with Haxe/JavaScript . . . . .	168
12.1.2 Using external JavaScript libraries . . . . .	169
12.1.3 Inject raw JavaScript . . . . .	171
12.1.4 JavaScript untyped functions . . . . .	171
12.1.5 Debugging JavaScript . . . . .	172
12.1.6 JavaScript target Metadata . . . . .	173
12.1.7 Exposing Haxe classes for JavaScript . . . . .	173
12.1.8 Loading extern classes using "require" function . . . . .	174
12.2 Flash . . . . .	175
12.2.1 Getting started with Haxe/Flash . . . . .	175
12.2.2 Embedding resources . . . . .	176
12.2.3 Using external Flash libraries . . . . .	176
12.2.4 Flash target Metadata . . . . .	176
12.3 Neko . . . . .	177
12.4 PHP . . . . .	177
12.4.1 Getting started with Haxe/PHP . . . . .	177
12.4.2 PHP untyped functions . . . . .	178
12.5 C++ . . . . .	178
12.5.1 Using C++ Defines . . . . .	178
12.5.2 Using C++ Pointers . . . . .	181
12.6 Java . . . . .	181
12.7 C# . . . . .	181
12.8 Python . . . . .	181



# Chapter 1

## Introduction

### 1.1 What is Haxe?

Could we have a big Haxe logo in the First Manual Page (Introduction) under the menu (a bit like a book cover ?) It looks a bit empty now and is a landing page for "Manual"

Haxe consists of a high-level, open source programming language and a compiler. It allows compilation of programs, written using an ECMAScript<sup>1</sup>-oriented syntax, to multiple target languages. Employing proper abstraction, it is possible to maintain a single code-base which compiles to multiple targets.

Haxe is strongly typed but the typing system can be subverted where required. Utilizing type information, the Haxe type system can detect errors at compile-time which would only be noticeable at run-time in the target language. Furthermore, type information can be used by the target generators to generate optimized and robust code.

Currently, there are nine supported target languages which allow for different use-cases:

Name	Output type	Main usages
JavaScript	Sourcecode	Browser, Desktop, Mobile, Server
Neko	Bytecode	Desktop, Server
PHP	Sourcecode	Server
Python	Sourcecode	Desktop, Server
C++	Sourcecode	Desktop, Mobile, Server
ActionScript 3	Sourcecode	Browser, Desktop, Mobile
Flash	Bytecode	Browser, Desktop, Mobile
Java	Sourcecode	Desktop, Server
C#	Sourcecode	Desktop, Mobile, Server

The remainder of section 1 gives a brief overview of what a Haxe program looks like and how Haxe evolved since its inception in 2005.

**Types** (Chapter 2) introduces the seven different kinds of types in Haxe and how they interact with each other. The discussion of types is continued in **Type System** (Chapter 3), where features like *unification*, *type parameters* and *type inference* are explained.

**Class Fields** (Chapter 4) is all about the structure of Haxe classes and, among other topics, deals with *properties*, *inline fields* and *generic functions*.

In **Expressions** (Chapter 5) we see how to actually get programs to do something by using *expressions*.

**Language Features** (Chapter 6) describes some of the Haxe features in detail such as *pattern matching*, *string interpolation* and *dead code elimination*. This concludes the Haxe language reference.

<sup>1</sup><http://www.ecma-international.org/publications/standards/Ecma-327.htm>

We continue with the Haxe compiler reference, which first handles the basics in [Compiler Usage](#) ([Chapter 7](#)) before getting into the advanced features in [Compiler Features](#) ([Chapter 8](#)). Finally, we will venture into the exciting land of *haxe macros* in [Macros](#) ([Chapter 9](#)) to see how some common tasks can be greatly simplified.

In the following chapter, [Standard Library](#) ([Chapter 10](#)), we explore important types and concepts from the Haxe Standard Library. We then learn about Haxe’s package manager Haxelib in [Haxelib](#) ([Chapter 11](#)).

Haxe abstracts away many target differences, but sometimes it is important to interact with a target directly, which is the subject of [Target Details](#) ([Chapter 12](#)).

## 1.2 About this Document

This document is the official manual for Haxe 3. As such, it is not a beginner’s tutorial and does not teach programming. However, the topics are roughly designed to be read in order and there are references to topics “previously seen” and topics “yet to come”. In some cases, an earlier section makes use of the information of a later section if it simplifies the explanation. These references are linked accordingly and it should generally not be a problem to read ahead on other topics.

We use a lot of Haxe source code to keep a practical connection of theoretical matters. These code examples are often complete programs that come with a main function and can be compiled as-is. However, sometimes only the most important parts are shown. Source code looks like this:

```
1 Haxe code here
```

Occasionally, we demonstrate how Haxe code is generated, for which we usually show the JavaScript target.

Furthermore, we define a set of terms in this document. Predominantly, this is done when introducing a new type or when a term is specific to Haxe. We do not define every new aspect we introduce, e.g. what a class is, to avoid cluttering the text. A definition looks like this:

**Definition: Definition name**  
Definition description

In a few places, this document has *trivia*-boxes. These include off-the-record information such as why certain decisions were made during Haxe’s development or how a particular feature has been changed in past Haxe versions. This information is generally not important and can be skipped as it is only meant to convey trivia:

**Trivia: About Trivia**  
This is trivia.

### 1.2.1 Authors and contributions

Most of this document’s content was written by Simon Krajewski while working for the Haxe Foundation. We would like to thank these people for their contributions:

- Dan Korostelev: Additional content and editing
- Caleb Harper: Additional content and editing
- Josefiene Pertosa: Editing

- Miha Lunar: Editing
- Nicolas Cannasse: Haxe creator

### 1.2.2 License

The Haxe Manual by [Haxe Foundation](#) is licensed under a [Creative Commons Attribution 4.0 International License](#).

Based on a work at <https://github.com/HaxeFoundation/HaxeManual>.

## 1.3 Hello World

The following program prints “Hello World” after being compiled and run:

```
1 class Main {
2     static public function main():Void {
3         trace("Hello World");
4     }
5 }
```

This can be tested by saving the above code to a file named `Main.hx` and invoking the Haxe Compiler like so: `haxe -main Main --interp`. It then generates the following output: `Main.hx:3: Hello world`. There are several things to learn from this:

- Haxe programs are saved in files with an extension of `.hx`.
- The Haxe Compiler is a command-line tool which can be invoked with parameters such as `-main Main` and `--interp`.
- Haxe programs have classes (`Main`, upper-case), which have functions (`main`, lower-case).
- The name of the file containing main Haxe class is the same as name of the class itself (in this case `Main.hx`).

This generates the following output: too many ‘this’. You may like a passive sentence: the following output will be generated...though this is to be avoided, generally

## 1.4 History

The Haxe project was started on 22 October 2005 by French developer *Nicolas Cannasse* as a successor to the popular open-source ActionScript 2 compiler *MTASC* (Motion-Twin Action Script Compiler) and the in-house *MTypes* language, which experimented with the application of type inference to an object oriented language. Nicolas’ long-time passion for programming language design and the rise of new opportunities to mix different technologies as part of his game developer work at *Motion-Twin* led to the creation of a whole new language.

Being spelled *haXe* back then, its beta version was released in February 2006 with the first supported targets being AVM<sup>2</sup>-bytecode and Nicolas’ own *Neko* virtual machine<sup>3</sup>.

Nicolas Cannasse, who remains leader of the Haxe project to this date, kept on designing Haxe with a clear vision, subsequently leading to the Haxe 1.0 release in May 2006. This first major release came with support for JavaScript code generation and already had some of the features that define Haxe today such as type inference and structural sub-typing.

Haxe 1 saw several minor releases over the course of two years, adding the Flash AVM2 target along with the *haxelib*-tool in August 2006 and the ActionScript 3 target in March 2007. During

<sup>2</sup>Adobe Virtual Machine

<sup>3</sup><http://neko.vm.org>

these months, there was a strong focus on improving stability, which resulted in several minor bug-fix releases.

Haxe 2.0 was released in July 2008, including the PHP target, courtesy of *Franco Ponticelli*. A similar effort by *Hugh Sanderson* lead to the addition of the C++ target in July 2009 with the Haxe 2.04 release.

Just as with Haxe 1, what followed were several months of stability releases. In January 2011, Haxe 2.07 was released with the support of *macros*. Around that time, *Bruno Garcia* joined the team as maintainer of the JavaScript target, which saw vast improvements in the subsequent 2.08 and 2.09 releases.

After the release of 2.09, *Simon Krajewski* joined the team and work towards Haxe 3 began. Furthermore, *Cauê Waneck's* Java and C# targets found their way into the Haxe builds. It was then decided to make one final Haxe 2 release, which happened in July 2012 with the release of Haxe 2.10.

In late 2012, the Haxe 3 switch was flipped and the Haxe Compiler team, now backed by the newly established *Haxe Foundation*<sup>4</sup>, focused on this next major version. Haxe 3 was subsequently released in May 2013.

---

<sup>4</sup><http://haxe-foundation.org>

**Part I**

**Language Reference**

## Chapter 2

# Types

The Haxe Compiler employs a rich type system which helps detecting type-related errors in a program at compile-time. A type error is an invalid operation on a given type such as dividing by a `String`, trying to access a field of an `Integer` or calling a function with not enough (or too many) arguments.

In some languages this additional safety comes at a price because programmers are forced to explicitly assign types to syntactic constructs:

```
1 var myButton:MySpecialButton = new MySpecialButton(); // As3
2 MySpecialButton* myButton = new MySpecialButton(); // C++
```

The explicit type annotations are not required in Haxe, because the compiler can *infer* the type:

```
1 var myButton = new MySpecialButton(); // Haxe
```

We will explore type inference in detail later in [Type Inference](#) (Section 3.6). For now, it is sufficient to say that the variable `myButton` in the above code is known to be an *instance of class* `MySpecialButton`.

The Haxe type system knows seven type groups:

**Class instance:** an object of a given class or interface

**Enum instance:** a value of a Haxe enumeration

**Structure:** an anonymous structure, i.e. a collection of named fields

**Function:** a compound type of several arguments and one return

**Dynamic:** a wildcard type which is compatible with any type

**Abstract:** a compile-time type which is represented by a different type at runtime

**Monomorph:** an unknown type which may later become a different type

We will describe each of these type groups and how they relate to each other in the next chapters.

### Definition: Compound Type

A compound type is a type which has sub-types. This includes any type with type parameters (3.2) and the function (2.6) type.

## 2.1 Basic Types

*Basic types* are `Bool`, `Float` and `Int`. They can easily be identified in the syntax by values such as

- `true` and `false` for `Bool`,
- `1`, `0`, `-1` and `0xFF0000` for `Int` and
- `1.0`, `0.0`, `-1.0`, `1e10` for `Float`.

Basic types are not classes (2.3) in Haxe. They are implemented as abstract types (2.8) and are tied to the compiler's internal operator-handling as described in the following sections.

### 2.1.1 Numeric types

#### Type: `Float`

Represents a double-precision IEEE 64-bit floating point number.

#### Type: `Int`

Represents an integral number.

While every `Int` can be used where a `Float` is expected (that is, `Int` is assignable to or unifies with `Float`), the reverse is not true: Assigning a `Float` to an `Int` might lose precision and is not allowed implicitly.

### 2.1.2 Overflow

For performance reasons, the Haxe Compiler does not enforce any overflow behavior. The burden of checking for overflows falls to the target platform. Here are some platform specific notes on overflow behavior:

**C++, Java, C#, Neko, Flash:** 32-bit signed integers with usual overflow practices

**PHP, JS, Flash 8:** No native `Int` type, loss of precision will occur if they reach their float limit ( $2^{52}$ )

Alternatively, the `haxe.Int32` and `haxe.Int64` classes can be used to ensure correct overflow behavior regardless of the platform at the cost of additional computations depending on the platform.

### 2.1.3 Numeric Operators

This the list of numeric operators in Haxe, grouped by descending priority.

make sure the types are right for inc, dec, negate, and bitwise negate

While introducing the different operations, we should include that information as well, including how they differ with the "C" standard, see <http://haxe.org/manual/operators>

Arithmetic				
Operator	Operation	Operand 1	Operand 2	Return
++	increment	Int	N/A	Int
		Float	N/A	Float
--	decrement	Int	N/A	Int
		Float	N/A	Float
+	addition	Float	Float	Float
		Float	Int	Float
		Int	Float	Float
		Int	Int	Int
-	subtraction	Float	Float	Float
		Float	Int	Float
		Int	Float	Float
		Int	Int	Int
*	multiplication	Float	Float	Float
		Float	Int	Float
		Int	Float	Float
		Int	Int	Int
/	division	Float	Float	Float
		Float	Int	Float
		Int	Float	Float
		Int	Int	Float
%	modulo	Float	Float	Float
		Float	Int	Float
		Int	Float	Float
		Int	Int	Int
Comparison				
Operator	Operation	Operand 1	Operand 2	Return
==	equal	Float/Int	Float/Int	Bool
!=	not equal	Float/Int	Float/Int	Bool
<	less than	Float/Int	Float/Int	Bool
<=	less than or equal	Float/Int	Float/Int	Bool
>	greater than	Float/Int	Float/Int	Bool
>=	great than or equal	Float/Int	Float/Int	Bool
Bitwise				
Operator	Operation	Operand 1	Operand 2	Return
~	bitwise negation	Int	N/A	Int
&	bitwise and	Int	Int	Int
	bitwise or	Int	Int	Int
^	bitwise xor	Int	Int	Int
<<	shift left	Int	Int	Int
>>	shift right	Int	Int	Int
>>>	unsigned shift right	Int	Int	Int

**Equality** *For enums:*

**Enum without parameters** Are always represent the same value, so `MyEnum.A == MyEnum.A`.

**Enum with parameters** Can be compared with `a.equals(b)` (which is a short for `Type.enumEquals()`).

*Dynamic:* Comparison involving at least one Dynamic value is unspecified and platform-specific.



## 2.1.4 Bool

### Type: Bool

Represents a value which can be either *true* or *false*.

Values of type `Bool` are a common occurrence in *conditions* such as `if` (5.16) and `while` (5.14). The following *operators* accept and return `Bool` values:

- `&&` (and)
- `||` (or)
- `!` (not)

Haxe guarantees that compound boolean expressions are evaluated from left to right and only as far as necessary at run-time. For instance, an expression like `A && B` will evaluate `A` first and evaluate `B` only if the evaluation of `A` yielded `true`. Likewise, the expressions `A || B` will not evaluate `B` if the evaluation of `A` yielded `true`, because the value of `B` is irrelevant in that case. This is important in cases such as this:

```
1 if (object != null && object.field == 1) { }
```

Accessing `object.field` if `object` is `null` would lead to a run-time error, but the check for `object != null` guards against it.

## 2.1.5 Void

### Type: Void

`Void` denotes the absence of a type. It is used to express that something (usually a function) has no value.

`Void` is a special case in the type system because it is not actually a type. It is used to express the absence of a type, which applies mostly to function arguments and return types. We have already “seen” `Void` in the initial “Hello World” example:

please review, doubled content

```
1 class Main {
2     static public function main():Void {
3         trace("Hello World");
4     }
5 }
```

The function type will be explored in detail in the section [Function Type](#) (Section 2.6) but a quick preview helps here: The type of the function `main` in the example above is `Void->Void`, which reads as “it has no arguments and returns nothing”. Haxe does not allow fields and variables of type `Void` and will complain if an attempt at declaring such is made:

review please, sounds weird

```
1 // Arguments and variables of type Void are not allowed
2 var x:Void;
```

## 2.2 Nullability

### Definition: nullable

A type in Haxe is considered *nullable* if `null` is a valid value for it.

It is common for programming languages to have a single, clean definition for nullability. However, Haxe has to find a compromise in this regard due to the nature of Haxe's target languages: While some of them allow and; in fact, default to `null` for anything, others do not even allow `null` for certain types. This necessitates the distinction of two types of target languages:

### Definition: Static target

Static targets employ their own type system where `null` is not a valid value for basic types. This is true for the Flash, C++, Java and C# targets.

### Definition: Dynamic target

Dynamic targets are more lenient with their types and allow `null` values for basic types. This applies to the JavaScript, PHP, Neko and Flash 6-8 targets.

There is nothing to worry about when working with `null` on dynamic targets; however, static ones may require some thought. For starters, basic types are initialized to their default values.

for starters...please re-view

### Definition: Default values

Basic types have the following default values on static targets:

**Int:** 0

**Float:** NaN on Flash, 0.0 on other static targets

**Bool:** false

As a consequence, the Haxe Compiler does not allow the assignment of `null` to a basic type on static targets. In order to achieve this, the basic type has to be wrapped as `Null<T>`:

```
1 // error on static platforms
2 var a:Int = null;
3 var b:Null<Int> = null; // allowed
```

Similarly, basic types cannot be compared to `null` unless wrapped:

```
1 var a : Int = 0;
2 // error on static platforms
3 if( a == null ) { ... }
4 var b : Null<Int> = 0;
5 if( b != null ) { ... } // allowed
```

This restriction extends to all situations where unification (3.5) is performed.

### Type: `Null<T>`

On static targets the types `Null<Int>`, `Null<Float>` and `Null<Bool>` can be used to allow `null` as a value. On dynamic targets this has no effect. `Null<T>` can also be used with other types in order to document that `null` is an allowed value.

If a `null`-value is “hidden” in `Null<T>` or `Dynamic` and assigned to a basic type, the default value is used:

```
1 var n : Null<Int> = null;
2 var a : Int = n;
3 trace(a); // 0 on static platforms
```

## 2.2.1 Optional Arguments and Nullability

Optional arguments also have to be accounted for when considering nullability.

In particular, there must be a distinction between *native* optional arguments which are not nullable and Haxe-specific optional arguments which might be. The distinction is made by using the question-mark optional argument:

```
1 // x is a native Int (not nullable)
2 function foo(x : Int = 0) {}
3 // y is Null<Int> (nullable)
4 function bar(?y : Int) {}
5 // z is also Null<Int>
6 function opt(?z : Int = -1) {}
```

Is there a difference between `?y : Int` and `y : Null<Int>` or can you even do the latter? Some more explanation and examples with native optional and Haxe optional arguments and how they relate to nullability would be nice.

### Trivia: Argument vs. Parameter

In some other programming languages, *argument* and *parameter* are used interchangeably. In Haxe, *argument* is used when referring to methods and *parameter* refers to [Type Parameters \(Section 3.2\)](#).

## 2.3 Class Instance

Similar to many object-oriented languages, classes are the primary data structure for the majority of programs in Haxe. Each Haxe class has an explicit name, an implied path and zero or more class fields. Here we will focus on the general structure of classes and their relations, while leaving the details of class fields for [Class Fields \(Chapter 4\)](#).

The following code example serves as basis for the remainder of this section:

```
1 class Point {
2     var x : Int;
3     var y : Int;
4     public function new(x,y) {
5         this.x = x;
6         this.y = y;
7     }
8     public function toString() {
9         return "Point (" + x + ", " + y + ")";
10    }
```

please review future tense

```
10 }
11 }
```

Semantically, this class represents a point in discrete 2-dimensional space - but this is not important here. Let us instead describe the structure:

- The keyword `class` denotes that we are declaring a class.
- `Point` is the name of the class and could be anything conforming to the rules for type identifiers (5).
- Enclosed in curly braces `{}` are the class fields,
- Which consist of two *variable* fields `x` and `y` of type `Int`,
- followed by a special *function* field named `new`, which is the *constructor* of the class,
- as well as a normal function `toString`.

There is a special type in Haxe which is compatible with all classes:

#### **Type: `Class<T>`**

This type is compatible with all class types which means that all classes (not their instances) can be assigned to it. At compile-time, `Class<T>` is the common base type of all class types. However, this relation is not reflected in generated code.

This type is useful when an API requires a value to be *a* class, but not a specific one. This applies to several methods of the Haxe reflection API (10.7).

### 2.3.1 Class constructor

Instances of classes are created by calling the class constructor - a process commonly referred to as *instantiation*. Another name for class instances is *object*. Nevertheless, we prefer the term class instance to emphasize the analogy between classes/class instances and enums/enum instances (2.4).

```
1 var p = new Point(-1, 65);
```

This will yield an instance of class `Point`, which is assigned to a variable named `p`. The constructor of `Point` receives the two arguments `-1` and `65` and assigns them to the instance variables `x` and `y` respectively (compare its definition in [Class Instance \(Section 2.3\)](#)). We will revisit the exact meaning of the `new` expression later in the section 5.12. For now, we just think of it as calling the class constructor and returning the appropriate object.

### 2.3.2 Inheritance

Classes may inherit from other classes, which in Haxe is denoted by the `extends` keyword:

```
1 class Point3 extends Point {
2     var z : Int;
3     public function new(x, y, z) {
4         super(x, y);
5         this.z = z;
6     }
7 }
```

This relation is often described as “is-a”: Any instance of class `Point3` is also an instance of `Point`. `Point` is then known as the *parent class* of `Point3`, which is a *child class* of `Point`. A class may have many child classes, but only one parent class. The term “a parent class of class `X`” usually refers to its direct parent class, the parent class of its parent class and so on.

The code above is very similar to the original `Point` class, with two new constructs being shown:

- `extends Point` denotes that this class inherits from class `Point`
- `super(x, y)` is the call to the constructor of the parent class, in this case `Point.new`

It is not necessary for child classes to define their own constructors, but if they do, a call to `super()` is mandatory. Unlike some other object-oriented languages, this call can appear anywhere in the constructor code and does not have to be the first expression.

A class may override methods (4.3) of its parent class, which requires the explicit `override` keyword. The effects and restrictions of this are detailed in [Overriding Methods \(Section 4.3.1\)](#).

### 2.3.3 Interfaces

An interface can be understood as the signature of a class because it describes the public fields of a class. Interfaces do not provide implementations but pure structural information:

```
1 interface Printable {
2     public function toString():String;
3 }
```

The syntax is similar to classes, with the following exceptions:

- `interface` keyword is used instead of `class` keyword
- functions do not have any expressions (5)
- every field must have an explicit type

Interfaces, unlike structural subtyping (3.5.2), describe a *static relation* between classes. A given class is only considered to be compatible to an interface if it explicitly states so:

```
1 class Point implements Printable { }
```

Here, the `implements` keyword denotes that `Point` has a “is-a” relationship to `Printable`, i.e. each instance of `Point` is also an instance of `Printable`. While a class may only have one parent class, it may implement multiple interfaces through multiple `implements` keywords:

```
1 class Point implements Printable
2     implements Serializable
```

The compiler checks if the `implements` assumption holds. That is, it makes sure the class actually does implement all the fields required by the interface. A field is considered implemented if the class or any of its parent classes provide an implementation.

Interface fields are not limited to methods. They can be variables and properties as well:

```
1 interface Placeable {
2     public var x:Float;
3     public var y:Float;
4 }
5
6 class Main implements Placeable {
```

```

7 public var x:Float;
8 public var y:Float;
9 static public function main() { }
10 }

```

Interfaces can extend multiple other interfaces using the `extends` keyword:

```

1 interface Debuggable extends Printable extends Serializable

```

#### Trivia: Implements Syntax

Haxe versions prior to 3.0 required multiple `implements` keywords to be separated by a comma. We decided to adhere to the de-facto standard of Java and got rid of the comma. This was one of the breaking changes between Haxe 2 and 3.

## 2.4 Enum Instance

Haxe provides powerful enumeration (short: enum) types, which are actually an *algebraic data type* (ADT)<sup>1</sup>. While they cannot have any expressions (5), they are very useful for describing data structures:

```

1 enum Color {
2     Red;
3     Green;
4     Blue;
5     Rgb(r:Int, g:Int, b:Int);
6 }

```

Semantically, this enum describes a color which is either red, green, blue or a specified RGB value. The syntactic structure is as follows:

- The keyword `enum` denotes that we are declaring an enum.
- `Color` is the name of the enum and could be anything conforming to the rules for type identifiers (5).
- Enclosed in curly braces `{}` are the *enum constructors*,
- which are `Red`, `Green` and `Blue` taking no arguments,
- as well as `Rgb` taking three `Int` arguments named `r`, `g` and `b`.

The Haxe type system provides a type which unifies with all enum types:

#### Type: `Enum<T>`

This type is compatible with all enum types. At compile-time, `Enum<T>` can be seen as the common base type of all enum types. However, this relation is not reflected in generated code.

Same as in 2.2, what is `Enum<T>` syntax?

<sup>1</sup>[http://en.wikipedia.org/wiki/Algebraic\\_data\\_type](http://en.wikipedia.org/wiki/Algebraic_data_type)

### 2.4.1 Enum Constructor

Similar to classes and their constructors, enums provide a way of instantiating them by using one of their constructors. However, unlike classes, enums provide multiple constructors which can easily be used through their name:

```
1 var a = Red;
2 var b = Green;
3 var c = Rgb(255, 255, 0);
```

In this code the type of variables `a`, `b` and `c` is `Color`. Variable `c` is initialized using the `Rgb` constructor with arguments.

All enum instances can be assigned to a special type named `EnumValue`.

#### Type: `EnumValue`

`EnumValue` is a special type which unifies with all enum instances. It is used by the Haxe Standard Library to provide certain operations for all enum instances and can be employed in user-code accordingly in cases where an API requires *an* enum instance, but not a specific one.

It is important to distinguish enum types and enum constructors, as this example demonstrates:

```
1 enum Color {
2   Red;
3   Green;
4   Blue;
5   Rgb(r:Int, g:Int, b:Int);
6 }
7
8 class Main {
9   static public function main() {
10    var ec:EnumValue = Red; // valid
11    var en:Enum<Color> = Color; // valid
12    // Error: Color should be Enum<Color>
13    //var x:Enum<Color> = Red;
14   }
15 }
```

If the commented line is uncommented, the program does not compile because `Red` (an enum constructor) cannot be assigned to a variable of type `Enum<Color>` (an enum type). The relation is analogous to a class and its instance.

#### Trivia: Concrete type parameter for `Enum<T>`

One of the reviewers of this manual was confused about the difference between `Color` and `Enum<Color>` in the example above. Indeed, using a concrete type parameter there is pointless and only serves the purpose of demonstration. Usually we would omit the type there and let type inference (3.6) deal with it.

However, the inferred type would be different from `Enum<Color>`. The compiler infers a pseudo-type which has the enum constructors as “fields”. As of Haxe 3.2.0, it is not possible to express this type in syntax but also, it is never necessary to do so.

list arguments

## 2.4.2 Using enums

Enums are a good choice if only a finite set of values should be allowed. The individual constructors (2.4.1) then represent the allowed variants and enable the compiler to check if all possible values are respected. This can be seen here:

```
1 enum Color {
2   Red;
3   Green;
4   Blue;
5   Rgb(r:Int, g:Int, b:Int);
6 }
7
8 class Main {
9   static function main() {
10    var color = getColor();
11    switch (color) {
12      case Red: trace("Color was red");
13      case Green: trace("Color was green");
14      case Blue: trace("Color was blue");
15      case Rgb(r, g, b): trace("Color had a red value of " +r);
16    }
17  }
18
19  static function getColor():Color {
20    return Rgb(255, 0, 255);
21  }
22 }
```

After retrieving the value of `color` by assigning the return value of `getColor()` to it, a `switch` expression (5.17) is used to branch depending on the value. The first three cases `Red`, `Green` and `Blue` are trivial and correspond to the constructors of `Color` that have no arguments. The final case `Rgb(r, g, b)` shows how the argument values of a constructor can be extracted: they are available as local variables within the case body expression, just as if a `var` expression (5.10) had been used.

Advanced information on using the `switch` expression will be explored later in the section on pattern matching (6.4).

## 2.5 Anonymous Structure

Anonymous structures can be used to group data without explicitly creating a type. The following example creates a structure with two fields `x` and `name`, and initializes their values to `12` and `"foo"` respectively:

```
1 class Main {
2   static public function main() {
3     var myStructure = { x: 12, name: "foo" };
4   }
5 }
```

The general syntactic rules follow:

1. A structure is enclosed in curly braces `{}` and



2. Has a *comma-separated* list of key-value-pairs.
3. A *colon* separates the key, which must be a valid identifier (5), from the value.
4. The value can be any Haxe expression.

Rule 4 implies that structures can be nested and complex, e.g.:

please reformat

```
1 var user = {
2   name : "Nicolas",
3   age : 32,
4   pos : [
5     { x : 0, y : 0 },
6     { x : 1, y : -1 }
7   ],
8 };
```

Fields of structures, like classes, are accessed using a *dot* (.) like so:

```
1 // get value of name, which is "Nicolas"
2 user.name;
3 // set value of age to 33
4 user.age = 33;
```

It is worth noting that using anonymous structures does not subvert the typing system. The compiler ensures that only available fields are accessed, which means the following program does not compile:

```
1 class Test {
2   static public function main() {
3     var point = { x: 0.0, y: 12.0 };
4     // { y : Float, x : Float } has no field z
5     point.z;
6   }
7 }
```

The error message indicates that the compiler knows the type of `point`: It is a structure with fields `x` and `y` of type `Float`. Since it has no field `z`, the access fails. The type of `point` is known through type inference (3.6), which thankfully saves us from using explicit types for local variables. However, if `point` was a field, explicit typing would be necessary:

```
1 class Path {
2   var start : { x : Int, y : Int };
3   var target : { x : Int, y : Int };
4   var current : { x : Int, y : Int };
5 }
```

To avoid this kind of redundant type declaration, especially for more complex structures, it is advised to use a *typedef* (3.1):

```
1 typedef Point = { x : Int, y : Int }
2
3 class Path {
4   var start : Point;
5   var target : Point;
6   var current : Point;
7 }
```

## 2.5.1 JSON for Structure Values

It is also possible to use *JavaScript Object Notation* for structures by using *string literals* for the keys:

```
1 var point = { "x" : 1, "y" : -5 };
```

While any string literal is allowed, the field is only considered part of the type if it is a valid Haxe identifier (5). Otherwise, Haxe syntax does not allow expressing access to such a field, and reflection (10.7) has to be employed through the use of `Reflect.field` and `Reflect.setField`.

## 2.5.2 Class Notation for Structure Types

When defining a structure type, Haxe allows using the same syntax as described in [Class Fields \(Chapter 4\)](#). The following `typedef` (3.1) declares a `Point` type with variable fields `x` and `y` of type `Int`:

```
1 typedef Point = {  
2     var x : Int;  
3     var y : Int;  
4 }
```

## 2.5.3 Optional Fields

I don't really know how these work yet.

## 2.5.4 Impact on Performance

Using structures and, by extension, structural subtyping (3.5.2) has no impact on performance when compiling to dynamic targets (2.2). However, on static targets (2.2) a dynamic lookup has to be performed which is typically slower than a static field access.

## 2.6 Function Type

It seems a bit convoluted explanations. Should we maybe start by "decoding" the meaning of `Void` -> `Void`, then `Int` -> `Bool` -> `Float`, then maybe have samples using `$type`

The function type, along with the monomorph (2.9), is a type which is usually well-hidden from Haxe users, yet present everywhere. We can make it surface by using `$type`, a special Haxe identifier which outputs the type its expression has during compilation :

```
1 class Main {  
2     static public function main() {  
3         // i : Int -> s : String -> Bool  
4         $type(test);  
5         $type(test(1, "foo")); // Bool  
6     }  
7  
8     static function test(i:Int, s:String):Bool {  
9         return true;  
10    }  
11 }
```

There is a strong resemblance between the declaration of function `test` and the output of the first `$type` expression, yet also a subtle difference:

- *Function arguments* are separated by the special arrow token `->` instead of commas, and
- the *function return type* appears at the end after another `->`.

In either notation it is obvious that the function `test` accepts a first argument of type `Int`, a second argument of type `String` and returns a value of type `Bool`. If a call to this function, such as `test(1, "foo")`, is made within the second `$type` expression, the Haxe typer checks if `1` can be assigned to `Int` and if `"foo"` can be assigned to `String`. The type of the call is then equal to the type of the value `test` returns, which is `Bool`.

If a function type has other function types as argument or return type, parentheses can be used to group them correctly. For example, `Int -> (Int -> Void) -> Void` represents a function which has a first argument of type `Int`, a second argument of function type `Int -> Void` and a return of `Void`.

## 2.6.1 Optional Arguments

Optional arguments are declared by prefixing an argument identifier with a question mark `?`:

```

1 class Main {
2   static public function main() {
3     // ?i : Int -> ?s : String -> String
4     $type(test);
5     trace(test()); // i: null, s: null
6     trace(test(1)); // i: 1, s: null
7     trace(test(1, "foo")); // i: 1, s: foo
8     trace(test("foo")); // i: null, s: foo
9   }
10
11  static function test(?i:Int, ?s:String) {
12    return "i: " +i + ", s: " +s;
13  }
14 }
```

Function `test` has two optional arguments: `i` of type `Int` and `s` of `String`. This is directly reflected in the function type output by line 3. This example program calls `test` four times and prints its return value.

1. The first call is made without any arguments.
2. The second call is made with a singular argument `1`.
3. The third call is made with two arguments `1` and `"foo"`.
4. The fourth call is made with a singular argument `"foo"`.

The output shows that optional arguments which are omitted from the call have a value of `null`. This implies that the type of these arguments must admit `null` as value, which raises the question of its nullability (2.2). The Haxe Compiler ensures that optional basic type arguments are nullable by inferring their type as `Null<T>` when compiling to a static target (2.2).

While the first three calls are intuitive, the fourth one might come as a surprise: It is indeed allowed to skip optional arguments if the supplied value is assignable to a later argument.

## 2.6.2 Default values

Haxe allows default values for arguments by assigning a *constant value* to them:

```
1 class Main {
2     static public function main() {
3         // ?i : Int -> ?s : String -> String
4         $type(test);
5         trace(test()); // i: 12, s: bar
6         trace(test(1)); // i: 1, s: bar
7         trace(test(1, "foo")); // i: 1, s: foo
8         trace(test("foo")); // i: 12, s: foo
9     }
10
11     static function test(?i = 12, s = "bar") {
12         return "i: " + i + ", s: " + s;
13     }
14 }
```

This example is very similar to the one from [Optional Arguments \(Section 2.6.1\)](#), with the only difference being that the values 12 and "bar" are assigned to the function arguments `i` and `s` respectively. The effect is that the default values are used instead of `null` should an argument be omitted from the call.

Default values in Haxe are not part of the type and are not replaced at call-site (unless the function is inlined ([4.4.2](#)), which can be considered as a more typical approach. On some targets the compiler may still pass `null` for omitted argument values and generate code similar to this into the function:

```
1     static function test(i = 12, s = "bar") {
2         if (i == null) i = 12;
3         if (s == null) s = "bar";
4         return "i: " + i + ", s: " + s;
5     }
```

This should be considered in performance-critical code where a solution without default values may sometimes be more viable.

## 2.7 Dynamic

While Haxe has a static type system, this type system can, in effect, be turned off by using the `Dynamic` type. A *dynamic value* can be assigned to anything; and anything can be assigned to it. This has several drawbacks:

- The compiler can no longer type-check assignments, function calls and other constructs where specific types are expected.
- Certain optimizations, in particular when compiling to static targets, can no longer be employed.
- Some common errors, e.g. a typo in a field access, can not be caught at compile-time and likely cause an error at runtime.
- [Dead Code Elimination \(Section 8.2\)](#) cannot detect used fields if they are used through `Dynamic`.

It is very easy to come up with examples where the usage of `Dynamic` can cause problems at runtime. Consider compiling the following two lines to a static target:

```
1 var d:Dynamic = 1;
2 d.foo;
```

Trying to run a compiled program in the Flash Player yields an error `Property foo not found on Number` and there is no default value. Without `Dynamic`, this would have been detected at compile-time.

#### Trivia: Dynamic Inference before Haxe 3

The Haxe 3 compiler never infers a type to `Dynamic`, so users must be explicit about it. Previous Haxe versions used to infer arrays of mixed types, e.g. `[1, true, "foo"]`, as `Array<Dynamic>`. We found that this behavior introduced too many type problems and thus removed it for Haxe 3.

Use of `Dynamic` should be minimized as there are better options in many situations but sometimes it is just practical to use it. Parts of the Haxe [Reflection](#) (Section 10.7) API use it and it is sometimes the best option when dealing with custom data structures that are not known at compile-time.

`Dynamic` behaves in a special way when being unified (3.5) with a monomorph (2.9). Monomorphs are never bound to `Dynamic` which can have surprising results in examples such as this:

```
1 class Main {
2     static function main() {
3         var jsonData = '[1, 2, 3]';
4         var json = haxe.Json.parse(jsonData);
5         $type(json); // Unknown<0>
6         for (i in 0...json.length) {
7             // Array access is not allowed on
8             // {+ length : Int }
9             trace(json[0]);
10        }
11    }
12 }
```

Although the return type of `Json.parse` is `Dynamic`, the type of local variable `json` is not bound to it and remains a monomorph. It is then inferred as an anonymous structure (2.5) upon the `json.length` field access, which causes the following `json[0]` array access to fail. In order to avoid this, the variable `json` can be explicitly typed as `Dynamic` by using `var json:Dynamic`.

#### Trivia: Dynamic in the Standard Library

`Dynamic` was quite frequent in the Haxe Standard Library before Haxe 3. With the continuous improvements of the Haxe type system the occurrences of `Dynamic` were reduced over the releases leading to Haxe 3.

### 2.7.1 Dynamic with Type Parameter

`Dynamic` is a special type because it allows explicit declaration with and without a type parameter (3.2). If such a type parameter is provided, the semantics described in [Dynamic](#) (Section 2.7) are constrained to all fields being compatible with the parameter type:

```

1 var att : Dynamic<String> = xml.attributes;
2 // valid, value is a String
3 att.name = "Nicolas";
4 // dito (this documentation is quite old)
5 att.age = "26";
6 // error, value is not a String
7 att.income = 0;

```

## 2.7.2 Implementing Dynamic

Classes can implement (2.3.3) `Dynamic` and `Dynamic<T>` which enables arbitrary field access. In the former case, fields can have any type, in the latter, they are constrained to be compatible with the parameter type:

```

1 class ImplementsDynamic
2   implements Dynamic<String> {
3   public var present:Int;
4   public function new() {}
5 }
6
7 class Main {
8   static public function main() {
9     var c = new ImplementsDynamic();
10    // valid, present is an existing field
11    c.present = 1;
12    // valid, assigned value is a String
13    c.stringField = "foo";
14    // error, Int should be String
15    //c.intField = 1;
16  }
17 }

```

Implementing `Dynamic` does not satisfy the requirements of other implemented interfaces. The expected fields still have to be implemented explicitly.

Classes that implement `Dynamic` (with or without type parameter) can also utilize a special method named `resolve`. If a read access (4.2) is made and the field in question does not exist, the `resolve` method is called with the field name as argument:

```

1 class Resolve implements Dynamic<String> {
2   public var present:Int;
3   public function new() {}
4   function resolve(field:String) {
5     return "Tried to resolve " +field;
6   }
7 }
8
9 class Main {
10  static public function main() {
11    var c = new Resolve();
12    c.present = 2;
13    trace(c.present);
14    trace(c.resolveMe);
15  }

```

## 2.8 Abstract

An abstract type is a type which is actually a different type at run-time. It is a compile-time feature which defines types “over” concrete types in order to modify or augment their behavior:

```
1 abstract AbstractInt(Int) {
2   inline public function new(i:Int) {
3     this = i;
4   }
5 }
```

We can derive the following from this example:

- The keyword `abstract` denotes that we are declaring an abstract type.
- `AbstractInt` is the name of the abstract and could be anything conforming to the rules for type identifiers.
- Enclosed in parenthesis `()` is the *underlying type* `Int`.
- Enclosed in curly braces `{}` are the fields,
- which are a constructor function `new` accepting one argument `i` of type `Int`.

### Definition: Underlying Type

The underlying type of an abstract is the type which is used to represent said abstract at runtime. It is usually a concrete (i.e. non-abstract) type but could be another abstract type as well.

The syntax is reminiscent of classes and the semantics are indeed similar. In fact, everything in the “body” of an abstract (that is everything after the opening curly brace) is parsed as class fields. Abstracts may have method (4.3) fields and non-physical (4.2.3) property (4.2) fields.

Furthermore, abstracts can be instantiated and used just like classes:

```
1 class Main {
2   static public function main() {
3     var a = new AbstractInt(12);
4     trace(a); //12
5   }
6 }
```

As mentioned before, abstracts are a compile-time feature, so it is interesting to see what the above actually generates. A suitable target for this is JavaScript, which tends to generate concise and clean code. Compiling the above (using `haxe -main MyAbstract -js myabstract.js`) shows this JavaScript code:

```
1 var a = 12;
2 console.log(a);
```

The abstract type `Abstract` completely disappeared from the output and all that is left is a value of its underlying type, `Int`. This is because the constructor of `Abstract` is inlined - something we shall learn about later in the section [Inline](#) (Section 4.4.2) - and its inlined expression assigns

a value to `this`. This might be surprising when thinking in terms of classes. However, it is precisely what we want to express in the context of abstracts. Any *inlined member method* of an abstract can assign to `this`, and thus modify the “internal value”.

A good question at this point is “What happens if a member function is not declared inline” because the code obviously has to go somewhere. Haxe creates a private class, known to be the *implementation class*, which has all the abstract member functions as static functions accepting an additional first argument `this` of the underlying type. While technically this is an implementation detail, it can be used for selective functions (2.8.4).

#### Trivia: Basic Types and abstracts

Before the advent of abstract types, all basic types were implemented as extern classes or enums. While this nicely took care of some aspects such as `Int` being a “child class” of `Float`, it caused issues elsewhere. For instance, with `Float` being an extern class, it would unify with the empty structure `{}`, making it impossible to constrain a type to accepting only real objects.

### 2.8.1 Implicit Casts

Unlike classes, abstracts allow defining implicit casts. There are two kinds of implicit casts:

**Direct:** Allows direct casting of the abstract type to or from another type. This is defined by adding `to` and `from` rules to the abstract type and is only allowed for types which unify with the underlying type of the abstract.

**Class field:** Allows casting via calls to special cast functions. These functions are defined using `@:to` and `@:from` metadata. This kind of cast is allowed for all types.

The following code example shows an example of *direct* casting:

```
1 abstract MyAbstract(Int) from Int to Int {
2     inline function new(i:Int) {
3         this = i;
4     }
5 }
6
7 class Main {
8     static public function main() {
9         var a:MyAbstract = 12;
10        var b:Int = a;
11    }
12 }
```

We declare `MyAbstract` as being `from Int` and `to Int`, meaning it can be assigned from `Int` and assigned to `Int`. This is shown in lines 9 and 10, where we first assign the `Int` 12 to variable `a` of type `MyAbstract` (this works due to the `from Int` declaration) and then that abstract back to variable `b` of type `Int` (this works due to the `to Int` declaration).

Class field casts have the same semantics, but are defined completely differently:

```
1 abstract MyAbstract(Int) {
2     inline function new(i:Int) {
3         this = i;
4     }
```



```

5
6  @:from
7  static public function fromString(s:String) {
8      return new MyAbstract(Std.parseInt(s));
9  }
10
11  @:to
12  public function toArray() {
13      return [this];
14  }
15 }
16
17 class Main {
18     static public function main() {
19         var a:MyAbstract = "3";
20         var b:Array<Int> = a;
21         trace(b); // [3]
22     }
23 }

```

By adding `@:from` to a static function, that function qualifies as implicit cast function from its argument type to the abstract. These functions must return a value of the abstract type. They must also be declared `static`.

Similarly, adding `@:to` to a function qualifies it as implicit cast function from the abstract to its return type. These functions are typically member-functions but they can be made `static` and then serve as selective function (2.8.4).

In the example the method `fromString` allows the assignment of value `"3"` to variable `a` of type `MyAbstract` while the method `toArray` allows assigning that abstract to variable `b` of type `Array<Int>`.

When using this kind of cast, calls to the cast-functions are inserted where required. This becomes obvious when looking at the JavaScript output:

```

1 var a = _ImplicitCastField.MyAbstract_Impl_.fromString("3");
2 var b = _ImplicitCastField.MyAbstract_Impl_.toArray(a);

```

This can be further optimized by inlining (4.4.2) both cast functions, turning the output into the following:

```

1 var a = Std.parseInt("3");
2 var b = [a];

```

The *selection algorithm* when assigning a type `A` to a type `B` with at least one of them being an abstract is simple:

1. If `A` is not an abstract, go to 3.
2. If `A` defines a *to*-conversions that admits `B`, go to 6.
3. If `B` is not an abstract, go to 5.
4. If `B` defines a *from*-conversions that admits `A`, go to 6.
5. Stop, unification fails.
6. Stop, unification succeeds.

please review your use of "this" and try to vary somewhat to avoid too much word repetition




Figure 2.1: Selection algorithm flow chart.

By design, implicit casts are *not transitive*, as the following example shows:

```

1 abstract A(Int) {
2   public function new() this = 0;
3   @:to public function toB() return new B();
4 }
5
6 abstract B(Int) {
7   public function new() this = 0;
8   @:to public function toC() return new C();
9 }
10
11 abstract C(Int) {
12   public function new() this = 0;
13 }
14
15 class Main {
16   static public function main() {
17     var a = new A();
18     var b:B = a; // valid, uses A.toB
19     var c:C = b; // valid, uses B.toC
20     var c:C = a; // error, A should be C
21   }
22 }
  
```

While the individual casts from A to B and from B to C are allowed, a transitive cast from A to C is not. This is to avoid ambiguous cast-paths and retain a simple selection algorithm.

## 2.8.2 Operator Overloading

Abstracts allow overloading of unary and binary operators by adding the `@:op` metadata to class fields:

```
1 abstract MyAbstract(String) {
2   public inline function new(s:String) {
3     this = s;
4   }
5
6   @:op(A * B)
7   public function repeat(rhs:Int):MyAbstract {
8     var s:StringBuf = new StringBuf();
9     for (i in 0...rhs)
10       s.add(this);
11     return new MyAbstract(s.toString());
12   }
13 }
14
15 class Main {
16   static public function main() {
17     var a = new MyAbstract("foo");
18     trace(a * 3); // foofoofoo
19   }
20 }
```

By defining `@:op(A * B)`, the function `repeat` serves as operator method for the multiplication `*` operator when the type of the left value is `MyAbstract` and the type of the right value is `Int`. The usage is shown in line 17, which turns into this when compiled to JavaScript:

```
1 console.log(_AbstractOperatorOverload.
2   MyAbstract_Impl_.repeat(a, 3));
```

Similar to implicit casts with class fields (2.8.1), a call to the overload method is inserted where required.

The example `repeat` function is not commutative: While `MyAbstract * Int` works, `Int * MyAbstract` does not. If this should be allowed as well, the `@:commutative` metadata can be added. If it should work *only* for `Int * MyAbstract`, but not for `MyAbstract * Int`, the overload method can be made static, accepting `Int` and `MyAbstract` as first and second type respectively.

Overloading unary operators is analogous:

```
1 abstract MyAbstract(String) {
2   public inline function new(s:String) {
3     this = s;
4   }
5
6   @:op(++A) public function pre()
7     return "pre" + this;
8   @:op(A++) public function post()
9     return this + "post";
10 }
11
12 class Main {
13   static public function main() {
```

```

14     var a = new MyAbstract("foo");
15     trace(++a); // prefoo
16     trace(a++); // foopost
17 }
18 }

```

Both binary and unary operator overloads can return any type.

**Exposing underlying type operations** It is also possible to omit the method body of a `@:op` function, but only if the underlying type of the abstract allows the operation in question and if the resulting type can be assigned back to the abstract.

```

1 abstract MyAbstractInt(Int) from Int to Int {
2     // The following line exposes the (A > B) operation from the
3     // underlying Int
4     // type. Note that no function body is used:
5     @:op(A > B) static function gt( a:MyAbstractInt, b:MyAbstractInt ) :
6         Bool;
7 }
8
9 class Main {
10     static function main() {
11         var a:MyAbstractInt = 42;
12         if(a > 0) trace('Works fine, > operation implemented!');
13
14         // The < operator is not implemented.
15         // This will cause an 'Cannot compare MyAbstractInt and Int' error
16         :
17         if(a < 100) { }
18     }
19 }

```

please review for correctness

### 2.8.3 Array Access

Array access describes the particular syntax traditionally used to access the value in an array at a certain offset. This is usually only allowed with arguments of type `Int`. Nevertheless, with abstracts it is possible to define custom array access methods. The Haxe Standard Library (10) uses this in its `Map` type, where the following two methods can be found:

You have marked "Map" for some reason

```

1 @:arrayAccess
2 public inline function get(key:K) {
3     return this.get(key);
4 }
5 @:arrayAccess
6 public inline function arrayWrite(k:K, v:V):V {
7     this.set(k, v);
8     return v;
9 }

```

There are two kinds of array access methods:

- If an `@:arrayAccess` method accepts one argument, it is a getter.

- If an `@:arrayAccess` method accepts two arguments, it is a setter.

The methods `get` and `arrayWrite` seen above then allow the following usage:

```
1 class Main {
2   public static function main() {
3     var map = new Map();
4     map["foo"] = 1;
5     trace(map["foo"]);
6   }
7 }
```

At this point it should not be surprising to see that calls to the array access fields are inserted in the output:

```
1 map.set("foo", 1);
2 console.log(map.get("foo")); // 1
```

**Order of array access resolving** Due to a bug in Haxe versions before 3.2 the order of checked `:arrayAccess` fields was undefined. This was fixed for Haxe 3.2 so that the fields are now consistently checked from top to bottom:

```
1 abstract AString(String) {
2   public function new(s) this = s;
3   @:arrayAccess function getInt1(k:Int) {
4     return this.charAt(k);
5   }
6   @:arrayAccess function getInt2(k:Int) {
7     return this.charAt(k).toUpperCase();
8   }
9 }
10
11 class Main {
12   static function main() {
13     var a = new AString("foo");
14     trace(a[0]); // f
15   }
16 }
```

The array access `a[0]` is resolved to the `getInt1` field, leading to lower case `f` being returned. The result might be different in Haxe versions before 3.2.

Fields which are defined earlier take priority even if they require an implicit cast (2.8.1).

## 2.8.4 Selective Functions

Since the compiler promotes abstract member functions to static functions, it is possible to define static functions by hand and use them on an abstract instance. The semantics here are similar to those of static extensions (6.3), where the type of the first function argument determines for which types a function is defined:

```
1 abstract MyAbstract<T>(T) from T {
2   public function new(t:T) this = t;
3
4   function get() return this;
```

```

5
6  @:impl
7  static public function getString(v:MyAbstract<String>):String {
8      return v.get();
9  }
10 }
11
12 class Main {
13     static public function main() {
14         var a = new MyAbstract("foo");
15         a.getString();
16         var b = new MyAbstract(1);
17         // Int should be MyAbstract<String>
18         b.getString();
19     }
20 }

```

The method `getString` of abstract `MyAbstract` is defined to accept a first argument of `MyAbstract<String>`. This causes it to be available on variable `a` on line 14 (because the type of `a` is `MyAbstract<String>`), but not on variable `b` whose type is `MyAbstract<Int>`.

#### Trivia: Accidental Feature

Rather than having actually been designed, selective functions were discovered. After the idea was first mentioned, it required only minor adjustments in the compiler to make them work. Their discovery also led to the introduction of multi-type abstracts, such as `Map`.

## 2.8.5 Enum abstracts

Since Haxe 3.1.0

By adding the `:enum` metadata to an abstract definition, that abstract can be used to define finite value sets:

```

1  @:enum
2  abstract HttpStatus(Int) {
3      var NotFound = 404;
4      var MethodNotAllowed = 405;
5  }
6
7  class Main {
8      static public function main() {
9          var status = HttpStatus.NotFound;
10         var msg = printStatus(status);
11     }
12
13     static function printStatus(status:HttpStatus) {
14         return switch(status) {
15             case NotFound:
16                 "Not found";
17             case MethodNotAllowed:
18                 "Method not allowed";

```

```

19     }
20 }
21 }

```

The Haxe Compiler replaces all field access to the `HttpStatus` abstract with their values, as evident in the JavaScript output:

```

1 Main.main = function() {
2     var status = 404;
3     var msg = Main.printStatus(status);
4 };
5 Main.printStatus = function(status) {
6     switch(status) {
7     case 404:
8         return "Not found";
9     case 405:
10        return "Method not allowed";
11    }
12 };

```

This is similar to accessing variables declared as inline (4.4.2), but has several advantages:

- The typer can ensure that all values of the set are typed correctly.
- The pattern matcher checks for exhaustiveness (6.4.10) when matching (6.4) an enum abstract.
- Defining fields requires less syntax.

## 2.8.6 Forwarding abstract fields

Since Haxe 3.1.0

When wrapping an underlying type, it is sometimes desirable to “keep” parts of its functionality. Because writing forwarding functions by hand is cumbersome, Haxe allows adding the `:forward` metadata to an abstract type:

```

1 @:forward(push, pop)
2 abstract MyArray<S>(Array<S>) {
3     public inline function new() {
4         this = [];
5     }
6 }
7
8 class Main {
9     static public function main() {
10         var myArray = new MyArray();
11         myArray.push(12);
12         myArray.pop();
13         // MyArray<Int> has no field length
14         //myArray.length;
15     }
16 }

```

The `MyArray` abstract in this example wraps `Array`. Its `:forward` metadata has two arguments which correspond to the field names to be forwarded to the underlying type. In this example, the `main` method instantiates `MyArray` and accesses its `push` and `pop` methods. The commented line demonstrates that the `length` field is not available.

As usual we can look at the JavaScript output to see how the code is being generated:

```
1 Main.main = function() {  
2     var myArray = [];  
3     myArray.push(12);  
4     myArray.pop();  
5 };
```

It is also possible to use `:forward` without any arguments in order to forward all fields. Of course the Haxe Compiler still ensures that the field actually exists on the underlying type.

#### Trivia: Implemented as macro

Both the `:enum` and `:forward` functionality were originally implemented using build macros (9.5). While this worked nicely in non-macro code, it caused issues if these features were used from within macros. The implementation was subsequently moved to the compiler.

## 2.8.7 Core-type abstracts

The Haxe Standard Library defines a set of basic types as core-type abstracts. They are identified by the `:coreType` metadata and the lack of an underlying type declaration. These abstracts can still be understood to represent a different type. Still, that type is native to the Haxe target.

Introducing custom core-type abstracts is rarely necessary in user code as it requires the Haxe target to be able to make sense of it. However, there could be interesting use-cases for authors of macros and new Haxe targets.

In contrast to opaque abstracts, core-type abstracts have the following properties:

- They have no underlying type.
- They are considered nullable unless they are annotated with `:notNull` metadata.
- They are allowed to declare array access (2.8.3) functions without expressions.
- Operator overloading fields (2.8.2) that have no expression are not forced to adhere to the Haxe type semantics.

## 2.9 Monomorph

A monomorph is a type which may, through unification (3.5), morph into a different type later. We shall see details about this type when talking about type inference (3.6).



# Chapter 3

## Type System

We learned about the different kinds of types in [Types \(Chapter 2\)](#) and it is now time to see how they interact with each other. We start off easy by introducing `typedef` ([3.1](#)), a mechanism to give a name (or alias) to a more complex type. Among other things, this will come in handy when working with types having type parameters ([3.2](#)).

A lot of type-safety is achieved by checking if two given types of the type groups above are compatible. Meaning, the compiler tries to perform *unification* between them as detailed in [Unification \(Section 3.5\)](#).

All types are organized in *modules* and can be addressed through *paths*. [Modules and Paths \(Section 3.7\)](#) will give a detailed explanation of the related mechanics.

### 3.1 Typedef

We briefly looked at typedefs while talking about anonymous structures ([2.5](#)) and saw how we could shorten a complex structure type ([2.5](#)) by giving it a name. This is precisely what typedefs are good for. Giving names to structure types might even be considered their primary use. In fact, it is so common that the distinction appears somewhat blurry and many Haxe users consider typedefs to actually *be* the structure.

A typedef can give a name to any other type:

```
1 typedef IA = Array<Int>;
```

This enables us to use `IA` in places where we would normally use `Array<Int>`. While this saves only a few keystrokes in this particular case, it can make a much bigger difference for more complex, compound types. Again, this is why typedef and structures seem so connected:

```
1 typedef User = {  
2     var age : Int;  
3     var name : String;  
4 }
```

A typedef is not a textual replacement but actually a real type. It can even have type parameters ([3.2](#)) as the `Iterable` type from the Haxe Standard Library demonstrates:

```
1 typedef Iterable<T> = {  
2     function iterator() : Iterator<T>;  
3 }
```

**Optional fields** Mark the field of a structure as optional using the `@:optional` metadata.

```
1 typedef User = {
2     var age : Int;
3     var name : String;
4     @:optional var phoneNumber : String;
5 }
```

### 3.1.1 Extensions

Extensions are used to express that a structure has all the fields of a given type as well as some additional fields of its own:

```
1 typedef IterableWithLength<T> = {
2     > Iterable<T>,
3     // read only property
4     var length(default, null):Int;
5 }
6
7 class Main {
8     static public function main() {
9         var array = [1, 2, 3];
10        var t:IterableWithLength<Int> = array;
11    }
12 }
```

The greater-than operator `>` denotes that an extension of `Iterable<T>` is being created, with the additional class fields following. In this case, a read-only property (4.2) `length` of type `Int` is required.

In order to be compatible with `IterableWithLength<T>`, a type then must be compatible with `Iterable<T>` and also provide a read-only `length` property of type `Int`. The example assigns an `Array`, which happens to fulfill these requirements.

**Since Haxe 3.1.0**

It is also possible to extend multiple structures:

```
1 typedef WithLength = {
2     var length(default, null):Int;
3 }
4
5 typedef IterableWithLengthAndPush<T> = {
6     > Iterable<T>,
7     > WithLength,
8     function push(a:T):Int;
9 }
10
11 class Main {
12     static public function main() {
13         var array = [1, 2, 3];
14         var t:IterableWithLengthAndPush<Int> = array;
15     }
16 }
```

## 3.2 Type Parameters

Haxe allows parametrization of a number of types, as well as class fields (4) and enum constructors (2.4.1). Type parameters are defined by enclosing comma-separated type parameter names in angle brackets <>. A simple example from the Haxe Standard Library is `Array`:

```
1 class Array<T> {
2     function push(x : T) : Int;
3 }
```

Whenever an instance of `Array` is created, its type parameter `T` becomes a monomorph (2.9). That is, it can be bound to any type, but only one at a time. This binding can happen

**explicitly** by invoking the constructor with explicit types (`new Array<String>()`) or

**implicitly** by type inference (3.6), e.g. when invoking `arrayInstance.push("foo")`.

Inside the definition of a class with type parameters, these type parameters are an unspecific type. Unless constraints (3.2.1) are added, the compiler has to assume that the type parameters could be used with any type. As a consequence, it is not possible to access fields of type parameters or cast (5.23) to a type parameter type. It is also not possible to create a new instance of a type parameter type, unless the type parameter is generic (3.3) and constrained accordingly.

The following table shows where type parameters are allowed:

Parameter on	Bound upon	Notes
Class	instantiation	Can also be bound upon member field access.  Allowed for methods and named local lvalue functions.
Enum	instantiation	
Enum Constructor	instantiation	
Function	invocation	
Structure	instantiation	

With function type parameters being bound upon invocation, such a type parameter (if unconstrained) accepts any type. However, only one type per invocation is accepted. This can be utilized if a function has multiple arguments:

```
1 class Main {
2     static public function main() {
3         equals(1, 1);
4         // runtime message: bar should be foo
5         equals("foo", "bar");
6         // compiler error: String should be Int
7         equals(1, "foo");
8     }
9
10    static function equals<T>(expected:T, actual:T) {
11        if (actual != expected) {
12            trace('$actual should be $expected');
13        }
14    }
15 }
```

Both arguments `expected` and `actual` of the `equals` function have type `T`. This implies that for each invocation of `equals` the two arguments must be of the same type. The compiler admits the first call (both arguments being of `Int`) and the second call (both arguments being of `String`) but the third attempt causes a compiler error.

#### Trivia: Type parameters in expression syntax

We often get the question why a method with type parameters cannot be called as `method<String>(x)`. The error messages the compiler gives are not very helpful. However, there is a simple reason for that: The above code is parsed as if both `<` and `>` were binary operators, yielding `(method < String) > (x)`.

### 3.2.1 Constraints

Type parameters can be constrained to multiple types:

```
1 typedef Measurable = {
2   public var length(default, null):Int;
3 }
4
5 class Main {
6   static public function main() {
7     trace(test([]));
8     trace(test(["bar", "foo"]));
9     // String should be Iterable<String>
10    //test("foo");
11  }
12
13  static function test<T:(Iterable<String>, Measurable)>(a:T) {
14    if (a.length == 0) return "empty";
15    return a.iterator().next();
16  }
17 }
```

Type parameter `T` of method `test` is constrained to the types `Iterable<String>` and `Measurable`. The latter is defined using a `typedef` (3.1) for convenience and requires compatible types to have a read-only property (4.2) named `length` of type `Int`. The constraints then say that a type is compatible if

- it is compatible with `Iterable<String>` and
- has a `length`-property of type `Int`.

We can see that invoking `test` with an empty array in line 7 and an `Array<String>` in line 8 works fine. This is because `Array` has both a `length`-property and an `iterator`-method. However, passing a `String` as argument in line 9 fails the constraint check because `String` is not compatible with `Iterable<T>`.

### 3.3 Generic

Usually, the Haxe Compiler generates only a single class or function even if it has type parameters. This results in a natural abstraction where the code generator for the target language has to assume that a type parameter could be of any type. The generated code then might have to perform some type checks which can be detrimental to performance.

A class or function can be made *generic* by attributing it with the `:generic` metadata (6.9). This causes the compiler to emit a distinct class/function per type parameter combination with mangled names. A specification like this can yield a boost in performance-critical code portions on static targets (2.2) at the cost of a larger output size:

```

1 @:generic
2 class MyValue<T> {
3     public var value:T;
4     public function new(value:T) {
5         this.value = value;
6     }
7 }
8
9 class Main {
10     static public function main() {
11         var a = new MyValue<String>("Hello");
12         var b = new MyValue<Int>(42);
13     }
14 }

```

It seems unusual to see the explicit type `MyValue<String>` here as we usually let type inference (3.6) deal with this. Nonetheless, it is indeed required in this case. The compiler has to know the exact type of a generic class upon construction. The JavaScript output shows the result:

```

1 (function () { "use strict";
2 var Test = function() { };
3 Test.main = function() {
4     var a = new MyValue_String("Hello");
5     var b = new MyValue_Int(5);
6 };
7 var MyValue_Int = function(value) {
8     this.value = value;
9 };
10 var MyValue_String = function(value) {
11     this.value = value;
12 };
13 Test.main();
14 }) ();

```

We can identify that `MyValue<String>` and `MyValue<Int>` have become `MyValue_String` and `MyValue_Int` respectively. This is similar for generic functions:

```

1 class Main {
2     static public function main() {
3         method("foo");
4         method(1);
5     }
6
7     @:generic static function method<T>(t:T) { }
8 }

```

Again, the JavaScript output makes it obvious:

```

1 (function () { "use strict";
2 var Main = function() { }
3 Main.method_Int = function(t) {
4 }
5 Main.method_String = function(t) {
6 }
7 Main.main = function() {

```

```

8     Main.method_String("foo");
9     Main.method_Int(1);
10 }
11 Main.main();
12 }) ();

```

### 3.3.1 Construction of generic type parameters

#### Definition: Generic Type Parameter

A type parameter is said to be generic if its containing class or method is generic.

It is not possible to construct normal type parameters, e.g. `new T()` is a compiler error. The reason for this is that Haxe generates only a single function and the construct makes no sense in that case. This is different when the type parameter is generic: Since we know that the compiler will generate a distinct function for each type parameter combination, it is possible to replace the `T new T()` with the real type.

```

1 typedef Constructible = {
2     public function new(s:String):Void;
3 }
4
5 class Main {
6     static public function main() {
7         var s:String = make();
8         var t:haxe.Template = make();
9     }
10
11     @:generic
12     static function make<T:Constructible>():T {
13         return new T("foo");
14     }
15 }

```

It should be noted that top-down inference (3.6.1) is used here to determine the actual type of `T`. There are two requirements for this kind of type parameter construction to work: The constructed type parameter must be

1. generic and
2. be explicitly constrained (3.2.1) to having a constructor (2.3.1).

Here, 1. is given by `make` having the `@:generic` metadata, and 2. by `T` being constrained to `Constructible`. The constraint holds for both `String` and `haxe.Template` as both have a constructor accepting a singular `String` argument. Sure enough, the relevant JavaScript output looks as expected:

```

1 var Main = function() { }
2 Main.__name__ = true;
3 Main.make_haxe_Template = function() {
4     return new haxe.Template("foo");
5 }

```

```

6 Main.make_String = function() {
7   return new String("foo");
8 }
9 Main.main = function() {
10  var s = Main.make_String();
11  var t = Main.make_haxe_Template();
12 }

```

### 3.4 Variance

While variance is also relevant in other places, it occurs particularly often with type parameters and comes as a surprise in this context. Additionally, it is very easy to trigger variance errors:

```

1 class Base {
2   public function new() { }
3 }
4
5 class Child extends Base { }
6
7 class Main {
8   public static function main () {
9     var children = [new Child()];
10    // Array<Child> should be Array<Base>
11    // Type parameters are invariant
12    // Child should be Base
13    var bases:Array<Base> = children;
14  }
15 }

```

Apparently, an `Array<Child>` cannot be assigned to an `Array<Base>`, even though `Child` can be assigned to `Base`. The reason for this might be somewhat unexpected: It is not allowed because arrays can be written to, e.g. via their `push()` method. It is easy to generate problems by ignoring variance errors:

```

1 class Base {
2   public function new() { }
3 }
4
5 class Child extends Base { }
6
7 class OtherChild extends Base { }
8
9 class Main {
10  public static function main () {
11    var children = [new Child()];
12    // subvert type checker
13    var bases:Array<Base> = cast children;
14    bases.push(new OtherChild());
15    for(child in children) {
16      trace(child);
17    }
18  }
19 }

```

19 }

Here we subvert the type checker by using a cast (5.23), thus allowing the assignment after the commented line. With that we hold a reference `bases` to the original array, typed as `Array<Base>`. This allows pushing another type compatible with `Base` (`OtherChild`) onto that array. However, our original reference `children` is still of type `Array<Child>` and things go bad when we encounter the `OtherChild` instance in one of its elements while iterating.

If `Array` had no `push()` method and no other means of modification, the assignment would be safe because no incompatible type could be added to it. In Haxe, we can achieve this by restricting the type accordingly using structural subtyping (3.5.2):

```
1 class Base {
2     public function new() { }
3 }
4
5 class Child extends Base { }
6
7 typedef MyArray<T> = {
8     public function pop():T;
9 }
10
11 class Main {
12     public static function main () {
13         var a = [new Child()];
14         var b:MyArray<Base> = a;
15     }
16 }
```

We can safely assign with `b` being typed as `MyArray<Base>` and `MyArray` only having a `pop()` method. There is no method defined on `MyArray` which could be used to add incompatible types, it is thus said to be *covariant*.

#### Definition: Covariance

A compound type (2) is considered covariant if its component types can be assigned to less specific components, i.e. if they are only read, but never written.

#### Definition: Contravariance

A compound type (2) is considered contravariant if its component types can be assigned to less generic components, i.e. if they are only written, but never read.

## 3.5 Unification

Mention `toString()/String` conversion somewhere in this chapter.

Unification is the heart of the type system and contributes immensely to the robustness of Haxe programs. It describes the process of checking if a type is compatible to another type.

#### Definition: Unification

Unification between two types `A` and `B` is a directional process which answers the question if `A` can be assigned to `B`. It may *mutate* either type if it is or has a monomorph (2.9).



Unification errors are very easy to trigger:

```
1 class Main {  
2   static public function main() {  
3     // Int should be String  
4     var s:String = 1;  
5   }  
6 }
```

We try to assign a value of type `Int` to a variable of type `String`, which causes the compiler to try and *unify* `Int` with `String`. This is, of course, not allowed and makes the compiler emit the error `Int should be String`.

In this particular case, the unification is triggered by an *assignment*, a context in which the “is assignable to” definition is intuitive. It is one of several cases where unification is performed:

**Assignment:** If `a` is assigned to `b`, the type of `a` is unified with the type of `b`.

**Function call:** We have briefly seen this one while introducing the function (2.6) type. In general, the compiler tries to unify the first given argument type with the first expected argument type, the second given argument type with the second expected argument type and so on until all argument types are handled.

**Function return:** Whenever a function has a `return e` expression, the type of `e` is unified with the function return type. If the function has no explicit return type, it is inferred to the type of `e` and subsequent `return` expressions are inferred against it.

**Array declaration:** The compiler tries to find a minimal type between all given types in an array declaration. Refer to [Common Base Type \(Section 3.5.5\)](#) for details.

**Object declaration:** If an object is declared “against” a given type, the compiler unifies each given field type with each expected field type.

**Operator unification:** Certain operators expect certain types which the given types are unified against. For instance, the expression `a && b` unifies both `a` and `b` with `Bool` and the expression `a == b` unifies `a` with `b`.

### 3.5.1 Between Class/Interface

When defining unification behavior between classes, it is important to remember that unification is directional: We can assign a more specialized class (e.g. a child class) to a generic class (e.g. a parent class) but the reverse is not valid.

The following assignments are allowed:

- child class to parent class
- class to implementing interface
- interface to base interface

These rules are transitive, meaning that a child class can also be assigned to the base class of its base class, an interface its base class implements, the base interface of an implementing interface and so on.

“parent class” should probably be used here, but I have no idea what it means, so I will refrain from changing it myself.

## 3.5.2 Structural Subtyping

### Definition: Structural Subtyping

Structural subtyping defines an implicit relation between types that have the same structure.

Structural sub-typing in Haxe is allowed when unifying

- a class (2.3) with a structure (2.5) and
- a structure with another structure.

The following example is part of the `Lambda` class of the Haxe Standard Library (10):

```
1 public static function empty<T>(it : Iterable<T>):Bool {  
2     return !it.iterator().hasNext();  
3 }
```

The `empty`-method checks if an `Iterable` has an element. For this purpose, it is not necessary to know anything about the argument type other than the fact that it is considered an iterable. This allows calling the `empty`-method with any type that unifies with `Iterable<T>` which applies to a lot of types in the Haxe Standard Library.

This kind of typing can be very convenient but extensive use may be detrimental to performance on static targets, which is detailed in [Impact on Performance](#) (Section 2.5.4).

## 3.5.3 Monomorphs

Unification of types having or being a monomorph (2.9) is detailed in [Type Inference](#) (Section 3.6).

## 3.5.4 Function Return

Unification of function return types may involve the `Void`-type (2.1.5) and requires a clear definition of what unifies with `Void`. With `Void` describing the absence of a type, it is not assignable to any other type, not even `Dynamic`. This means that if a function is explicitly declared as returning `Dynamic`, it cannot return `Void`.

The opposite applies as well: If a function declares a return type of `Void`, it cannot return `Dynamic` or any other type. However, this direction of unification is allowed when assigning function types:

```
1 var func:Void->Void = function() return "foo";
```

The right-hand function clearly is of type `Void->String`, yet we can assign it to the variable `func` of type `Void->Void`. This is because the compiler can safely assume that the return type is irrelevant, given that it could not be assigned to any non-`Void` type.

## 3.5.5 Common Base Type

Given a set of multiple types, a *common base type* is a type which all types of the set unify against:

```
1 class Base {  
2     public function new() { }  
3 }  
4  
5 class Child1 extends Base { }
```

```

6 class Child2 extends Base { }
7
8 class Main {
9     static public function main() {
10         var a = [new Child1(), new Child2()];
11         $type(a); // Array<Base>
12     }
13 }

```

Although `Base` is not mentioned, the Haxe Compiler manages to infer it as the common type of `Child1` and `Child2`. The Haxe Compiler employs this kind of unification in the following situations:

- array declarations
- if/else
- cases of a switch

## 3.6 Type Inference

The effects of type inference have been seen throughout this document and will continue to be important. A simple example shows type inference at work:

```

1 class Main {
2     public static function main() {
3         var x = null;
4         $type(x); // Unknown<0>
5         x = "foo";
6         $type(x); // String
7     }
8 }

```

The special construct `$type` was previously mentioned in order to simplify the explanation of the [Function Type](#) (Section 2.6) type, so let us now introduce it officially:

### Construct: `$type`

`$type` is a compile-time mechanism being called like a function, with a single argument. The compiler evaluates the argument expression and then outputs the type of that expression.

In the example above, the first `$type` prints `Unknown<0>`. This is a monomorph (2.9), a type that is not yet known. The next line `x = "foo"` assigns a `String` literal to `x`, which causes the unification (3.5) of the monomorph with `String`. We then see that the type of `x` indeed has changed to `String`.

Whenever a type other than [Dynamic](#) (Section 2.7) is unified with a monomorph, that monomorph *becomes* that type: it *morphs* into that type. Therefore it cannot morph into a different type afterwards, a property expressed in the *mono* part of its name.

Following the rules of unification, type inference can occur in compound types:

```

1 class Main {
2     public static function main() {
3         var x = [];
4         $type(x); // Array<Unknown<0>>

```

```

5     x.push("foo");
6     $type(x); // Array<String>
7 }
8 }

```

Variable `x` is first initialized to an empty `Array`. At this point we can tell that the type of `x` is an array, but we do not yet know the type of the array elements. Consequentially, the type of `x` is `Array<Unknown<0>>`. It is only after pushing a `String` onto the array that we know the type to be `Array<String>`.

### 3.6.1 Top-down Inference

Most of the time, types are inferred on their own and may then be unified with an expected type. In a few places, however, an expected type may be used to influence inference. We then speak of *top-down inference*.

#### Definition: Expected Type

Expected types occur when the type of an expression is known before that expression has been typed, e.g. because the expression is argument to a function call. They can influence typing of that expression through what is called top-down inference (3.6.1).

A good example are arrays of mixed types. As mentioned in [Dynamic](#) (Section 2.7), the compiler refuses `[1, "foo"]` because it cannot determine an element type. Employing top-down inference, this can be overcome:

```

1 class Main {
2     static public function main() {
3         var a:Array<Dynamic> = [1, "foo"];
4     }
5 }

```

Here, the compiler knows while typing `[1, "foo"]` that the expected type is `Array<Dynamic>`, so the element type is `Dynamic`. Instead of the usual unification behavior where the compiler would attempt (and fail) to determine a common base type (3.5.5), the individual elements are typed against and unified with `Dynamic`.

We have seen another interesting use of top-down inference when construction of generic type parameters (3.3.1) was introduced:

```

1 typedef Constructible = {
2     public function new(s:String):Void;
3 }
4
5 class Main {
6     static public function main() {
7         var s:String = make();
8         var t:haxe.Template = make();
9     }
10
11     @:generic
12     static function make<T:Constructible>():T {
13         return new T("foo");
14     }
15 }

```

The explicit types `String` and `haxe.Template` are used here to determine the return type of `make`. This works because the method is invoked as `make()`, so we know the return type will be assigned to the variables. Utilizing this information, it is possible to bind the unknown type `T` to `String` and `haxe.Template` respectively.

### 3.6.2 Limitations

Type inference saves a lot of manual type hints when working with local variables, but sometimes the type system still needs some help. In fact, it does not even try to infer the type of a variable (4.1) or property (4.2) field unless it has a direct initialization.

There are also some cases involving recursion where type inference has limitations. If a function calls itself recursively while its type is not (completely) known yet, type inference may infer a wrong, too specialized type.

A different kind of limitation involves the readability of code. If type inference is overused it might be difficult to understand parts of a program due to the lack of visible types. This is particularly true for method signatures. It is recommended to find a good balance between type inference and explicit type hints.

## 3.7 Modules and Paths

### Definition: Module

All Haxe code is organized in modules, which are addressed using paths. In essence, each `.hx` file represents a module which may contain several types. A type may be `private`, in which case only its containing module can access it.

The distinction of a module and its containing type of the same name is blurry by design. In fact, addressing `haxe.ds.StringMap<Int>` can be considered shorthand for `haxe.ds.StringMap.StringMap<Int>`. The latter version consists of four parts:

1. the package `haxe.ds`
2. the module name `StringMap`
3. the type name `StringMap`
4. the type parameter `Int`

If the module and type name are equal, the duplicate can be removed, leading to the `haxe.ds.StringMap<Int>` short version. However, knowing about the extended version helps with understanding how module sub-types (3.7.1) are addressed.

Paths can be shortened further by using an `import` (3.7.2), which typically allows omitting the package part of a path. This may lead to usage of unqualified identifiers, for which understanding the resolution order (3.7.3) is required.

### Definition: Type path

The (dot-)path to a type consists of the package, the module name and the type name. Its general form is `pack1.pack2.packN.ModuleName.TypeName`.

### 3.7.1 Module Sub-Types

A module sub-type is a type declared in a module with a different name than that module. This allows a single .hx file to contain multiple types, which can be accessed unqualified from within the module, and by using `package.Module.Type` from other modules:

```
1 var e:haxe.macro.Expr.ExprDef;
```

Here the sub-type `ExprDef` within module `haxe.macro.Expr` is accessed.

The sub-type relation is not reflected at run-time. That is, public sub-types become a member of their containing package, which could lead to conflicts if two modules within the same package tried to define the same sub-type. Naturally, the Haxe compiler detects these cases and reports them accordingly. In the example above `ExprDef` is generated as `haxe.macro.ExprDef`.

Sub-types can also be made private:

```
1 private class C { ... }
2 private enum E { ... }
3 private typedef T { ... }
4 private abstract A { ... }
```

#### Definition: Private type

A type can be made private by using the `private` modifier. As a result, the type can only be directly accessed from within the module (3.7) it is defined in.

Private types, unlike public ones, do not become a member of their containing package.

The accessibility of types can be controlled more fine-grained by using access control (6.10).

### 3.7.2 Import

If a type path is used multiple times in a .hx file, it might make sense to use an `import` to shorten it. This allows omitting the package when using the type:

```
1 import haxe.ds.StringMap;
2
3 class Main {
4     static public function main() {
5         // instead of: new haxe.ds.StringMap();
6         new StringMap();
7     }
8 }
```

With `haxe.ds.StringMap` being imported in the first line, the compiler is able to resolve the unqualified identifier `StringMap` in the `main` function to this package. The module `StringMap` is said to be *imported* into the current file.

In this example, we are actually importing a *module*, not just a specific type within that module. This means that all types defined within the imported module are available:

```
1 import haxe.macro.Expr;
2
3 class Main {
4     static public function main() {
5         var e:Binop = OpAdd;
6     }
7 }
```

The type `Binop` is an enum (2.4) declared in the module `haxe.macro.Expr`, and thus available after the import of said module. If we were to import only a specific type of that module, e.g. `import haxe.macro.Expr.ExprDef`, the program would fail to compile with `Class not found : Binop`.

There are several aspects worth knowing about importing:

- The bottommost import takes priority (detailed in [Resolution Order \(Section 3.7.3\)](#)).
- The static extension (6.3) keyword `using` implies the effect of `import`.
- If an enum is imported (directly or as part of a module import), all its enum constructors (2.4.1) are also imported (this is what allows the `OpAdd` usage in the above example).

Furthermore, it is also possible to import static fields (4) of a class and use them unqualified:

```
1 import Math.random;
2
3 class Main {
4     static public function main() {
5         random();
6     }
7 }
```

Special care has to be taken with field names or local variable names that conflict with a package name: Since they take priority over packages, a local variable named `haxe` blocks off usage the entire `haxe` package.

**Wildcard import** Haxe allows using `*` to allow import of all modules in a package, all types in a module or all static fields in a type. It is important to understand that this kind of import only crosses a single level as we can see in the following example:

```
1 import haxe.macro.*;
2
3 class Main {
4     static function main() {
5         var expr:Expr = null;
6         //var expr:ExprDef = null; // Class not found : ExprDef
7     }
8 }
```

Using the wildcard import on `haxe.macro` allows accessing `Expr` which is a module in this package, but it does not allow accessing `ExprDef` which is a sub-type of the `Expr` module. This rule extends to static fields when a module is imported.

When using wildcard imports on a package the compiler does not eagerly process all modules in that package. This means that these modules are never actually seen by the compiler unless used explicitly and are then not part of the generated output.

**Import with alias** If a type or static field is used a lot in an importing module it might help to alias it to a shorter name. This can also be used to disambiguate conflicting names by giving them a unique identifier.

```
1 import String.fromCharCode in f;
2
3 class Main {
4     static function main() {
```

```

5     var c1 = f(65);
6     var c2 = f(66);
7     trace(c1 + c2); // AB
8 }
9 }

```

Here we import `String.fromCharCode` as `f` which allows us to use `f(65)` and `f(66)`. While the same could be achieved with a local variable, this method is compile-time exclusive and guaranteed to have no run-time overhead.

**Since Haxe 3.2.0**

Haxe also allows the more natural `as` in place of `in`.

### 3.7.3 Resolution Order

Resolution order comes into play as soon as unqualified identifiers are involved. These are expressions (5) in the form of `foo()`, `foo = 1` and `foo.field`. The last one in particular includes module paths such as `haxe.ds.StringMap`, where `haxe` is an unqualified identifier.

We describe the resolution order algorithm here, which depends on the following state:

- the declared local variables (5.10) (including function arguments)
- the imported (3.7.2) modules, types and statics
- the available static extensions (6.3)
- the kind (static or member) of the current field
- the declared member fields on the current class and its parent classes
- the declared static fields on the current class
- the expected type (3.6.1)
- the expression being `untyped` or not

Given an identifier `i`, the algorithm is as follows:

1. If `i` is `true`, `false`, `this`, `super` or `null`, resolve to the matching constant and halt.
2. If a local variable named `i` is accessible, resolve to it and halt.
3. If the current field is static, go to 6.
4. If the current class or any of its parent classes has a field named `i`, resolve to it and halt.
5. If a static extension with a first argument of the type of the current class is available, resolve to it and halt.
6. If the current class has a static field named `i`, resolve to it and halt.
7. If an enum constructor named `i` is declared on an imported enum, resolve to it and halt.
8. If a static named `i` is explicitly imported, resolve to it and halt.
9. If `i` starts with a lower-case character, go to 11.

proper label and caption + code/identifier styling for diagram



10. If a type named `i` is available, resolve to it and halt.
11. If the expression is not in untyped mode, go to 14
12. If `i` equals `__this__`, resolve to the `this` constant and halt.
13. Generate a local variable named `i`, resolve to it and halt.
14. Fail

For step 10, it is also necessary to define the resolution order of types:

1. If a type named `i` is imported (directly or as part of a module), resolve to it and halt.
2. If the current package contains a module named `i` with a type named `i`, resolve to it and halt.
3. If a type named `i` is available at top-level, resolve to it and halt.
4. Fail

For step 1 of this algorithm as well as steps 5 and 7 of the previous one, the order of import resolution is important:

- Imported modules and static extensions are checked from bottom to top with the first match being picked.
- Within a given module, types are checked from top to bottom.
- For imports, a match is made if the name equals.
- For static extensions (6.3), a match is made if the name equals and the first argument unifies (3.5). Within a given type being used as static extension, the fields are checked from top to bottom.




Figure 3.1: Resolution order of identifier 'i'

## Chapter 4

# Class Fields

### Definition: Class Field

A class field is a variable, property or method of a class which can either be static or non-static. Non-static fields are referred to as *member* fields, so we speak of e.g. a *static method* or a *member variable*.

So far we have seen how types and Haxe programs in general are structured. This section about class fields concludes the structural part and at the same time bridges to the behavioral part of Haxe. This is because class fields are the place where expressions (5) are at home.

There are three kinds of class fields:

**Variable:** A variable (4.1) class field holds a value of a certain type, which can be read or written.

**Property:** A property (4.2) class field defines a custom access behavior for something that, outside the class, looks like a variable field.

**Method:** A method (4.3) is a function which can be called to execute code.

Strictly speaking, a variable could be considered to be a property with certain access modifiers. Indeed, the Haxe Compiler does not distinguish variables and properties during its typing phase, but they remain separated at syntax level.

Regarding terminology, a method is a (static or non-static) function belonging to a class. Other functions, such as a local functions (5.11) in expressions, are not considered methods.

## 4.1 Variable

We have already seen variable fields in several code examples of previous sections. Variable fields hold values, a characteristic which they share with most (but not all) properties:

```
1 class Main {
2     static var member:String = "bar";
3
4     public static function main() {
5         trace(member);
6         member = "foo";
7         trace(member);
8     }
9 }
```

We can learn from this that a variable

1. has a name (here: `member`),
2. has a type (here: `String`),
3. may have a constant initialization (here: `"bar"`) and
4. may have access modifiers (4.4) (here: `static`)

The example first prints the initialization value of `member`, then sets it to `"foo"` before printing its new value. The effect of access modifiers is shared by all three class field kinds and explained in a separate section.

It should be noted that the explicit type is not required if there is an initialization value. The compiler will infer (3.6) it in this case.




Figure 4.1: Initialization values of a variable field.

## 4.2 Property

Next to variables (4.1), properties are the second option for dealing with data on a class. Unlike variables however, they offer more control of which kind of field access should be allowed and how it should be generated. Common use cases include:

- Have a field which can be read from anywhere, but only be written from within the defining class.

- Have a field which invokes a *getter*-method upon read-access.
- Have a field which invokes a *setter*-method upon write-access.

When dealing with properties, it is important to understand the two kinds of access:

**Definition: Read Access**

A read access to a field occurs when a right-hand side field access expression (5.7) is used. This includes calls in the form of `obj.field()`, where `field` is accessed to be read.

**Definition: Write Access**

A write access to a field occurs when a field access expression (5.7) is assigned a value in the form of `obj.field = value`. It may also occur in combination with read access (4.2) for special assignment operators such as `+=` in expressions like `obj.field += value`.

Read access and write access are directly reflected in the syntax, as the following example shows:

```
1 class Main {
2   public var x(default, null):Int;
3   static public function main() { }
4 }
```

For the most part, the syntax is similar to variable syntax, and the same rules indeed apply. Properties are identified by

- the opening parenthesis ( after the field name,
- followed by a special *access identifier* (here: `default`),
- with a comma , separating
- another special access identifier (here: `null`)
- before a closing parenthesis ).

The access identifiers define the behavior when the field is read (first identifier) and written (second identifier). The accepted values are:

**default:** Allows normal field access if the field has public visibility, otherwise equal to `null` access.

**null:** Allows access only from within the defining class.

**get/set:** Access is generated as a call to an *accessor method*. The compiler ensures that the accessor is available.

**dynamic:** Like `get/set` access, but does not verify the existence of the accessor field.

**never:** Allows no access at all.

**Definition: Accessor method**

An *accessor method* (or short *accessor*) for a field named `field` of type `T` is a *getter* named `get_field` of type `Void->T` or a *setter* named `set_field` of type `T->T`.

### Trivia: Accessor names

In Haxe 2, arbitrary identifiers were allowed as access identifiers and would lead to custom accessor method names to be admitted. This made parts of the implementation quite tricky to deal with. In particular, `Reflect.getProperty()` and `Reflect.setProperty()` had to assume that any name could have been used, requiring the target generators to generate meta-information and perform lookups.

We disallowed these identifiers and went for the `get_` and `set_` naming convention which greatly simplified implementation. This was one of the breaking changes between Haxe 2 and 3.

## 4.2.1 Common accessor identifier combinations

The next example shows common access identifier combinations for properties:

```
1 class Main {
2     // read from outside, write only within Main
3     public var ro(default, null):Int;
4
5     // write from outside, read only within Main
6     public var wo(null, default):Int;
7
8     // access through getter get_x and setter
9     // set_x
10    public var x(get, set):Int;
11
12    // read access through getter, no write
13    // access
14    public var y(get, never):Int;
15
16    // required by field x
17    function get_x() return 1;
18
19    // required by field x
20    function set_x(x) return x;
21
22    // required by field y
23    function get_y() return 1;
24
25    function new() {
26        var v = x;
27        x = 2;
28        x += 1;
29    }
30
31    static public function main() {
32        new Main();
33    }
34 }
```

The JavaScript output helps understand what the field access in the `main`-method is compiled to:

```

1 var Main = function() {
2   var v = this.get_x();
3   this.set_x(2);
4   var _g = this;
5   _g.set_x(_g.get_x() + 1);
6 };

```

As specified, the read access generates a call to `get_x()`, while the write access generates a call to `set_x(2)` where 2 is the value being assigned to `x`. The way the `+=` is being generated might look a little odd at first, but can easily be justified by the following example:

```

1 class Main {
2   public var x(get, set):Int;
3   function get_x() return 1;
4   function set_x(x) return x;
5
6   public function new() { }
7
8   static public function main() {
9     new Main().x += 1;
10  }
11 }

```

What happens here is that the expression part of the field access to `x` in the `main` method is *complex*: It has potential side-effects, such as the construction of `Main` in this case. Thus, the compiler cannot generate the `+=` operation as `new Main().x = new Main().x + 1` and has to cache the complex expression in a local variable:

```

1 Main.main = function() {
2   var _g = new Main();
3   _g.set_x(_g.get_x() + 1);
4 }

```

## 4.2.2 Impact on the type system

The presence of properties has several consequences on the type system. Most importantly, it is necessary to understand that properties are a compile-time feature and thus *require the types to be known*. If we were to assign a class with properties to `Dynamic`, field access would *not* respect accessor methods. Likewise, access restrictions no longer apply and all access is virtually public.

When using `get` or `set` access identifier, the compiler ensures that the getter and setter actually exists. The following problem does not compile:

```

1 class Main {
2   // Method get_x required by property x is missing
3   public var x(get, null):Int;
4   static public function main() {}
5 }

```

The method `get_x` is missing, but it need not be declared on the class defining the property itself as long as a parent class defines it:

```

1 class Base {
2   public function get_x() return 1;
3 }
4

```

```

5 class Main extends Base {
6   // ok, get_x is declared by parent class
7   public var x(get, null):Int;
8
9   static public function main() {}
10 }

```

The dynamic access modifier works exactly like `get` or `set`, but does not check for the existence

### 4.2.3 Rules for getter and setter

Visibility of accessor methods has no effect on the accessibility of its property. That is, if a property is `public` and defined to have a getter, that getter may be defined as `private` regardless.

Both getter and setter may access their physical field for data storage. The compiler ensures that this kind of field access does not go through the accessor method if made from within the accessor method itself, thus avoiding infinite recursion:

```

1 class Main {
2   public var x(default, set):Int;
3
4   function set_x(newX) {
5     return x = newX;
6   }
7
8   static public function main() {}
9 }

```

However, the compiler assumes that a physical field exists only if at least one of the access identifiers is `default` or `null`.

#### Definition: Physical field

A field is considered to be *physical* if it is either

- a variable (4.1)
- a property (4.2) with the read-access or write-access identifier being `default` or `null`
- a property (4.2) with `:isVar` metadata (6.9)

If this is not the case, access to the field from within an accessor method causes a compilation error:

```

1 class Main {
2   // This field cannot be accessed because it is not a real variable
3   public var x(get, set):Int;
4
5   function get_x() {
6     return x;
7   }
8
9   function set_x(x) {
10    return this.x = x;

```



```

11     }
12
13     static public function main() {}
14 }

```

If a physical field is indeed intended, it can be forced by attributing the field in question with the `:isVar` metadata (6.9):

```

1 class Main {
2     // @isVar forces the field to be physical allowing the program to
3     // compile.
4     @:isVar public var x(get, set):Int;
5
6     function get_x() {
7         return x;
8     }
9
10    function set_x(x) {
11        return this.x = x;
12    }
13
14    static public function main() {}
15 }

```

#### Trivia: Property setter type

It is not uncommon for new Haxe users to be surprised by the type of a setter being required to be `T->T` instead of the seemingly more natural `T->Void`. After all, why would a *setter* have to return something?

The rationale is that we still want to be able to use field assignments using setters as right-side expressions. Given a chain like `x = y = 1`, it is evaluated as `x = (y = 1)`. In order to assign the result of `y = 1` to `x`, the former must have a value. If `y` had a setter returning `Void`, this would not be possible.

## 4.3 Method

While variables (4.1) hold data, methods are defining behavior of a program by hosting expressions (5). We have seen method fields in every code example of this document with even the initial Hello World (1.3) example containing a `main` method:

```

1 class Main {
2     static public function main():Void {
3         trace("Hello World");
4     }
5 }

```

Methods are identified by the `function` keyword. We can also learn that they

1. have a name (here: `main`),
2. have an argument list (here: empty `()`),
3. have a return type (here: `Void`),

4. may have access modifiers (4.4) (here: `static` and `public`) and
5. may have an expression (here: `{trace("Hello World");}`).

We can also look at the next example to learn more about arguments and return types:

```
1 class Main {
2     static public function main() {
3         myFunc("foo", 1);
4     }
5
6     static function myFunc(f:String, i) {
7         return true;
8     }
9 }
```

Arguments are given by an opening parenthesis ( after the field name, a comma , separated list of argument specifications and a closing parenthesis ). Additional information on the argument specification is described in [Function Type \(Section 2.6\)](#).

The example demonstrates how type inference (3.6) can be used for both argument and return types. The method `myFunc` has two arguments but only explicitly gives the type of the first one, `f`, as `String`. The second one, `i`, is not type-hinted and it is left to the compiler to infer its type from calls made to it. Likewise, the return type of the method is inferred from the `return true` expression as `Bool`.

### 4.3.1 Overriding Methods

Overriding fields is instrumental for creating class hierarchies. Many design patterns utilize it, but here we will explore only the basic functionality. In order to use overrides in a class, it is required that this class has a parent class (2.3.2). Let us consider the following example:

```
1 class Base {
2     public function new() { }
3     public function myMethod() {
4         return "Base";
5     }
6 }
7
8 class Child extends Base {
9     public override function myMethod() {
10         return "Child";
11     }
12 }
13
14 class Main {
15     static public function main() {
16         var child:Base = new Child();
17         trace(child.myMethod()); // Child
18     }
19 }
```

The important components here are:

- the class `Base` which has a method `myMethod` and a constructor,

- the class `Child` which extends `Base` and also has a method `myMethod` being declared with `override`, and
- the `Main` class whose `main` method creates an instance of `Child`, assigns it to a variable `child` of explicit type `Base` and calls `myMethod()` on it.

The variable `child` is explicitly typed as `Base` to highlight an important difference: At compile-time the type is known to be `Base`, but the runtime still finds the correct method `myMethod` on class `Child`. This is because field access is resolved dynamically at runtime.

The `Child` class can access methods it has overridden by calling `super.methodName()`:

```

1 class Base {
2     public function new() { }
3     public function myMethod() {
4         return "Base";
5     }
6 }
7
8 class Child extends Base {
9     public override function myMethod() {
10         return "Child";
11     }
12
13     public function callHome() {
14         return super.myMethod();
15     }
16 }
17
18
19 class Main {
20     static public function main() {
21         var child = new Child();
22         trace(child.callHome()); // Base
23     }
24 }

```

The section on [Inheritance](#) (Section 2.3.2) explains the use of `super()` from within a new constructor.

### 4.3.2 Effects of variance and access modifiers

Overriding adheres to the rules of variance (3.4). That is, their argument types allow *contravariance* (less specific types) while their return type allows *covariance* (more specific types):

```

1 class Base {
2     public function new() { }
3 }
4
5 class Child extends Base {
6     private function method(obj:Child):Child {
7         return obj;
8     }
9 }
10

```

```

11 class ChildChild extends Child {
12     public override function method(obj:Base):ChildChild {
13         return null;
14     }
15 }
16
17 class Main {
18     static public function main() { }
19 }

```

Intuitively, this follows from the fact that arguments are “written to” the function and the return value is “read from” it.

The example also demonstrates how visibility (4.4.1) may be changed: An overriding field may be *public* if the overridden field is *private*, but not the other way around.

It is not possible to override fields which are declared as *inline* (4.4.2). This is due to the conflicting concepts: While inlining is done at compile-time by replacing a call with the function body, overriding fields necessarily have to be resolved at runtime.

## 4.4 Access Modifier

### 4.4.1 Visibility

Fields are by default *private*, meaning that only the class and its sub-classes may access them. They can be made *public* by using the public access modifier, allowing access from anywhere.

```

1 class MyClass {
2     static public function available() {
3         unavailable();
4     }
5     static private function unavailable() { }
6 }
7
8 class Main {
9     static public function main() {
10         MyClass.available();
11         // Cannot access private field unavailable
12         MyClass.unavailable();
13     }
14 }

```

Access to field `available` of class `MyClass` is allowed from within `Main` because it is denoted as being *public*. However, while access to field `unavailable` is allowed from within class `MyClass`, it is not allowed from within class `Main` because it is *private* (explicitly, although this identifier is redundant here).

The example demonstrates visibility through *static* fields, but the rules for member fields are equivalent. The following example demonstrates visibility behavior for when inheritance (2.3.2) is involved.

```

1 class Base {
2     public function new() { }
3     private function baseField() { }
4 }
5

```

```

6 class Child1 extends Base {
7   private function child1Field() { }
8 }
9
10 class Child2 extends Base {
11   public function child2Field() {
12     var child1 = new Child1();
13     child1.baseField();
14     // Cannot access private field child1Field
15     child1.child1Field();
16   }
17 }
18
19 class Main {
20   static public function main() { }
21 }

```

We can see that access to `child1.baseField()` is allowed from within `Child2` even though `child1` is of a different type, `Child1`. This is because the field is defined on their common ancestor class `Base`, contrary to field `child1Field` which can not be accessed from within `Child2`.

Omitting the visibility modifier usually defaults the visibility to `private`, but there are exceptions where it becomes `public` instead:

1. If the class is declared as `extern`.
2. If the field is declared on an interface (2.3.3).
3. If the field overrides (4.3.1) a public field.

#### Trivia: Protected

Haxe has no notion of a `protected` keyword known from Java, C++ and other object-oriented languages. However, its `private` behavior is equal to those language's `protected` behavior, so Haxe actually lacks their real `private` behavior.

## 4.4.2 Inline

The `inline` keyword allows function bodies to be directly inserted in place of calls to them. This can be a powerful optimization tool, but should be used judiciously as not all functions are good candidates for inline behavior. The following example demonstrates the basic usage:

```

1 class Main {
2   static inline function mid(s1:Int, s2:Int) {
3     return (s1 + s2) / 2;
4   }
5
6   static public function main() {
7     var a = 1;
8     var b = 2;
9     var c = mid(a, b);
10  }
11 }

```

The generated JavaScript output reveals the effect of `inline`:

```

1 (function () { "use strict";
2 var Main = function() { }
3 Main.main = function() {
4     var a = 1;
5     var b = 2;
6     var c = (a + b) / 2;
7 }
8 Main.main();
9 }) ();

```

As evident, the function body  $(s1 + s2) / 2$  of field `mid` was generated in place of the call to `mid(a, b)`, with `s1` being replaced by `a` and `s2` being replaced by `b`. This avoids a function call which, depending on the target and frequency of occurrences, may yield noticeable performance improvements.

It is not always easy to judge if a function qualifies for being inline. Short functions that have no writing expressions (such as `a = assignment`) are usually a good choice, but even more complex functions can be candidates. However, in some cases inlining can actually be detrimental to performance, e.g. because the compiler has to create temporary variables for complex expressions.

Inline is not guaranteed to be done. The compiler might cancel inlining for various reasons or a user could supply the `--no-inline` command line argument to disable inlining. The only exception is if the class is `extern` (6.2) or if the class field has the `:extern` metadata (6.9), in which case inline is forced. If it cannot be done, the compiler emits an error.

It is important to remember this when relying on inline:

```

1 class Main {
2     public static function main () { }
3
4     static function test() {
5         if (Math.random() > 0.5) {
6             return "ok";
7         } else {
8             error("random failed");
9         }
10    }
11
12    @:extern static inline function error(s:String) {
13        throw s;
14    }
15 }

```

If the call to `error` is inlined the program compiles correctly because the control flow checker is satisfied due to the inlined `throw` (5.22) expression. If inline is not done, the compiler only sees a function call to `error` and emits the error `A return is missing here`.

### 4.4.3 Dynamic

Methods can be denoted with the `dynamic` keyword to make them (re-)bindable:

```

1 class Main {
2     static dynamic function test() {
3         return "original";
4     }

```

```

5
6 static public function main() {
7     trace(test()); // original
8     test = function() { return "new"; }
9     trace(test()); // new
10 }
11 }

```

The first call to `test()` invokes the original function which returns the `String` "original". In the next line, `test` is *assigned* a new function. This is precisely what `dynamic` allows: Function fields can be assigned a new function. As a result, the next invocation of `test()` returns the `String` "new".

Dynamic fields cannot be `inline` for obvious reasons: While inlining is done at compile-time, dynamic functions necessarily have to be resolved at runtime.

#### 4.4.4 Override

The access modifier `override` is required when a field is declared which also exists on a parent class (2.3.2). Its purpose is to ensure that the author of a class is aware of the override as this may not always be obvious in large class hierarchies. Likewise, having `override` on a field which does not actually override anything (e.g. due to a misspelled field name) triggers an error.

The effects of overriding fields are detailed in [Overriding Methods \(Section 4.3.1\)](#). This modifier is only allowed on method (4.3) fields.

#### 4.4.5 Static

All fields are member fields unless the modifier `static` is used. Static fields are used “on the class” whereas non-static fields are used “on a class instance”:

```

1 class Main {
2     static function main() {
3         Main.staticField; // static read
4         Main.staticField = 2; // static write
5     }
6
7     static var staticField:Int;
8 }

```

Static variable (4.1) and property (4.2) fields can have arbitrary initialization expressions (5).

## Chapter 5

# Expressions

Expressions in Haxe define what a program *does*. Most expressions are found in the body of a method (4.3), where they are combined to express what that method should do. This section explains the different kinds of expressions. Some definitions help here:

**Definition: Name**

A general name may refer to

- a type,
- a local variable,
- a local function or
- a field.

**Definition: Identifier**

Haxe identifiers start with an underscore `_`, a dollar `$`, a lower-case character `a-z` or an upper-case character `A-Z`. After that, any combination and number of `_`, `A-Z`, `a-z` and `0-9` may follow.

Further limitations follow from the usage context, which are checked upon typing:

- Type names must start with an upper-case letter `A-Z` or an underscore `_`.
- Leading dollars are not allowed for any kind of name (5) (dollar-names are mostly used for macro reification (9.3)).

**Keywords** The following Haxe keywords may not be used as identifiers:

- abstract
- break
- case
- cast



- catch
- class
- continue
- default
- do
- dynamic
- else
- enum
- extends
- extern
- false
- for
- function
- if
- implements
- import
- in
- inline
- interface
- macro
- new
- null
- override
- package
- private
- public
- return
- static
- switch
- this
- throw

- true
- try
- typedef
- untyped
- using
- var
- while

## 5.1 Blocks

A block in Haxe starts with an opening curly brace `{` and ends with a closing curly brace `}`. A block may contain several expressions, each of which is followed by a semicolon `;`. The general syntax is thus:

```
1 {
2     expr1;
3     expr2;
4     ...
5     exprN;
6 }
```

The value and by extension the type of a block-expression is equal to the value and the type of the last sub-expression.

Blocks can contain local variables declared by `var` expression (5.10), as well as local functions declared by `function` expressions (5.11). These are available within the block and within sub-blocks, but not outside the block. Also, they are available only after their declaration. The following example uses `var`, but the same rules apply to `function` usage:

```
1 {
2     a; // error, a is not declared yet
3     var a = 1; // declare a
4     a; // ok, a was declared
5     {
6         a; // ok, a is available in sub-blocks
7     }
8     // ok, a is still available after
9     // sub-blocks
10    a;
11 }
12 a; // error, a is not available outside
```

At runtime, blocks are evaluated from top to bottom. Control flow (e.g. exceptions (5.18) or return expressions (5.19)) may leave a block before all expressions are evaluated.

## 5.2 Constants

The Haxe syntax supports the following constants:

**Int:** An integer (2.1.1), such as 0, 1, 97121, -12, 0xFF0000.

**Float:** A floating point number (2.1.1), such as `0.0`, `1.`, `.3`, `-93.2`.

**String:** A string of characters (10.1), such as `"`, `"foo"`, `'`, `'bar'`.

**true,false:** A boolean (2.1.4) value.

**null:** The null value.

Furthermore, the internal syntax structure treats identifiers (5) as constants, which may be relevant when working with macros (9).

## 5.3 Binary Operators

## 5.4 Unary Operators

## 5.5 Array Declaration

Arrays are initialized by enclosing comma , separated values in brackets []. A plain [] represents the empty array, whereas [1, 2, 3] initializes an array with three elements 1, 2 and 3.

The generated code may be less concise on platforms that do not support array initialization. Essentially, such initialization code then looks like this:

```
1 var a = new Array();
2 a.push(1);
3 a.push(2);
4 a.push(3);
```

This should be considered when deciding if a function should be inlined (4.4.2) as it may inline more code than visible in the syntax.

Advanced initialization techniques are described in [Array Comprehension](#) (Section 6.6).

## 5.6 Object Declaration

Object declaration begins with an opening curly brace { after which key:value-pairs separated by comma , follow, and which ends in a closing curly brace }.

```
1 {
2   key1:value1,
3   key2:value2,
4   ...
5   keyN:valueN
6 }
```

Further details of object declaration are described in the section about anonymous structures (2.5).

## 5.7 Field Access

Field access is expressed by using the dot . followed by the name of the field.

```
1 object.fieldName
```

This syntax is also used to access types within packages in the form of `pack.Type`.

The typer ensures that an accessed field actually exist and may apply transformations depending on the nature of the field. If a field access is ambiguous, understanding the resolution order (3.7.3) may help.

## 5.8 Array Access

Array access is expressed by using an opening bracket `[` followed by the index expression and a closing bracket `]`.

```
1 expr[indexExpr]
```

This notation is allowed with arbitrary expressions, but at typing level only certain combinations are admitted:

- `expr` is of `Array` or `Dynamic` and `indexExpr` is of `Int`
- `expr` is an abstract type (2.8) which defines a matching array access (2.8.3)

## 5.9 Function Call

Functions calls consist of an arbitrary subject expression followed by an opening parenthesis `(`, a comma `,` separated list of expressions as arguments and a closing parenthesis `)`.

```
1 subject(); // call with no arguments
2 subject(e1); // call with one argument
3 subject(e1, e2); // call with two arguments
4 // call with multiple arguments
5 subject(e1, ..., eN);
```

### 5.10 var

The `var` keyword allows declaring multiple variables, separated by comma `,`. Each variable has a valid identifier (5) and optionally a value assignment following the assignment operator `=`. Variables can also have an explicit type-hint.

```
1 var a; // declare local a
2 var b:Int; // declare variable b of type Int
3 // declare variable c, initialized to value 1
4 var c = 1;
5 // declare variable d and variable e
6 // initialized to value 2
7 var d,e = 2;
```

The scoping behavior of local variables is described in [Blocks](#) (Section 5.1).

### 5.11 Local functions

Haxe supports first-class functions and allows declaring local functions in expressions. The syntax follows class field methods (4.3):

```

1 class Main {
2     static public function main() {
3         var value = 1;
4         function myLocalFunction(i) {
5             return value + i;
6         }
7         trace(myLocalFunction(2)); // 3
8     }
9 }

```

We declare `myLocalFunction` inside the block expression (5.1) of the main class field. It takes one argument `i` and adds it to `value`, which is defined in the outside scope.

The scoping is equivalent to that of variables (5.10) and for the most part writing a named local function can be considered equal to assigning an unnamed local function to a local variable:

```

1 var myLocalFunction = function(a) { }

```

However, there are some differences related to type parameters and the position of the function. We speak of a “lvalue” function if it is not assigned to anything upon its declaration, and an “rvalue” function otherwise.

- Lvalue functions require a name and can have type parameters (3.2).
- Rvalue functions may have a name, but cannot have type parameters.

## 5.12 new

The `new` keyword signals that a class (2.3) or an abstract (2.8) is being instantiated. It is followed by the type path (3.7) of the type which is to be instantiated. It may also list explicit type parameters (3.2) enclosed in `<>` and separated by comma `,`. After an opening parenthesis `(` follow the constructor arguments, again separated by comma `,`, with a closing parenthesis `)` at the end.

```

1 class Main<T> {
2     static public function main() {
3         new Main<Int>(12, "foo");
4     }
5
6     function new(t:T, s:String) { }
7 }

```

Within the `main` method we instantiate an instance of `Main` itself, with an explicit type parameter `Int` and the arguments `12` and `"foo"`. As we can see, the syntax is very similar to the function call syntax (5.9) and it is common to speak of “constructor calls”.

## 5.13 for

Haxe does not support traditional for-loops known from C. Its `for` keyword expects an opening parenthesis `(`, then a variable identifier followed by the keyword `in` and an arbitrary expression used as iterating collection. After the closing parenthesis `)` follows an arbitrary loop body expression.

```

1 for (v in e1) e2;

```

The typer ensures that the type of `e1` can be iterated over, which is typically the case if it has an `iterator` method returning an `Iterator<T>`, or if it is an `Iterator<T>` itself.

Variable `v` is then available within loop body `e2` and holds the value of the individual elements of collection `e1`.

Haxe has a special range operator to iterate over intervals. It is a binary operator taking two `Int` operands: `min...max` returns an `IntIterator` instance that iterates from `min` (inclusive) to `max` (exclusive). Note that `max` may not be smaller than `min`.

```
1 for (i in 0...10) trace(i); // 0 to 9
```

The type of a `for` expression is always `Void`, meaning it has no value and cannot be used as right-side expression.

The control flow of loops can be affected by `break` (5.20) and `continue` (5.21) expressions.

## 5.14 while

A normal `while` loop starts with the `while` keyword, followed by an opening parenthesis `(`, the condition expression and a closing parenthesis `)`. After that follows the loop body expression:

```
1 while(condition) expression;
```

The condition expression has to be of type `Bool`.

Upon each iteration, the condition expression is evaluated. If it evaluates to `false`, the loop stops, otherwise it evaluates the loop body expression.

```
1 class Main {
2     static public function main() {
3         var f = 0.0;
4         while (f < 0.5) {
5             trace(f);
6             f = Math.random();
7         }
8     }
9 }
```

This kind of `while`-loop is not guaranteed to evaluate the loop body expression at all: If the condition does not hold from the start, it is never evaluated. This is different for `do-while` loops (5.15).

## 5.15 do-while

A `do-while` loop starts with the `do` keyword followed by the loop body expression. After that follows the `while` keyword, an opening parenthesis `(`, the condition expression and a closing parenthesis `)`:

```
1 do expression while(condition);
```

The condition expression has to be of type `Bool`.

As the syntax suggests, the loop body expression is always evaluated at least once, unlike `while` (5.14) loops.

## 5.16 if

Conditional expressions come in the form of a leading `if` keyword, a condition expression enclosed in parentheses `()` and a expression to be evaluated in case the condition holds:

```
1 if (condition) expression;
```

The condition expression has to be of type `Bool`.

Optionally, `expression` may be followed by the `else` keyword as well as another expression to be evaluated if the condition does not hold:

```
1 if (condition) expression1 else expression2;
```

Here, `expression2` may consist of another `if` expression:

```
1 if (condition1) expression1
2 else if (condition2) expression2
3 else expression3
```

If the value of an `if` expression is required, e.g. for `var x = if (condition) expression1 else expression2`, the typer ensures that the types of `expression1` and `expression2` unify (3.5). If no `else` expression is given, the type is inferred to be `Void`.

## 5.17 switch

A basic `switch` expression starts with the `switch` keyword and the `switch` subject expression, as well as the case expressions between curly braces `{}`. Case expressions either start with the `case` keyword and are followed by a pattern expression, or consist of the `default` keyword. In both cases a colon `:` and an optional case body expression follows:

```
1 switch subject {
2   case pattern1: case-body-expression-1;
3   case pattern2: case-body-expression-2;
4   default: default-expression;
5 }
```

Case body expressions never “fall through”, so the `break` (5.20) keyword is not supported in Haxe.

Switch expressions can be used as value; in that case the types of all case body expressions and the default expression must unify (3.5).

Further details on syntax of pattern expressions are detailed in [Pattern Matching](#) (Section 6.4).

## 5.18 try/catch

Haxe allows catching values using its `try/catch` syntax:

```
1 try try-expr
2 catch (varName1:Type1) catch-expr-1
3 catch (varName2:Type2) catch-expr-2
```

If during runtime the evaluation of `try-expression` causes a `throw` (5.22), it can be caught by any subsequent `catch` block. These blocks consist of

- a variable name which holds the thrown value,
- an explicit type annotation which determines which types of values to catch, and

- the expression to execute in that case.

Haxe allows throwing and catching any kind of value, it is not limited to types inheriting from a specific exception or error class. Catch blocks are checked from top to bottom with the first one whose type is compatible with the thrown value being picked.

This process has many similarities to the compile-time unification (3.5) behavior. However, since the check has to be done at runtime there are several restrictions:

- The type must exist at runtime: Class instances (2.3), enum instances (2.4), abstract core types (2.8.7) and Dynamic (2.7).
- Type parameters can only be Dynamic (2.7).

## 5.19 return

A `return` expression can come with or without an value expression:

```
1 return;
2 return expression;
```

It leaves the control-flow of the innermost function it is declared in, which has to be distinguished when local functions (5.11) are involved:

```
1 function f1() {
2     function f2() {
3         return;
4     }
5     f2();
6     expression;
7 }
```

The `return` leaves local function `f2`, but not `f1`, meaning `expression` is still evaluated.

If `return` is used without a value expression, the typer ensures that the return type of the function it returns from is of `Void`. If it has a value expression, the typer unifies (3.5) its type with the return type (explicitly given or inferred by previous `return` expressions) of the function it returns from.

## 5.20 break

The `break` keyword leaves the control flow of the innermost loop (`for` or `while`) it is declared in, stopping further iterations:

```
1 while(true) {
2     expression1;
3     if (condition) break;
4     expression2;
5 }
```

Here, `expression1` is evaluated for each iteration, but as soon as `condition` holds, `expression2` is not evaluated anymore.

The typer ensures that it appears only within a loop. The `break` keyword in `switch` cases (5.17) is not supported in Haxe.



## 5.21 continue

The `continue` keyword ends the current iteration of the innermost loop (`for` or `while`) it is declared in, causing the loop condition to be checked for the next iteration:

```
1 while(true) {  
2     expression1;  
3     if(condition) continue;  
4     expression2;  
5 }
```

Here, `expression1` is evaluated for each iteration, but if `condition` holds, `expression2` is not evaluated for the current iteration. Unlike `break`, iterations continue.

The typer ensures that it appears only within a loop.

## 5.22 throw

Haxe allows throwing any kind of value using its `throw` syntax:

```
1 throw expr
```

A value which is thrown like this can be caught by `catch` blocks (5.18). If no such block catches it, the behavior is target-dependent.

## 5.23 cast

Haxe allows two kinds of casts:

```
1 cast expr; // unsafe cast  
2 cast (expr, Type); // safe cast
```

### 5.23.1 unsafe cast

Unsafe casts are useful to subvert the type system. The compiler types `expr` as usual and then wraps it in a monomorph (2.9). This allows the expression to be assigned to anything.

Unsafe casts do not introduce any dynamic (2.7) types, as the following example shows:

```
1 class Main {  
2     public static function main() {  
3         var i = 1;  
4         $type(i); // Int  
5         var s = cast i;  
6         $type(s); // Unknown<0>  
7         Std.parseInt(s);  
8         $type(s); // String  
9     }  
10 }
```

Variable `i` is typed as `Int` and then assigned to variable `s` using the unsafe cast `cast i`. This causes `s` to be of an unknown type, a monomorph. Following the usual rules of unification (3.5), it can then be bound to any type, such as `String` in this example.

These casts are called “unsafe” because the runtime behavior for invalid casts is not defined. While most dynamic targets (2.2) are likely to work, it might lead to undefined errors on static targets (2.2).

Unsafe casts have little to no runtime overhead.

### 5.23.2 safe cast

Unlike unsafe casts (5.23.1), the runtime behavior in case of a failing cast is defined for safe casts:

```
1 class Base {
2   public function new() { }
3 }
4
5 class Child1 extends Base {}
6
7 class Child2 extends Base {}
8
9 class Main {
10  public static function main() {
11    var child1:Base = new Child1();
12    var child2:Base = new Child2();
13    cast(child1, Base);
14    // Exception: Class cast error
15    cast(child1, Child2);
16  }
17 }
```

In this example we first cast a class instance of type `Child1` to `Base`, which succeeds because `Child1` is a child class (2.3.2) of `Base`. We then try to cast the same class instance to `Child2`, which is not allowed because instances of `Child2` are not instances of `Child1`.

The Haxe compiler guarantees that an exception of type `String` is thrown (5.22) in this case. This exception can be caught using a `try/catch` block (5.18).

Safe casts have a runtime overhead. It is important to understand that the compiler already generates type checks, so it is redundant to add manual checks, e.g. using `Std.is`. The intended usage is to try the safe cast and catch the `String` exception.

## 5.24 type check

Since Haxe 3.1.0

It is possible to employ compile-time type checks using the following syntax:

```
1 (expr : type)
```

The parentheses are mandatory. Unlike safe casts (5.23.2) this construct has no run-time impact. It has two compile-time implications:

1. Top-down inference (3.6.1) is used to type `expr` with type `type`.
2. The resulting typed expression is unified (3.5) with type `type`.

This has the usual effect of both operations such as the given type being used as expected type when performing unqualified identifier resolution (3.7.3) and the unification checking for abstract casts (2.8.1).

## Chapter 6

# Language Features

### *Abstract types (2.8):*

An abstract type is a compile-time construct which is represented in a different way at run-time. This allows giving a whole new meaning to existing types.

### *Extern classes (6.2):*

Externs can be used to describe target-specific interaction in a type-safe manner.

### *Anonymous structures (2.5):*

Data can easily be grouped in anonymous structures, minimizing the necessity of small data classes.

```
1 var point = { x: 0, y: 10 };
2 point.x += 10;
```

### *Array Comprehension (6.6):*

Create and populate arrays quickly using for loops and logic.

```
1 var evenNumbers = [ for (i in 0...100) if (i%2==0) i ];
```

### *Classes, interfaces and inheritance (2.3):*

Haxe allows structuring code in classes, making it an object-oriented language. Common related features known from languages such as Java are supported, including inheritance and interfaces.

### *Conditional compilation (6.1):*

Conditional Compilation allows compiling specific code depending on compilation parameters. This is instrumental for abstracting target-specific differences, but can also be used for other purposes, such as more detailed debugging.

```
1 \#if js
2     js.Browser.alert("Hello");
3 \#elseif sys
4     Sys.println("Hello");
5 \#end
```

### *(Generalized) Algebraic Data Types (2.4):*

Structure can be expressed through algebraic data types (ADT), which are known as enums in the Haxe Language. Furthermore, Haxe supports their generalized variant known as GADT.

```
1 enum Result {
2     Success(data:Array<Int>);
3     UserError(msg:String);
4     SystemError(msg:String, position:PosInfos);
5 }
```

#### *Inline calls (4.4.2):*

Functions can be designated as being inline, allowing their code to be inserted at call-site. This can yield significant performance benefits without resorting to code duplication via manual inlining.

#### *Iterators (6.7):*

Iterating over a set of values, e.g. the elements of an array, is very easy in Haxe courtesy of iterators. Custom classes can quickly implement iterator functionality to allow iteration.

```
1 for (i in [1, 2, 3]) {  
2     trace(i);  
3 }
```

#### *Local functions and closures (5.11):*

Functions in Haxe are not limited to class fields and can be declared in expressions as well, allowing powerful closures.

```
1 var buffer = "";  
2 function append(s:String) {  
3     buffer += s;  
4 }  
5 append("foo");  
6 append("bar");  
7 trace(buffer); // foobar
```

#### *Metadata (6.9):*

Add metadata to fields, classes or expressions. This can communicate information to the compiler, macros, or runtime classes.

```
1 class MyClass {  
2     @range(1, 8) var value:Int;  
3 }  
4 trace(haxe.rtti.Meta.getFields(MyClass).value.range); // [1,8]
```

#### *Static Extensions (6.3):*

Existing classes and other types can be augmented with additional functionality through using static extensions.

```
1 using StringTools;  
2 " Me & You ".trim().htmlEscape();
```

#### *String Interpolation (6.5):*

Strings declared with a single quotes are able to access variables in the current context.

```
1 trace('My name is $name and I work in ${job.industry}');
```

#### *Partial function application (6.8):*

Any function can be applied partially, providing the values of some arguments and leaving the rest to be filled in later.

```
1 var map = new haxe.ds.IntMap();  
2 var setToTwelve = map.set.bind(_, 12);  
3 setToTwelve(1);  
4 setToTwelve(2);
```

#### *Pattern Matching (6.4):*

Complex structures can be matched against patterns, extracting information from an enum or a structure and defining specific operations for specific value combination.

```

1 var a = { foo: 12 };
2 switch (a) {
3     case { foo: i }: trace(i);
4     default:
5 }

```

*Properties (4.2):*

Variable class fields can be designed as properties with custom read and write access, allowing fine grained access control.

```

1 public var color(get, set);
2 function get_color() {
3     return element.style.backgroundColor;
4 }
5 function set_color(c:String) {
6     trace('Setting background of element to $c');
7     return element.style.backgroundColor = c;
8 }

```

*Access control (6.10):*

The access control language feature uses the Haxe metadata syntax to force or allow access classes or fields.

*Type Parameters, Constraints and Variance (3.2):*

Types can be parametrized with type parameters, allowing typed containers and other complex data structures. Type parameters can also be constrained to certain types and respect variance rules.

```

1 class Main<A> {
2     static function main() {
3         new Main<String>("foo");
4         new Main(12); // use type inference
5     }
6
7     function new(a:A) { }
8 }

```

## 6.1 Conditional Compilation

Haxe allows conditional compilation by using `#if`, `#elseif` and `#else` and checking for *compiler flags*.

### Definition: Compiler Flag

A compiler flag is a configurable value which may influence the compilation process. Such a flag can be set by invoking the command line with `-D key=value` or just `-D key`, in which case the value defaults to "1". The compiler also sets several flags internally to pass information between different compilation steps.

This example demonstrates usage of conditional compilation:

```

1 class Main {
2     public static function main() {
3         #if !debug

```

```

4     trace("ok");
5     #elseif (debug_level > 3)
6         trace(3);
7     #else
8         trace("debug level too low");
9     #end
10 }
11 }

```

Compiling this without any flags will leave only the `trace("ok");` line in the body of the `main` method. The other branches are discarded while parsing the file. These other branches must still contain valid Haxe syntax, but the code is not type-checked.

The conditions after `#if` and `#elseif` allow the following expressions:

- Any identifier is replaced by the value of the compiler flag by the same name. Note that `-D some-flag` from command line leads to the flags `some-flag` and `some_flag` to be defined.
- The values of `String`, `Int` and `Float` constants are used directly.
- The boolean operators `&&` (and), `||` (or) and `!` (not) work as expected, however the full expression must be completely contained by parentheses.
- The operators `==`, `!=`, `>`, `>=`, `<`, `<=` can be used to compare values.
- Parentheses `()` can be used to group expressions as usual.

The Haxe parser does not parse `some-flag` as a single token and instead reads it as a subtraction binary operator `some - flag`. In cases like this the underscore version `some_flag` has to be used.

**Built-in Compiler Flags** An exhaustive list of all built-in defines can be obtained by invoking the Haxe Compiler with the `--help-defines` argument. The Haxe Compiler allows multiple `-D` flags per compilation.

See also the Compiler Flags list (7.1).

## 6.2 Externs

Externs can be used to describe target-specific interaction in a type-safe manner. They are defined like normal classes, except that

- the `class` keyword is preceded by the `extern` keyword,
- methods (4.3) have no expressions and
- all argument and return types are explicit.

A common example from the Haxe Standard Library (10) is the `Math` class, as an excerpt shows:

```

1 extern class Math
2 {
3     static var PI(default,null) : Float;
4     static function floor(v:Float):Int;
5 }

```

We see that externs can define both methods and variables (actually, `PI` is declared as a read-only property (4.2)). Once this information is available to the compiler, it allows field access accordingly and also knows the types:

```
1 class Main {
2     static public function main() {
3         var pi = Math.floor(Math.PI);
4         $type(pi); // Int
5     }
6 }
```

This works because the return type of method `floor` is declared to be `Int`.

The Haxe Standard Library comes with many externs for the Flash and JavaScript target. They allow accessing the native APIs in a type-safe manner and are instrumental for designing higher-level APIs. There are also externs for many popular native libraries on haxelib (11).

The Flash, Java and C# targets allow direct inclusion of native libraries from command line (7). Target-specific details are explained in the respective sections of [Target Details \(Chapter 12\)](#).

Some targets such as Python or JavaScript may require generating additional “import” code that loads an `extern` class from a native module. Haxe provides ways to declare such dependencies also described in respective sections [Target Details \(Chapter 12\)](#).

### Rest arguments and type choices Since Haxe 3.2.0

The `haxe.extern` package provides two types that help mapping native semantics to Haxe:

**Rest<T>:** This type can be used as a final function argument to allow passing an arbitrary number of additional call arguments. The type parameter can be used to constrain these arguments to a specific type.

**EitherType<T1, T2>:** This type allows using either of its parameter types, thus representing a type choice. It can be nested to allow more than two different types.

We demonstrate the usage in this code sample:

```
1 import haxe.extern.Rest;
2 import haxe.extern.EitherType;
3
4 extern class MyExtern {
5     static function f1(s:String, r:Rest<Int>):Void;
6     static function f2(e:EitherType<Int, String>):Void;
7 }
8
9 class Main {
10     static function main() {
11         MyExtern.f1("foo", 1, 2, 3); // use 1, 2, 3 as rest argument
12         MyExtern.f1("foo"); // no rest argument
13         //MyExtern.f1("foo", "bar"); // String should be Int
14
15         MyExtern.f2("foo");
16         MyExtern.f2(12);
17         //MyExtern.f2(true); // Bool should be EitherType<Int, String>
18     }
19 }
```

## 6.3 Static Extension

### Definition: Static Extension

A static extension allows pseudo-extending existing types without modifying their source. In Haxe this is achieved by declaring a static method with a first argument of the extending type and then bringing the defining class into context through `using`.

Static extensions can be a powerful tool which allows augmenting types without actually changing them. The following example demonstrates the usage:

```
1 using Main.IntExtender;
2
3 class IntExtender {
4     static public function triple(i:Int) {
5         return i * 3;
6     }
7 }
8
9 class Main {
10     static public function main() {
11         trace(12.triple());
12     }
13 }
```

Clearly, `Int` does not natively provide a `triple` method, yet this program compiles and outputs 36 as expected. This is because the call to `12.triple()` is transformed into `IntExtender.triple(12)`. There are three requirements for this:

1. Both the literal `12` and the first argument of `triple` are known to be of type `Int`.
2. The class `IntExtender` is brought into context through `using Main.IntExtender`.
3. `Int` does not have a `triple` field by itself (if it had, that field would take priority over the static extension).

Static extensions are usually considered syntactic sugar and indeed they are, but it is worth noting that they can have a dramatic effect on code readability: Instead of nested calls in the form of `f1(f2(f3(f4(x))))`, chained calls in the form of `x.f4().f3().f2().f1()` can be used.

Following the rules previously described in [Resolution Order \(Section 3.7.3\)](#), multiple `using` expressions are checked from bottom to top, with the types within each module as well as the fields within each type being checked from top to bottom. Using a module (as opposed to a specific type of a module, see [Modules and Paths \(Section 3.7\)](#)) as static extension brings all its types into context.

### 6.3.1 In the Haxe Standard Library

Several classes in the Haxe Standard Library are suitable for static extension usage. The next example shows the usage of `StringTools`:

```
1 using StringTools;
2
3 class Main {
```



```

4 static public function main() {
5     "adc".replace("d", "b");
6 }
7 }

```

While `String` does not have a `replace` functionality by itself, the `using StringTools` static extension provides one. As usual, the JavaScript output nicely shows the transformation:

```

1 Main.main = function() {
2     StringTools.replace("adc", "d", "b");
3 }

```

The following classes from the Haxe Standard Library are designed to be used as static extensions:

**StringTools:** Provides extended functionality on strings, such as replacing or trimming.

**Lambda:** Provides functional methods on iterables.

**haxe.EnumTools:** Provides type information functionality on enums and their instances.

**haxe.macro.Tools:** Provides different extensions for working with macros (see [Tools \(Section 9.4\)](#)).

#### Trivia: “using” using

Since the `using` keyword was added to the language, it has been common to talk about certain problems with “using using” or the effect of “using using”. This makes for awkward English in many cases, so the author of this manual decided to call the feature by what it actually is: Static extension.

## 6.4 Pattern Matching

### 6.4.1 Introduction

Pattern matching is the process of branching depending on a value matching given, possibly deep patterns. In Haxe, all pattern matching is done within a `switch` expression ([5.17](#)) where the individual `case` expressions represent the patterns. Here we will explore the syntax for different patterns using this data structure as running example:

```

1 enum Tree<T> {
2     Leaf(v:T);
3     Node(l:Tree<T>, r:Tree<T>);
4 }

```

Some pattern matcher basics include:

- Patterns will always be matched from top to bottom.
- The topmost pattern that matches the input value has its expression executed.
- A `_` pattern matches anything, so `case _:` is equal to `default:`

## 6.4.2 Enum matching

Enums can be matched by their constructors in a natural way:

```
1  var myTree = Node(Leaf("foo"), Node(Leaf("bar"), Leaf("foobar")));
2  var match = switch(myTree) {
3    // matches any Leaf
4    case Leaf(_): "0";
5    // matches any Node that has r = Leaf
6    case Node(_, Leaf(_)): "1";
7    // matches any Node that has has
8    // r = another Node, which has
9    // l = Leaf("bar")
10   case Node(_, Node(Leaf("bar"), _)): "2";
11   // matches anything
12   case _: "3";
13 }
14 trace(match); // 2
```

The pattern matcher will check each case from top to bottom and pick the first one that matches the input value. The following manual interpretation of each case rule helps understanding the process:

**case Leaf(\_):** matching fails because myTree is a Node

**case Node(\_, Leaf(\_)):** matching fails because the right sub-tree of myTree is not a Leaf, but another Node

**case Node(\_, Node(Leaf("bar"), \_)):** matching succeeds

**case \_:** this is not checked here because the previous line matched

## 6.4.3 Variable capture

It is possible to catch any value of a sub-pattern by matching it against an identifier:

```
1  var myTree = Node(Leaf("foo"), Node(Leaf("bar"), Leaf("foobar")));
2  var name = switch(myTree) {
3    case Leaf(s): s;
4    case Node(Leaf(s), _): s;
5    case _: "none";
6  }
7  trace(name); // foo
```

This would return one of the following:

- If myTree is a Leaf, its name is returned.
- If myTree is a Node whose left sub-tree is a Leaf, its name is returned (this will apply here, returning "foo").
- Otherwise "none" is returned.

It is also possible to use = to capture values which are further matched:

```

1   var node = switch(myTree) {
2       case Node(leafNode = Leaf("foo"), _): leafNode;
3       case x: x;
4   }
5   trace(node); // Leaf(foo)

```

Here, `leafNode` is bound to `Leaf("foo")` if the input matches that. In all other cases, `myTree` itself is returned: case `x` works similar to case `_` in that it matches anything, but with an identifier name like `x` it also binds the matched value to that variable.

#### 6.4.4 Structure matching

It is also possible to match against the fields of anonymous structures and instances:

```

1   var myStructure = {
2       name: "haxe",
3       rating: "awesome"
4   };
5   var value = switch(myStructure) {
6       case { name: "haxe", rating: "poor" }:
7           throw false;
8       case { rating: "awesome", name: n }:
9           n;
10      case _:
11          "no awesome language found";
12  }
13  trace(value); // haxe

```

In the second case we bind the matched `name` field to identifier `n` if `rating` matches `"awesome"`. Of course this structure could also be put into the `Tree` from the previous example to combine structure and enum matching.

A limitation with regards to class instances is that you cannot match against fields of their parent class.

#### 6.4.5 Array matching

Arrays can be matched on fixed length:

```

1   var myArray = [1, 6];
2   var match = switch(myArray) {
3       case [2, _]: "0";
4       case [_, 6]: "1";
5       case []: "2";
6       case [_, _, _]: "3";
7       case _: "4";
8   }
9   trace(match); // 1

```

This will trace `1` because `array[1]` matches `6`, and `array[0]` is allowed to be anything.

#### 6.4.6 Or patterns

The `|` operator can be used anywhere within patterns to describe multiple accepted patterns:

```

1   var match = switch(7) {
2       case 4 | 1: "0";
3       case 6 | 7: "1";
4       case _: "2";
5   }
6   trace(match); // 1

```

If there is a captured variable in an or-pattern, it must appear in both its sub-patterns.

### 6.4.7 Guards

It is also possible to further restrict patterns with the `case ... if(condition) :` syntax:

```

1   var myArray = [7, 6];
2   var s = switch(myArray) {
3       case [a, b] if (b > a):
4       b + ">" + a;
5       case [a, b]:
6       b + "<=" + a;
7       case _: "found something else";
8   }
9   trace(s); // 6<=7

```

The first case has an additional guard condition `if (b > a)`. It will only be selected if that condition holds, otherwise matching continues with the next case.

### 6.4.8 Match on multiple values

Array syntax can be used to match on multiple values:

```

1   var s = switch [1, false, "foo"] {
2       case [1, false, "bar"]: "0";
3       case [_, true, _]: "1";
4       case [_, false, _]: "2";
5   }
6   trace(s); // 2

```

This is quite similar to usual array matching, but there are differences:

- The number of elements is fixed, so patterns of different array length will not be accepted.
- It is not possible to capture the switch value in a variable, i.e. `case x` is not allowed (`case _` still is).

### 6.4.9 Extractors

Since Haxe 3.1.0

Extractors allow applying transformations to values being matched. This is often useful when a small operation is required on a matched value before matching can continue:

```

1   enum Test {
2       TString(s:String);
3       TInt(i:Int);
4   }

```

```

5
6 class Main {
7     static public function main() {
8         var e = TString("fOo");
9         switch(e) {
10             case TString(temp):
11                 switch(temp.toLowerCase()) {
12                     case "foo": true;
13                     case _: false;
14                 }
15             case _: false;
16         }
17     }
18 }

```

Here we have to capture the argument value of the `TString` enum constructor in a variable `temp` and use a nested switch on `temp.toLowerCase()`. Obviously, we want matching to succeed if `TString` holds a value of "foo" regardless of its casing. This can be simplified with extractors:

```

1 enum Test {
2     TString(s:String);
3     TInt(i:Int);
4 }
5
6 class Main {
7     static public function main() {
8         var e = TString("fOo");
9         var success = switch(e) {
10             case TString(_.toLowerCase() => "foo"):
11                 true;
12             case _:
13                 false;
14         }
15     }
16 }

```

Extractors are identified by the `extractorExpression => match expression`. The compiler generates code which is similar to the previous example, but the original syntax was greatly simplified. Extractors consist of two parts, which are separated by the `=>` operator:

1. The left side can be any expression, where all occurrences of underscore `_` are replaced with the currently matched value.
2. The right side is a pattern which is matched against the result of the evaluation of the left side.

Since the right side is a pattern, it can contain another extractor. The following example "chains" two extractors:

```

1 class Main {
2     static public function main() {
3         switch(3) {
4             case add(_, 1) => mul(_, 3) => a:

```

```

5     trace(a);
6   }
7 }
8
9 static function add(i1:Int, i2:Int) {
10   return i1 + i2;
11 }
12
13 static function mul(i1:Int, i2:Int) {
14   return i1 * i2;
15 }
16 }

```

This traces 12 as a result of the calls to `add(3, 1)`, where 3 is the matched value, and `mul(4, 3)` where 4 is the result of the `add` call. It is worth noting that the `a` on the right side of the second `=>` operator is a capture variable (6.4.3).

It is currently not possible to use extractors within or-patterns (6.4.6):

```

1 class Main {
2   static public function main() {
3     switch("foo") {
4       // Extractors in or patterns are not allowed
5       case (_.toLowerCase() => "foo") | "bar":
6     }
7   }
8 }

```

However, it is possible to have or-patterns on the right side of an extractor, so the previous example would compile without the parentheses.

### 6.4.10 Exhaustiveness checks

The compiler ensures that no possible cases are forgotten:

```

1 switch(true) {
2   case false:
3 } // Unmatched patterns: true

```

The matched type `Bool` admits two values `true` and `false`, but only `false` is checked.

Figure out wtf our rules are now for when this is checked.

### 6.4.11 Useless pattern checks

In a similar fashion, the compiler detects patterns which will never match the input value:

```

1 switch(Leaf("foo")) {
2   case Leaf(_)
3     | Leaf("foo"): // This pattern is unused
4   case Node(l,r):
5   case _: // This pattern is unused
6 }

```

## 6.5 String Interpolation

With Haxe 3 it is no longer necessary to manually concatenate parts of a string due to the introduction of *String Interpolation*. Special identifiers, denoted by the dollar sign \$ within a String enclosed by single-quote ' characters, are evaluated as if they were concatenated identifiers:

```
1 var x = 12;
2 // The value of x is 12
3 trace('The value of x is $x');
```

Furthermore, it is possible to include whole expressions in the string by using `${expr}`, with `expr` being any valid Haxe expression:

```
1 var x = 12;
2 // The sum of 12 and 3 is 15
3 trace('The sum of $x and 3 is ${x + 3}');
```

String interpolation is a compile-time feature and has no impact on the runtime. The above example is equivalent to manual concatenation, which is exactly what the compiler generates:

```
1 trace("The sum of " + x +
2 " and 3 is " + (x + 3));
```

Of course the use of single-quote enclosed strings without any interpolation remains valid, but care has to be taken regarding the \$ character as it triggers interpolation. If an actual dollar-sign should be used in the string, \$\$ can be used.

### Trivia: String Interpolation before Haxe 3

String Interpolation has been a Haxe feature since version 2.09. Back then, the macro `Std.format` had to be used, being both slower and less comfortable than the new string interpolation syntax.

## 6.6 Array Comprehension

Comprehensions are only listing Arrays, not Maps

Array comprehension in Haxe uses existing syntax to allow concise initialization of arrays. It is identified by `for` or `while` constructs:

```
1 class Main {
2     static public function main() {
3         var a = [for (i in 0...10) i];
4         trace(a); // [0,1,2,3,4,5,6,7,8,9]
5
6         var i = 0;
7         var b = [while(i < 10) i++];
8         trace(b); // [0,1,2,3,4,5,6,7,8,9]
9     }
10 }
```

Variable `a` is initialized to an array holding the numbers 0 to 9. The compiler generates code which adds the value of each loop iteration to the array, so the following code would be equivalent:

```
1 var a = [];
2 for (i in 0...10) a.push(i);
```

Variable `b` is initialized to an array with the same values, but through a different comprehension style using `while` instead of `for`. Again, the following code would be equivalent:

```
1 var i = 0;
2 var a = [];
3 while(i < 10) a.push(i++);
```

The loop expression can be anything, including conditions and nested loops, so the following works as expected:

```
1 class Main {
2   static public function main() {
3     var a = [
4       for (a in 1...11)
5         for(b in 2...4)
6           if (a % b == 0)
7             a+ "/" +b
8     ];
9     // [2/2,3/3,4/2,6/2,6/3,8/2,9/3,10/2]
10    trace(a);
11  }
12 }
```

## 6.7 Iterators

With Haxe it is very easy to define custom iterators and iterable data types. These concepts are represented by the types `Iterator<T>` and `Iterable<T>` respectively:

```
1 typedef Iterator<T> = {
2   function hasNext() : Bool;
3   function next() : T;
4 }
5
6 typedef Iterable<T> = {
7   function iterator() : Iterator<T>;
8 }
```

Any class (2.3) which structurally unifies (3.5.2) with one of these types can be iterated over using a `for`-loop (5.13). That is, if the class defines methods `hasNext` and `next` with matching return types it is considered an iterator, if it defines a method `iterator` returning an `Iterator<T>` it is considered an iterable type.

```
1 class MyStringIterator {
2   var s:String;
3   var i:Int;
4
5   public function new(s:String) {
6     this.s = s;
7     i = 0;
8   }
9
10  public function hasNext() {
11    return i < s.length;
```



```

12     }
13
14     public function next() {
15         return s.charAt(i++);
16     }
17 }
18
19 class Main {
20     static public function main() {
21         var myIt = new MyStringIterator("string");
22         for (chr in myIt) {
23             trace(chr);
24         }
25     }
26 }

```

The type `MyStringIterator` in this example qualifies as iterator: It defines a method `hasNext` returning `Bool` and a method `next` returning `String`, making it compatible with `Iterator<String>`. The main method instantiates it, then iterates over it.

```

1 class MyArrayWrap<T> {
2     var a:Array<T>;
3     public function new(a:Array<T>) {
4         this.a = a;
5     }
6
7     public function iterator() {
8         return a.iterator();
9     }
10 }
11
12 class Main {
13     static public function main() {
14         var myWrap = new MyArrayWrap([1, 2, 3]);
15         for (elt in myWrap) {
16             trace(elt);
17         }
18     }
19 }

```

Here we do not setup a full iterator like in the previous example, but instead define that the `MyArrayWrap<T>` has a method `iterator`, effectively forwarding the iterator method of the wrapped `Array<T>` type.

## 6.8 Function Bindings

Haxe 3 allows binding functions with partially applied arguments. Each function type can be considered to have a `bind` field, which can be called with the desired number of arguments in order to create a new function. This is demonstrated here:

```

1 class Main {
2     static public function main() {
3         var map = new haxe.ds.IntMap<String>();

```

```

4     var f = map.set.bind(_, "12");
5     $type(map.set); // Int -> String -> Void
6     $type(f); // Int -> Void
7     f(1);
8     f(2);
9     f(3);
10    trace(map); // {1 => 12, 2 => 12, 3 => 12}
11  }
12 }

```

Line 4 binds the function `map.set` to a variable named `f`, and applies `12` as second argument. The underscore `_` is used to denote that this argument is not bound, which is shown by comparing the types of `map.set` and `f`: The bound `String` argument is effectively cut from the type, turning a `Int->String->Void` type into `Int->Void`.

A call to `f(1)` then actually invokes `map.set(1, "12")`, the calls to `f(2)` and `f(3)` are analogous. The last line proves that all three indices indeed are mapped to the value `"12"`.

The underscore `_` can be skipped for trailing arguments, so the first argument could be bound through `map.set.bind(1)`, yielding a `String->Void` function that sets a new value for index `1` on invocation.

#### Trivia: Callback

Prior to Haxe 3, Haxe used to know a `callback`-keyword which could be called with a function argument followed by any number of binding arguments. The name originated from a common usage where a callback-function is created with the `this`-object being bound. Callback would allow binding of arguments only from left to right as there was no support for the underscore `_`. The choice to use an underscore was controversial and several other suggestions were made, none of which were considered superior. After all, the underscore `_` at least looks like it's saying "fill value in here", which nicely describes its semantics.

## 6.9 Metadata

Several constructs can be attributed with custom metadata:

- `class` and `enum` declarations
- Class fields
- Enum constructors
- Expressions

These metadata information can be obtained at runtime through the `haxe.rtti.Meta` API:

```

1 import haxe.rtti.Meta;
2
3 @author("Nicolas")
4 @debug
5 class MyClass {
6     @range(1, 8)
7     var value: Int;
8
9     @broken

```

```

10  @:noCompletion
11  static function method() { }
12  }
13
14  class Main {
15  static public function main() {
16    // { author : ["Nicolas"], debug : null }
17    trace(Meta.getType(MyClass));
18    // [1,8]
19    trace(Meta.getFields(MyClass).value.range);
20    // { broken: null }
21    trace(Meta.getStatics(MyClass).method);
22  }
23  }

```

We can easily identify metadata by the leading @ character, followed by the metadata name and, optionally, by a number of comma-separated constant arguments enclosed in parentheses.

- Class `MyClass` has an `author` metadata with a single `String` argument `"Nicolas"`, as well as a `debug` metadata without arguments.
- The member variable `value` has a `range` metadata with two `Int` arguments `1` and `8`.
- The static method `method` has a `broken` metadata without arguments, as well as a `:noCompletion` metadata without arguments.

The `main` method accesses these metadata values using their API. The output reveals the structure of the obtained data:

- There is a field for each metadata, with the field name being the metadata name.
- The field values correspond to the metadata arguments. If there are no arguments, the field value is `null`. Otherwise the field value is an array with one element per argument.
- Metadata starting with `:` is omitted. This kind of metadata is known as *compiler metadata*.

Allowed values for metadata arguments are:

- Constants (5.2)
- Arrays declarations (5.5) (if all their elements qualify)
- Object declarations (5.6) (if all their field values qualify)

**Built-in Compiler Metadata** An exhaustive list of all defined metadata can be obtained by running `haxe --help-metas` from command line.

See also the Compiler Metadata list (8.1).

## 6.10 Access Control

Access control can be used if the basic visibility (4.4.1) options are not sufficient. It is applicable at *class-level* and at *field-level* and knows two directions:

**Allowing access:** The target is granted access to the given class or field by using the `:allow(target)` metadata (6.9).

**Forcing access:** A target is forced to allow access to the given class or field by using the `:access(target)` metadata (6.9).

In this context, a *target* can be the dot-path (3.7) to

- a *class field*,
- a *class* or *abstract* type, or
- a *package*.

Target does *not* respect imports, so the fully qualified path has to be used.

If it is a class or abstract type, access modification extends to all fields of that type. Likewise, if it is a package, access modification extends to all types of that package and recursively to all fields of these types.

```
1 @:allow(Main)
2 class MyClass {
3     static private var foo: Int;
4 }
5
6 class Main {
7     static public function main() {
8         MyClass.foo;
9     }
10 }
```

Here, `MyClass.foo` can be accessed from the `main`-method because `MyClass` is annotated with `@:allow(Main)`. This would also work with `@:allow(Main.main)` and both versions could alternatively be annotated to the field `foo` instead of the class `MyClass`:

```
1 class MyClass {
2     @:allow(Main.main)
3     static private var foo: Int;
4 }
5
6 class Main {
7     static public function main() {
8         MyClass.foo;
9     }
10 }
```

If a type cannot be modified to allow this kind of access, the accessing method may force access:

```
1 class MyClass {
2     static private var foo: Int;
3 }
4
5 class Main {
6     @:access(MyClass.foo)
7     static public function main() {
8         MyClass.foo;
9     }
10 }
```

The `@:access(MyClass.foo)` annotation effectively subverts the visibility of the `foo` field within the `main`-method.

#### Trivia: On the choice of metadata

The access control language feature uses the Haxe metadata syntax instead of additional language-specific syntax. There are several reasons for that:

- Additional syntax often adds complexity to the language parsing, and also adds (too) many keywords.
- Additional syntax requires additional learning by the language user, whereas metadata syntax is something that is already known.
- The metadata syntax is flexible enough to allow extension of this feature.
- The metadata can be accessed/generated/modified by Haxe macros.

Of course, the main drawback of using metadata syntax is that you get no error report in case you misspell either the metadata key (`@:acess` for instance) or the class/package name. However, with this feature you will get an error when you try to access a private field that you are not allowed to, therefore there is no possibility for silent errors.

#### Since Haxe 3.1.0

If access is allowed to an interface ([2.3.3](#)), it extends to all classes implementing that interface:

```
1 class MyClass {
2     @:allow(I)
3     static private var foo: Int;
4 }
5
6 interface I { }
7
8 class Main implements I {
9     static public function main() {
10         MyClass.foo;
11     }
12 }
```

This is also true for access granted to parent classes, in which case it extends to all child classes.

#### Trivia: Broken feature

Access extension to child classes and implementing classes was supposed to work in Haxe 3.0 and even documented accordingly. While writing this manual it was found that this part of the access control implementation was simply missing.

## 6.11 Inline constructors

#### Since Haxe 3.1.0

If a constructor is declared to be inline (4.4.2), the compiler may try to optimize it away in certain situations. There are several requirements for this to work:

- The result of the constructor call must be directly assigned to a local variable.
- The expression of the constructor field must only contain assignments to its fields.

The following example demonstrates constructor inlining:

```
1 class Point {
2   public var x:Float;
3   public var y:Float;
4
5   public inline function new(x:Float, y:Float) {
6     this.x = x;
7     this.y = y;
8   }
9 }
10
11 class Main {
12   static public function main() {
13     var pt = new Point(1.2, 9.3);
14   }
15 }
```

A look at the JavaScript output reveals the effect:

```
1 Main.main = function() {
2   var pt_x = 1.2;
3   var pt_y = 9.3;
4 };
```

# **Part II**

## **Compiler Reference**

## Chapter 7

# Compiler Usage

**Basic Usage** The Haxe Compiler is typically invoked from command line with several arguments which have to answer two questions:

- What should be compiled?
- What should the output be?

To answer the first question, it is usually sufficient to provide a class path via the `-cp path` argument, along with the main class to be compiled via the `-main dot_path` argument. The Haxe Compiler then resolves the main class file and begins compilation.

The second question usually comes down to providing an argument specifying the desired target. Each Haxe target has a dedicated command line switch, such as `-js file_name` for JavaScript and `-php directory` for PHP. Depending on the nature of the target, the argument value is either a file name (for `-js`, `-swf` and `neko`) or a directory path.

**Common arguments** *Input:*

**-cp path** Adds a class path where `.hx` source files or packages (sub-directories) can be found.

**-lib library\_name** Adds a [Haxelib](#) ([Chapter 11](#)) library. By default the most recent version in the local Haxelib repository is used. To use specific version, `-lib library_name:version` can be used.

**-main dot\_path** Sets the main class.

*Output:*

**-js file\_name** Generates JavaScript ([12.1](#)) source code in specified file.

**-as3 directory** Generates ActionScript 3 source code in specified directory.

**-swf file\_name** Generates the specified file as Flash ([12.2](#)) `.swf`.

**-neko file\_name** Generates Neko ([12.3](#)) binary as specified file.

**-php directory** Generates PHP ([12.4](#)) source code in specified directory.

**-cpp directory** Generates C++ ([12.5](#)) source code in specified directory and compiles it using native C++ compiler.

**-cs directory** Generates C# ([12.7](#)) source code in specified directory.



**-java directory** Generates Java ([12.6](#)) source code in specified directory and compiles it using the Java Compiler.

**-python file\_name** Generates Python ([12.8](#)) source code in the specified file.

## 7.1 Global Compiler Flags

Starting from Haxe 3.0, you can get the list of supported compiler flags ([6.1](#)) by running `haxe --help-defines`

Flag	Description
absolute-path	Print absolute file path in trace output
advanced-telemetry	Allow the SWF to be measured with Monocle tool
analyzer	Use static analyzer for optimization (experimental)
as3	Defined when outputting flash9 as3 source code
check-xml-proxy	Check the used fields of the xml proxy
core-api	Defined in the core api context
core-api-serialize	Mark some generated core api classes with the Serializable a
cppia	Generate experimental cpp instruction assembly
dce=<mode:std full no>	Set the dead code elimination (8.2) mode (default std)
dce-debug	Show dead code elimination (8.2) log
debug	Activated when compiling with -debug
display	Activated during completion
dll-export	GenCPP experimental linking
dll-import	GenCPP experimental linking
doc-gen	Do not perform any removal/change in order to correctly g
dump	Dump the complete typed AST for internal debugging in a c
dump-dependencies	Dump the classes dependencies in a dump subdirectory
dump-ignore-var-ids	Remove variable IDs from non-pretty dumps (helps with di
erase-generics	Erase generic classes on C#
fdb	Enable full flash debug infos for FDB interactive debugging
file-extension	Output filename extension for cpp source code
flash-strict	More strict typing for flash target
flash-use-stage	Keep the SWF library initial stage
force-lib-check	Force the compiler to check -net-lib and -java-lib added clas
force-native-property	Tag all properties with :nativeProperty metadata for 3.1
format-warning	Print a warning for each formatted string, for 2.x compatibili
gencommon-debug	GenCommon internal
haxe-boot	Given the name 'haxe' to the flash boot class instead of a gen
haxe-ver	The current Haxe version value
hxcpp-api-level	Provided to allow compatibility between hxcpp versions
include-prefix	prepend path to generated include files
interp	The code is compiled to be run with --interp
java-ver=[version:5-7]	Sets the Java version to be targeted
js-classic	Don't use a function wrapper and strict mode in JS output
js-es5	Generate JS for ES5-compliant runtimes
js-unflatten	Generate nested objects for packages and types
keep-old-output	Keep old source files in the output directory (for C#/Java)
loop-unroll-max-cost	Maximum cost (number of expressions * iterations) before l
macro	Defined when code is compiled in the macro context (9)
macro-times	Display per-macro timing when used with --times
net-ver=<version:20-45>	Sets the .NET version to be targeted
net-target=<name>	Sets the .NET target. Defaults to net. xbox, micro _(Micro Fr
neko-source	Output neko source instead of bytecode
neko-v1	Keep Neko 1.x compatibility
network-sandbox	Use local network sandbox instead of local file access one
no-compilation	Disable CPP final compilation
no-copt	Disable completion optimization _(for debug purposes)_
no-debug	Remove all debug macros from cpp output
no-deprecation-warnings	Do not warn if fields annotated with @:deprecated are us
no-flash-override	Change overrides on some basic classes into HX suffixed me
no-opt	105Disable optimizations
no-pattern-matching	Disable pattern matching (6.4)
no-inline	Disable inlining (4.4.2)
no-root	GenCS internal
no-macro-cache	Disable macro context caching
no-simplify	Disable simplification filter
no-swf-compress	Disable SWF output compression
no-traces	Disable all trace calls
no-prefix	Compiled with --no-prefix

## Chapter 8

# Compiler Features

### 8.1 Built-in Compiler Metadata

Starting from Haxe 3.0, you can get the list of defined compiler metadata by running `haxe --help-metas`

Global	
Metadata	Description
@:abi	Function ABI/calling convention
@:abstract	Sets the underlying class implementation as abstract
@:access <i>_(Target path)</i> _	Forces private access to package type or field
@:allow <i>_(Target path)</i> _	Allows private access from package type or field
@:analyzer	Used to configure the static analyzer
@:annotation	Annotation ( <code>@interface</code> ) definitions on <code>-j</code>
@:arrayAccess	Allows Array access (2.8.3) on an abstract
@:autoBuild <i>_(Build macro call)</i> _	Extends <code>@:build</code> metadata to all extending
@:bind	Override Swf class declaration
@:bitmap <i>_(Bitmap file path)</i> _	_Embeds given bitmap data into the class (m
@:bridgeProperties	Creates native property bridges for all Haxe
@:build <i>_(Build macro call)</i> _	Builds a class or enum from a macro. See Typ
@:buildXml	Specify xml data to be injected into Build.xml
@:callable	Abstract forwards call to its underlying type
@:classCode	Used to inject platform-native code into a cla
@:commutative	Declares an abstract operator as commutativ
@:compilerGenerated	Marks a field as generated by the compiler. S
@:coreApi	Identifies this class as a core api class (forces
@:coreType	Identifies an abstract as core type (2.8.7) so th
@:cppFileCode	Code to be injected into generated cpp file
@:cppInclude	File to be included in generated cpp file
@:cppNamespaceCode	
@:dce	Forces Dead Code Elimination (8.2) even wh
@:debug	Forces debug information to be generated in
@:decl	
@:defParam	
@:delegate	Automatically added by <code>-net-lib</code> on deleg
@:depend	
@:deprecated	Automatically added by <code>-java-lib</code> on clas
@:event	Automatically added by <code>-net-lib</code> on even
@:enum	Defines finite value sets to abstract definition
@:expose <i>_(?Name=Class path)</i> _	Makes the class available on the <code>window obj</code>
@:extern	Marks the field as extern so it is not generate
@:fakeEnum <i>_(Type name)</i> _	Treat enum as collection of values of the spec
@:file <i>(File path)</i>	Includes a given binary file into the target Sw
@:final	Prevents a class from being extended
@:font <i>_(TTF path Range String)</i> _	Embeds the given TrueType font into the clas
@:forward <i>_(List of field names)</i> _	Forwards field access (2.8.6) to underlying ty
@:from	Specifies that the field of the abstract is a cas
@:functionCode	
@:functionTailCode	
@:generic	Marks a class or class field as generic (3.3) so
@:genericBuild	Builds instances of a type using the specified
@:getter <i>_(Class field name)</i> _	Generates a native getter function on the giv
@:hack	Allows extending classes marked as <code>@:final</code>
@:headerClassCode	Code to be injected into the generated class,
@:headerCode	Code to be injected into the generated head
@:headerNamespaceCode	
@:hxGen	Annotates that an extern class was generated
@:iffFeature <i>_(Feature name)</i> _	Causes a field to be kept by DCE (8.2) if the g
@:include	
@:initPackage	
@:internal	Generates the annotated field/class with in
@:isVar	Forces a physical field to be generated for pr
@:javaCanonical <i>_(Output type package,Output type name)</i> _	Used by the Java target to annotate the canon
@:jsRequire	Generate javascript module require expressio
@:keep	Causes a field or type to be kept by DCE (8.2)
@:keepInit	Causes a class to be kept by DCE (8.2) even i
@:keepSub	Extends <code>@:keep</code> metadata to all implement

## 8.2 Dead Code Elimination

Dead Code Elimination or *DCE* is a compiler feature which removes unused code from the output. After typing, the compiler evaluates the DCE entry-points (usually the main-method) and recursively determines which fields and types are used. Used fields are marked accordingly and unmarked fields are then removed from their classes.

DCE has three modes which are set when invoking the command line:

**-dce std:** Only classes in the Haxe Standard Library are affected by DCE. This is the default setting on all targets.

**-dce no:** No DCE is performed.

**-dce full:** All classes are affected by DCE.

The DCE-algorithm works well with typed code, but may fail when dynamic (2.7) or reflection (10.7) is involved. This may require explicit marking of fields or classes as being used by attributing the following metadata:

**@:keep:** If used on a class, the class along with all fields is unaffected by DCE. If used on a field, that field is unaffected by DCE.

**@:keepSub:** If used on a class, it works like @:keep on the annotated class as well as all sub-classes.

**@:keepInit:** Usually, a class which had all fields removed by DCE (or is empty to begin with) is removed from the output. By using this metadata, empty classes are kept.

If a class needs to be marked with @:keep from the command line instead of editing its source code, there is a compiler macro available for doing so: `--macro keep('type dot path')` See the [haxe.compiler.Compiler.keep API](#) for details of this macro. It will mark package, module or sub-type to be kept by DCE and includes them for compilation.

The compiler automatically defines the flag `dce` with a value of either `"std"`, `"no"` or `"full"` depending on the active mode. This can be used in conditional compilation (6.1).

### Trivia: DCE-rewrite

DCE was originally implemented in Haxe 2.07. This implementation considered a function to be used when it was explicitly typed. The problem with that was that several features, most importantly interfaces, would cause all class fields to be typed in order to verify type-safety. This effectively subverted DCE completely, prompting the rewrite for Haxe 2.10.

### Trivia: DCE and try.haxe.org

DCE for the JavaScript target saw vast improvements when the website <http://try.haxe.org> was published. Initial reception of the generated JavaScript code was mixed, leading to a more fine-grained selection of which code to eliminate.

## 8.3 Compiler Services

### 8.3.1 Overview

The rich type system (3) of the Haxe Compiler makes it difficult for IDEs and editors to provide accurate completion information. Between type inference (3.6) and macros (9), it would require a

substantial amount of work to replicate the required processing. This is why the Haxe Compiler comes with a built-in completion mode for third-party software to use.

All completion is triggered using the `--display file@position[@mode]` compiler argument. The required arguments are:

**file:** The file to check for completion. This must be an absolute or relative path to a `.hx` file. It does not respect any class paths or libraries.

**position:** The byte position (not character position) of where to check for completion in the given file.

**mode:** The completion mode to use (see below).

We will look into the following completion modes in detail:

**Field access (8.3.2):** Provides a list of fields that can be accessed on a given type.

**Call argument (8.3.3):** Reports the type of the function which is currently being called.

**Type path (8.3.4):** Lists sub-packages, sub-types and static fields.

**Usage (8.3.5):** Lists all occurrences of a given type, field or variable in all compiled files. (mode: "usage")

**Position (8.3.6):** Reports the position of where a given type, field or variable is defined. (mode: "position")

**Top-level (8.3.7):** Lists all identifiers which are available at a given position. (mode: "toplevel")

Due to Haxe being a very fast compiler, it is often sufficient to rely on the normal compiler invocation for completion. For bigger projects Haxe provides a server mode (8.3.8) which ensures that only those files are re-compiled that actually changed or had any of their dependencies changes.

### General notes on the interface

- The position-argument can be set to 0 if the file in question contains a pipeline `|` character at the position of interest. This is very useful for demonstration and testing as it allows us to ignore the byte-counting process a real IDE would have to do. The examples in this section are making use of this feature. Note that this only works in places where `|` is not valid syntax otherwise, e.g. after dots (`. |`) and opening parentheses (`( |`).
- The output is HTML-escaped so that `&`, `<` and `>` become `&amp;`, `&lt;` and `&gt;` respectively.
- Otherwise any documentation output is preserved which means longer documentation might include new-line and tab-characters as it does in the source files.
- When run in completion mode, the compiler does not display errors but instead tries to ignore them or recover from them. If a critical error occurs while getting completion, the Haxe Compiler prints the error message instead of the completion output. Any non-XML output can be treated as a critical error message.

### 8.3.2 Field access completion

Field completion is triggered after a dot `.` character to list the fields that are available on the given type. The compiler parses and types everything up to the point of completion and then outputs the relevant information to stderr:

```
1 class Main {
2     public static function main() {
3         trace("Hello".|
4     }
5 }
```

If this file is saved to `Main.hx`, the completion can be invoked using the command `haxe --display Main.hx@0`. The output looks similar to this (we omit several fields for brevity and improve the formatting for readability):

```
1 <list>
2 <i n="length">
3     <t>Int</t>
4     <d>
5         The number of characters in this String.
6     </d>
7 </i>
8 <i n="charAt">
9     <t>index : Int -> String</t>
10    <d>
11        Returns the character at position index of this String.
12        If index is negative or exceeds this.length, the empty String
13        "" is returned.
14    </d>
15 </i>
16 <i n="charCodeAt">
17     <t>index : Int -> Null<Int></t>
18     <d>
19         Returns the character code at position index of this String.
20         If index is negative or exceeds this.length, null is returned.
21         To obtain the character code of a single character, "x".code can
22         be used instead to inline the character code at compile time.
23         Note that this only works on String literals of length 1.
24     </d>
25 </i>
26 </list>
```

The XML structure follows:

- The document node `list` encloses several nodes `i`, each representing a field.
- The `n` attribute contains the name of the field.
- The `t` node contains the type of the field.
- the `d` node contains the documentation of the field.

Since Haxe 3.2.0

When compiling with `-D display-details`, each field additionally has a `k` attribute which can either be `var` or `method`. This allows distinguishing method fields from variable fields that have a function type.

### 8.3.3 Call argument completion

Call argument completion is triggered after an opening parenthesis character `(` to show the type of the function that is currently being called. It works for any function call as well as constructor calls:

```
1 class Main {
2   public static function main() {
3     trace("Hello".split(|
4   }
5 }
```

If this file is saved to `Main.hx`, the completion can be invoked using the command `haxe --display Main.hx@0`. The output looks like this:

```
1 <type>
2 delimiter : String -&gt; Array<String>;
3 </type>
```

IDEs can parse this to recognize that the called function requires one argument named `delimiter` of type `String` and returns an `Array<String>`.

#### Trivia: Problems with the output structure

We acknowledge that the current format requires a bit of manual parsing which can be annoying. In the future we might look into providing a more structured output, especially for functions.

### 8.3.4 Type path completion

Type path completion can occur in import declarations (3.7.2), using declarations (6.3) or any place a type is referenced. We can identify three different cases:

**package completion** This lists all sub-packages of the `haxe` package as well as all modules in that package:

```
1 import haxe.|

1 <list>
2 <i n="CallStack"><t></t><d></d></i>
3 <i n="Constraints"><t></t><d></d></i>
4 <i n="DynamicAccess"><t></t><d></d></i>
5 <i n="EnumFlags"><t></t><d></d></i>
6 <i n="EnumTools"><t></t><d></d></i>
7 <i n="Http"><t></t><d></d></i>
8 <i n="Int32"><t></t><d></d></i>
9 <i n="Int64"><t></t><d></d></i>
10 <i n="Json"><t></t><d></d></i>
11 <i n="Log"><t></t><d></d></i>
12 <i n="PosInfos"><t></t><d></d></i>
```



```

13 <i n="Resource"><t></t><d></d></i>
14 <i n="Serializer"><t></t><d></d></i>
15 <i n="Template"><t></t><d></d></i>
16 <i n="Timer"><t></t><d></d></i>
17 <i n="Ucs2"><t></t><d></d></i>
18 <i n="Unserializer"><t></t><d></d></i>
19 <i n="Utf8"><t></t><d></d></i>
20 <i n="crypto"><t></t><d></d></i>
21 <i n="ds"><t></t><d></d></i>
22 <i n="extern"><t></t><d></d></i>
23 <i n="format"><t></t><d></d></i>
24 <i n="io"><t></t><d></d></i>
25 <i n="macro"><t></t><d></d></i>
26 <i n="remoting"><t></t><d></d></i>
27 <i n="rtti"><t></t><d></d></i>
28 <i n="unit"><t></t><d></d></i>
29 <i n="web"><t></t><d></d></i>
30 <i n="xml"><t></t><d></d></i>
31 <i n="zip"><t></t><d></d></i>
32 </list>

```

**import module completion** This lists all sub-types (3.7.1) of the module `haxe.Unserializer` as well as all its public static fields (because these can be imported too):

```

1 import haxe.Unserializer.|

1 <list>
2 <i n="DEFAULT_RESOLVER">
3   <t>haxe.TypeResolver</t>
4   <d>
5     This value can be set to use custom type resolvers.
6
7     A type resolver finds a Class or Enum instance from a given String
8     .
9     By default, the haxe Type Api is used.
10
11    A type resolver must provide two methods:
12
13    1. resolveClass(name:String):Class<Dynamic>; is called to
14       determine a Class from a class name
15    2. resolveEnum(name:String):Enum<Dynamic>; is called to
16       determine an Enum from an enum name
17
18    This value is applied when a new Unserializer instance is created.
19    Changing it afterwards has no effect on previously created
20    instances.
21  </d>
22 </i>
23 <i n="run">
24   <t>v : String -> Dynamic</t>
25   <d>

```

```

25     Unserializes `v` and returns the according value.
26
27     This is a convenience function for creating a new instance of
28     Unserializer with `v` as buffer and calling its unserialize()
29     method once.
30 </d>
31 </i>
32 <i n="TypeResolver"><t></t><d></d></i>
33 <i n="Unserializer"><t></t><d></d></i>
34 </list>

```

```
1 using haxe.Unserializer.|
```

**other module completion** This lists all sub-types ([3.7.1](#)) of the module `haxe.Unserializer`:

```
1 using haxe.Unserializer.|
```

```

1 class Main {
2     static public function main() {
3         var x:haxe.Unserializer.|
4     }
5 }

```

```

1 <list>
2 <i n="TypeResolver"><t></t><d></d></i>
3 <i n="Unserializer"><t></t><d></d></i>
4 </list>

```

### 8.3.5 Usage completion

Since Haxe 3.2.0

Usage completion is enabled by using the "usage" mode argument (see [Overview \(Section 8.3.1\)](#)). We demonstrate it here using a local variable. Note that it would work with fields and types the same way:

```

1 class Main {
2     public static function main() {
3         var a = 1;
4         var b = a + 1;
5         trace(a);
6         a.|
7     }
8 }

```

If this file is saved to `Main.hx`, the completion can be invoked using the command `haxe --display Main.hx@0@usage`. The output looks like this:

```

1 <list>
2 <pos>main.hx:4: characters 9-10</pos>
3 <pos>main.hx:5: characters 7-8</pos>
4 <pos>main.hx:6: characters 1-2</pos>
5 </list>

```

### 8.3.6 Position completion

Since Haxe 3.2.0

Position completion is enabled by using the "position" mode argument (see [Overview \(Section 8.3.1\)](#)). We demonstrate it using a field. Note that it would work with local variables and types the same way:

```
1 class Main {
2     static public function main() {
3         "foo".split.|
4     }
```

If this file is saved to Main.hx, the completion can be invoked using the command `haxe --display Main.hx@0@position`. The output looks like this:

```
1 <list>
2 <pos>std/string.hx:124: characters 1-54</pos>
3 </list>
```

#### Trivia: Effects of omitting a target specifier

In this example the compiler reports the standard String.hx definition which does not actually have an implementation. This happens because we did not specify any target, which is allowed in completion-mode. If the command line arguments included, say, `-neko neko.n`, the reported position would instead be `std/neko/_std/string.hx:84: lines 84-98`.

### 8.3.7 Top-level completion

Since Haxe 3.2.0

Top-level completion displays all identifiers which the Haxe Compiler knows about at a given compilation position. This is the only completion method for which we need a real position argument in order to demonstrate its effect:

```
1 class Main {
2     static public function main() {
3         var a = 1;
4     }
5 }
6
7 enum MyEnum {
8     MyConstructor1;
9     MyConstructor2(s:String);
10 }
```

If this file is saved to Main.hx, the completion can be invoked using the command `haxe --display Main.hx@63@toplevel`. The output looks similar to this (we omit several entries for brevity):

```
1 <il>
2 <i k="local" t="Int">a</i>
3 <i k="static" t="Void -&gt; Unknown<0>&gt;">main</i>
```

```

4 <i k="enum" t="MyEnum">MyConstructor1</i>
5 <i k="enum" t="s : String -&gt; MyEnum">MyConstructor2</i>
6 <i k="package">sys</i>
7 <i k="package">haxe</i>
8 <i k="type" p="Int">Int</i>
9 <i k="type" p="Float">Float</i>
10 <i k="type" p="MyEnum">MyEnum</i>
11 <i k="type" p="Main">Main</i>
12 </il>

```

The structure of the XML depends on the `k` attribute of each entry. In all cases the node value of the `i` nodes contains the relevant name.

**local, member, static, enum, global:** The `t` attribute holds the type of the variable or field.

**global, type:** The `p` attribute holds the path of the module which contains the type or field.

### 8.3.8 Completion server

To get the best speed for both compilation and completion, you can use the `--wait` command-line parameter to start a Haxe compilation server. You can also use `-v` to have the server print the log. Here's an example:

```
1 haxe -v --wait 6000
```

You can then connect to the Haxe server, send commandline parameters followed by a 0 byte and, then, read the response (either completion result or errors).

Use the `--connect` commandline parameter to have Haxe send its compilation commands to the server instead of executing them directly :

```
1 haxe --connect 6000 myproject.hxml
```

Please note that you can use the parameter `--cwd` at the first sent command line to change the Haxe server's current working directory. Usually class paths and other files are relative to your project.

**How it works** The compilation server will cache the following things:

**parsed files** the files will only get parsed again if they are modified or if there was a parse error

**haxelib calls** the previous results of haxelib calls will be reused (only for completion : they are ignored when doing a compilation)

**typed modules** compilation modules will be cached after a successful compilation and can be reused in later compilation/completions if none of their dependencies have been modified

You can get precise reading of the times spent by the compiler and how using the compilation server affects them by adding `--times` to the command line.

**Protocol** As the following Haxe/Neko example shows, you can simply connect on the server port and send all commands (one per line) ending with a 0 binary char. You can, then, read the results.

Macros and other commands can log events which are not errors. From the command line, we can see the difference between what is printed on `stdout` and what is print on `stderr`. This is not the case in socket mode. In order to differentiate between the two, log messages (not errors)

are prefixed with a

x01 character and all newline-characters in the message are replaced by the same x01 character.

Warnings and other messages can also be considered errors, but are not fatal ones. If a fatal error occurred, it will send a single

x02 message-line.

Here's some code that will treat connection to the server and handle the protocol details:

```
1 class Test {
2     static function main() {
3         var newline = "\textbackslash n";
4         var s = new neko.net.Socket();
5         s.connect(new neko.net.Host("127.0.0.1"), 6000);
6         s.write("--cwd /my/project" + newline);
7         s.write("myproject.html" + newline);
8         s.write("\textbackslash 000");
9
10        var hasError = false;
11        for (line in s.read().split(newline))
12        {
13            switch (line.charCodeAt(0)) {
14                case 0x01:
15                    neko.Lib.print(line.substr(1).split("\
16                        textbackslash x01").join(newline));
17                case 0x02:
18                    hasError = true;
19                default:
20                    neko.io.File.stderr().writeString(line + newline);
21            }
22        }
23        if (hasError) neko.Sys.exit(1);
24    }
25 }
```

**Effect on macros** The compilation server can have some side effects on macro execution (9).

## 8.4 Resources

Haxe provides a simple resource embedding system that can be used for embedding files directly into the compiled application.

While it may be not optimal to embed large assets such as images or music in the application file, it comes in very handy for embedding smaller resources like configuration or XML data.

### 8.4.1 Embedding resources

External files are embedded using the *-resource* compiler argument:

what to use for listing  
of non-haxe code like  
html?

```
1 -resource hello_message.txt@welcome
```

The string after the @ symbol is the *resource identifier* which is used in the code for retrieving the resource. If it is omitted (together with the @ symbol) then the file name will become the resource identifier.

### 8.4.2 Retrieving text resources

To retrieve the content of an embedded resource we use the static method `getString` of `haxe.Resource`, passing a *resource identifier* to it:

```
1 class Main {
2     static function main() {
3         trace(haxe.Resource.getString("welcome"));
4     }
5 }
```

The code above will display the content of the `hello_message.txt` file that we included earlier using `welcome` as the identifier.

### 8.4.3 Retrieving binary resources

While it's not recommended to embed large binary files in the application, it still may be useful to embed binary data. The binary representation of an embedded resource can be accessed using the static method `getBytes` of `haxe.Resource`:

```
1 class Main {
2     static function main() {
3         var bytes = haxe.Resource.getBytes("welcome");
4         trace(bytes.readString(0, bytes.length));
5     }
6 }
```

The return type of `getBytes` method is `haxe.io.Bytes`, which is an object providing access to individual bytes of the data.

### 8.4.4 Implementation details

Haxe uses the target platform's native resource embedding if there is one, otherwise it provides its own implementation.

- *Flash* resources are embedded as `ByteArray` definitions
- *C#* resources are included in the compiled assembly
- *Java* resources are packed in the resulting JAR file
- *C++* resources are stored in global byte array constants.
- *JavaScript* resources are serialized in Haxe serialization format and stored in a static field of `haxe.Resource` class.
- *Neko* resources are stored as strings in a static field of `haxe.Resource` class.

## 8.5 Runtime Type Information

The Haxe compiler generates runtime type information (RTTI) for classes that are annotated or extend classes that are annotated with the `:rtti` metadata. This information is stored as a XML string in a static field `__rtti` and can be processed through `haxe.rtti.XmlParser`. The resulting structure is described in [RTTI structure](#) (Section 8.5.1).

Since Haxe 3.2.0

The type `haxe.rtti.Rtti` has been introduced in order to simplify working with RTTI. Retrieving this information is now very easy:

```
1 @:rtti
2 class Main {
3     var x:String;
4     static function main() {
5         var rtti = haxe.rtti.Rtti.getRtti(Main);
6         trace(rtti);
7     }
8 }
```

### 8.5.1 RTTI structure

#### General type information

**path:** The type path (3.7) of the type.

**module:** The type path of the module (3.7) containing the type.

**file:** The full slash path of the .hx file containing the type. This might be `null` in case there is no such file, e.g. if the type is defined through a macro (9).

**params:** An array of strings representing the names of the type parameters (3.2) the type has. As of Haxe 3.2.0, this does not include the constraints (3.2.1).

**doc:** The documentation of the type. This information is only available if the compiler flag (6.1) `-D use_rtti_doc` was in place. Otherwise, or if the type has no documentation, the value is `null`.

**isPrivate:** Whether or not the type is private (3.7.1).

**platforms:** A list of strings representing the targets where the type is available.

**meta:** The meta data the type was annotated with.

#### Class type information

**isExtern:** Whether or not the class is extern (6.2).

**isInterface:** Whether or not the class is actually an interface (2.3.3).

**superClass:** The class' parent class defined by its type path and list of type parameters.

**interfaces:** The list of interfaces defined by their type path and list of type parameters.

**fields:** The list of member class fields (4), described in [Class field information](#) (Section 8.5.1).

**statics:** The list of static class fields, described in [Class field information](#) (Section 8.5.1).

**tdynamic:** The type which is dynamically implemented (2.7.2) by the class or `null` if no such type exists.

### Enum type information

**isExtern:** Whether or not the enum is extern (6.2).

**constructors:** The list of enum constructors.

### Abstract type information

**to:** An array containing the defined implicit to casts (2.8.1).

**from:** An array containing the defined implicit from casts (2.8.1).

**impl:** The class type information (8.5.1) of the implementing class.

**athis:** The underlying type (2.8) of the abstract.

### Class field information

**name:** The name of the field.

**type:** The type of the field.

**isPublic:** Whether or not the field is public (4.4.1).

**isOverride:** Whether or not the field overrides (4.4.4) another field.

**doc:** The documentation of the field. This information is only available if the compiler flag (6.1) `-D use_rtti_doc` was in place. Otherwise, or if the field has no documentation, the value is `null`.

**get:** The read access behavior (4.2) of the field.

**set:** The write access behavior (4.2) of the field.

**params:** An array of strings representing the names of the type parameters (3.2) the field has. As of Haxe 3.2.0, this does not include the constraints (3.2.1).

**platforms:** A list of strings representing the targets where the field is available.

**meta:** The meta data the field was annotated with.

**line:** The line number where the field is defined. This information is only available if the field has an expression. Otherwise the value is `null`.

**overloads:** The list of available overloads for the fields or `null` if no overloads exists.



### Enum constructor information

**name:** The name of the constructor.

**args:** The list of arguments the constructor has or `null` if no arguments are available.

**doc:** The documentation of the constructor. This information is only available if the compiler flag (6.1) `-D use_rtti_doc` was in place. Otherwise, or if the constructor has no documentation, the value is `null`.

**platforms:** A list of strings representing the targets where the constructor is available.

**meta:** The meta data the constructor was annotated with.

## Chapter 9

# Macros

Macros are without a doubt the most advanced feature in Haxe. They are often perceived as dark magic that only a select few are capable of mastering, yet there is nothing magical (and certainly nothing dark) about them.

**Definition: Abstract Syntax Tree (AST)**

The AST is the result of *parsing* Haxe code into a typed structure. This structure is exposed to macros through the types defined in the file `haxe/macro/Expr.hx` of the Haxe Standard Library.




Figure 9.1: The role of macros during compilation.

A basic macro is a *syntax-transformation*. It receives zero or more expressions (5) and also

returns an expression. If a macro is called, it effectively inserts code at the place it was called from. In that respect, it could be compared to a preprocessor like `#define` in C++, but a Haxe macro is not a textual replacement tool.

We can identify different kinds of macros, which are run at specific compilation stages:

**Initialization Macros:** These are provided by command line using the `--macro` compiler parameter. They are executed after the compiler arguments were processed and the *type context* has been created, but before any typing was done (see [Initialization macros \(Section 9.7\)](#)).

**Build Macros:** These are defined for classes, enums and abstracts through the `@:build` or `@:autoBuild` metadata ([6.9](#)). They are executed per-type, after the type has been set up (including its relation to other types, such as inheritance for classes) but before its fields are typed (see [Type Building \(Section 9.5\)](#)).

**Expression Macros:** These are normal functions which are executed as soon as they are typed.

## 9.1 Macro Context

### Definition: Macro Context

The macro context is the environment in which the macro is executed. Depending on the macro type, it can be considered to be a class being built or a function being typed. Contextual information can be obtained through the `haxe.macro.Context` API.

Haxe macros have access to different contextual information depending on the macro type. Other than querying such information, the context also allows some modifications such as defining a new type or registering certain callbacks. It is important to understand that not all information is available for all macro kinds, as the following examples demonstrate:

- Initialization macros will find that the `Context.getLocal*()` methods return `null`. There is no local type or method in the context of an initialization macro.
- Only build macros get a proper return value from `Context.getBuildFields()`. There are no fields being built for the other macro kinds.
- Build macros have a local type (if incomplete), but no local method, so `Context.getLocalMethod()` returns `null`.

The context API is complemented by the `haxe.macro.Compiler` API detailed in [Initialization macros \(Section 9.7\)](#). While this API is available to all macro kinds, care has to be taken for any modification outside of initialization macros. This stems from the natural limitation of undefined build order ([9.6.3](#)), which could cause e.g. a flag definition through `Compiler.define()` to take effect before or after a conditional compilation ([6.1](#)) check against that flag.

## 9.2 Arguments

Most of the time, arguments to macros are expressions represented as an instance of `enum Expr`. As such, they are parsed but not typed, meaning they can be anything conforming to Haxe's syntax rules. The macro can then inspect their structure, or (try to) get their type using `haxe.macro.Context.typeof()`.

It is important to understand that arguments to macros are not guaranteed to be evaluated, so any intended side-effect is not guaranteed to occur. On the other hand, it is also important to understand that an argument expression may be duplicated by a macro and used multiple times in the returned expression:

```
1 import haxe.macro.Expr;
2
3 class Main {
4     static public function main() {
5         var x = 0;
6         var b = add(x++);
7         trace(x); // 2
8     }
9
10    macro static function add(e:Expr) {
11        return macro $e + $e;
12    }
13 }
```

The macro `add` is called with `x++` as argument and thus returns `x++ + x++` using expression reification (9.3.1), causing `x` to be incremented twice.

### 9.2.1 ExprOf

Since `Expr` is compatible with any possible input, Haxe provides the type `haxe.macro.ExprOf<T>`. For the most part, this type is identical to `Expr`, but it allows constraining the type of accepted expressions. This is useful when combining macros with static extensions (6.3):

```
1 import haxe.macro.Expr;
2 using Main;
3
4 class Main {
5     static public function main() {
6         identity("foo");
7         identity(1);
8         "foo".identity();
9         // Int has no field identity
10        //1.identity();
11    }
12
13    macro static function identity(e:ExprOf<String>) {
14        return e;
15    }
16 }
```

The two direct calls to `identity` are accepted, even though the argument is declared as `ExprOf<String>`. It might come as a surprise that the `Int 1` is accepted, but it is a logical consequence of what was explained about macro arguments (9.2): The argument expressions are never typed, so it is not possible for the compiler to check their compatibility by unifying (3.5).

This is different for the next two lines which are using static extensions (note the `using Main`): For these it is mandatory to type the left side ("`foo`" and `1`) first in order to make sense of the `identity` field access. This makes it possible to check the types against the argument types, which causes `1.identity()` to not consider `Main.identity()` as a suitable field.

## 9.2.2 Constant Expressions

A macro can be declared to expect constant (5.2) arguments:

```
1 class Main {
2     static public function main() {
3         const("foo", 1, 1.5, true);
4     }
5
6     macro static function const(s:String, i:Int, f:Float, b:Bool) {
7         trace(s);
8         trace(i);
9         trace(f);
10        trace(b);
11        return macro null;
12    }
13 }
```

With these it is not necessary to detour over expressions as the compiler can use the provided constants directly.

## 9.2.3 Rest Argument

If the final argument of a macro is of type `Array<Expr>`, the macro accepts an arbitrary number of extra arguments which are available from that array:

```
1 import haxe.macro.Expr;
2
3 class Main {
4     static public function main() {
5         myMacro("foo", a, b, c);
6     }
7
8     macro static function myMacro(e1:Expr, extra:Array<Expr>) {
9         for (e in extra) {
10            trace(e);
11        }
12        return macro null;
13    }
14 }
```

## 9.3 Reification

The Haxe Compiler allows *reification* of expressions, types and classes to simplify working with macros. The syntax for reification is `macro expr`, where `expr` is any valid Haxe expression.

### 9.3.1 Expression Reification

Expression reification is used to create instances of `haxe.macro.Expr` in a convenient way. The Haxe Compiler accepts the usual Haxe syntax and translates it to an expression object. It supports several escaping mechanisms, all of which are triggered by the `$` character:

**`${} and $e{}:`** `Expr -> Expr` This can be used to compose expressions. The expression within the delimiting `{ }` is executed, with its value being used in place.

**`$a{}:`** `Expr -> Array<Expr>` If used in a place where an `Array<Expr>` is expected (e.g. call arguments, block elements), `$a{}`  treats its value as that array. Otherwise it generates an array declaration.

**`$b{}:`** `Array<Expr> -> Expr` Generates a block expression from the given expression array.

**`$i{}:`** `String -> Expr` Generates an identifier from the given string.

**`$p{}:`** `Array<String> -> Expr` Generates a field expression from the given string array.

**`$v{}:`** `Dynamic -> Expr` Generates an expression depending on the type of its argument. This is only guaranteed to work for basic types (2.1) and enum instances (2.4).

Additionally the metadata (6.9) `@:pos(p)` can be used to map the position of the annotated expression to `p` instead of the place it is reified at.

This kind of reification only works in places where the internal structure expects an expression. This disallows `object.${fieldName}`, but `object.$fieldName` works. This is true for all places where the internal structure expects a string:

- field access `object.$name`
- variable name `var $name = 1;`

Since Haxe 3.1.0

- field name `{ $name: 1 }`
- function name `function $name() { }`
- catch variable name `try e() catch($name:Dynamic) { }`

Furthermore, a new expression can be reified by providing `haxe.macro.TypePath` argument: `new $typePath()`

### 9.3.2 Type Reification

Type reification is used to create instances of `haxe.macro.Expr.ComplexType` in a convenient way. It is identified by a `macro : Type`, where `Type` can be any valid type path expression. This is similar to explicit type hints in normal code, e.g. for variables in the form of `var x:Type`.

Each constructor of `ComplexType` has a distinct syntax:

**`TPath:`** `macro : pack.Type`

**`TFunction:`** `macro : Arg1 -> Arg2 -> Return`

**`TAnonymous:`** `macro : { field: Type }`

**`TParent:`** `macro : (Type)`

**`TExtend:`** `macro : {> Type, field: Type }`

**`TOptional:`** `macro : ?Type`

### 9.3.3 Class Reification

It is also possible to use reification to obtain an instance of `haxe.macro.Expr.TypeDefinition`. This is indicated by the `macro class` syntax as shown here:

```
1 class Main {
2     macro static function generateClass(funcName:String) {
3         var c = macro class MyClass {
4             public function new() { }
5             public function $funcName() {
6                 trace($v{funcName} + " was called");
7             }
8         }
9         haxe.macro.Context.defineType(c);
10        return macro new MyClass();
11    }
12
13    public static function main() {
14        var c = generateClass("myFunc");
15        c.myFunc();
16    }
17 }
```

The generated `TypeDefinition` instance is typically passed to `haxe.macro.Context.defineType` in order to add a new type to the calling context (not the macro context itself).

This kind of reification can also be useful to obtain instances of `haxe.macro.Expr.Field`, which are available from the `fields` array of the generated `TypeDefinition`.

## 9.4 Tools

The Haxe Standard Library comes with a set of tool-classes to simplify working with macros. These classes work best as static extensions (6.3) and can be brought into context either individually or as a whole through using `haxe.macro.Tools`. These classes are:

**ComplexTypeTools:** Allows printing `ComplexType` instances in a human-readable way. Also allows determining the `Type` corresponding to a `ComplexType`.

**ExprTools:** Allows printing `Expr` instances in a human-readable way. Also allows iterating and mapping expressions.

**MacroStringTools:** Offers useful operations on strings and string expressions in macro context.

**TypeTools:** Allows printing `Type` instances in a human-readable way. Also offers several useful operations on types, such as unifying (3.5) them or getting their corresponding `ComplexType`.

Furthermore the `haxe.macro.Printer` class has public methods for printing various types as a human-readable format. This can be helpful when debugging macros.

#### Trivia: The tinkerbell library and why `Tools.hx` works

We learned about static extensions that using a *module* implies that all its types are brought into static extension context. As it turns out, such a type can also be a `typedef` (3.1) to another

type. The compiler then considers this type part of the module, and extends static extension accordingly.

This “trick” was first used in Juraj Kirchheim’s *tinkerbelt*<sup>1</sup> library for exactly the same purpose. Tinkerbelt provided many useful macro tools long before they made it into the Haxe Compiler and Haxe Standard Library. It remains the primary library for additional macro tools and offers other useful functionality as well.

## 9.5 Type Building

Type-building macros are different from expression macros in several ways:

- They do not return expressions, but an array of class fields. Their return type must be set explicitly to `Array<haxe.macro.Expr.Field>`.
- Their context (9.1) has no local method and no local variables.
- Their context does have build fields, available from `haxe.macro.Context.getBuildFields()`.
- They are not called directly, but are argument to a `@:build` or `@:autoBuild` metadata (6.9) on a class (2.3) or enum (2.4) declaration.

The following example demonstrates type building. Note that it is split up into two files for a reason: If a module contains a macro function, it has to be typed into macro context as well. This is often a problem for type-building macros because the type to be built could only be loaded in its incomplete state, before the building macro has run. We recommend to always define type-building macros in their own module.

```
1 import haxe.macro.Context;
2 import haxe.macro.Expr;
3
4 class TypeBuildingMacro {
5     macro static public function build(fieldName:String):Array<Field> {
6         var fields = Context.getBuildFields();
7         var newField = {
8             name: fieldName,
9             doc: null,
10            meta: [],
11            access: [AStatic, APublic],
12            kind: FVar(macro : String, macro "my default"),
13            pos: Context.currentPos()
14        };
15        fields.push(newField);
16        return fields;
17    }
18 }
```

```
1 @:build(TypeBuildingMacro.build("myFunc"))
2 class Main {
3     static public function main() {
4         trace(Main.myFunc); // my default
5     }
6 }
```



The `build` method of `TypeBuildingMacro` performs three steps:

1. It obtains the build fields using `Context.getBuildFields()`.
2. It declares a new `haxe.macro.expr.Field` field using the `funcName` macro argument as field name. This field is a `String` variable (4.1) with a default value `"my default"` (from the `kind` field) and is public and static (from the `access` field).
3. It adds the new field to the build field array and returns it.

This macro is argument to the `@:build` metadata on the `Main` class. As soon as this type is required, the compiler does the following:

1. It parses the module file, including the class fields.
2. It sets up the type, including its relation to other types through inheritance (2.3.2) and interfaces (2.3.3).
3. It executes the type-building macro according to the `@:build` metadata.
4. It continues typing the class normally with the fields returned by the type-building macro.

This allows adding and modifying class fields at will in a type-building macro. In our example, the macro is called with a `"myFunc"` argument, making `Main.myFunc` a valid field access.

If a type-building macro should not modify anything, the macro can return `null`. This indicates to the compiler that no changes are intended and is preferable to returning `Context.getBuildFields`

### 9.5.1 Enum building

Building enums (2.4) is analogous to building classes with a simple mapping:

- Enum constructors without arguments are variable fields `FVar`.
- Enum constructors with arguments are method fields `FFun`.

Check if we can build GADTs this way.

```
1 import haxe.macro.Context;
2 import haxe.macro.Expr;
3
4 class EnumBuildingMacro {
5     macro static public function build():Array<Field> {
6         var noArgs = makeEnumField("A", FVar(null, null));
7         var eFunc = macro function(value:Int) { };
8         var fInt = switch (eFunc.expr) {
9             case EFunction(_, f): f;
10            case _: throw "false";
11        }
12        var intArg = makeEnumField("B", FFun(fInt));
13        return [noArgs, intArg];
14    }
15
16    static function makeEnumField(name, kind) {
17        return {
18            name: name,
```

```

19     doc: null,
20     meta: [],
21     access: [],
22     kind: kind,
23     pos: Context.currentPos()
24 }
25 }
26 }

```

```

1 @:build(EnumBuildingMacro.build())
2 enum E { }
3
4 class Main {
5     static public function main() {
6         switch (E.A) {
7             case A:
8             case B(v):
9         }
10    }
11 }

```

Because enum `E` is annotated with a `:build` metadata, the called macro builds two constructors `A` and `B` “into” it. The former is added with the kind being `FVar(null, null)`, meaning it is a constructor without argument. For the latter, we use reification (9.3.1) to obtain an instance of `haxe.macro.Expr.Function` with a singular `Int` argument.

The main method proves the structure of our generated enum by matching (6.4) it. We can see that the generated type is equivalent to this:

```

1 enum E {
2     A;
3     B(value:Int);
4 }

```

## 9.5.2 @:autoBuild

If a class has the `:autoBuild` metadata, the compiler generates `:build` metadata on all extending classes. If an interface has the `:autoBuild` metadata, the compiler generates `:build` metadata on all implementing classes and all extending interfaces. Note that `:autoBuild` does not imply `:build` on the class/interface itself.

```

1 import haxe.macro.Context;
2 import haxe.macro.Expr;
3
4 class AutoBuildingMacro {
5     macro static public
6     function fromInterface():Array<Field> {
7         trace("fromInterface: " + Context.getLocalType());
8         return null;
9     }
10
11     macro static public
12     function fromBaseClass():Array<Field> {
13         trace("fromBaseClass: " + Context.getLocalType());

```

```

14     return null;
15 }
16 }

1 @:autoBuild(AutoBuildingMacro.fromInterface())
2 interface I { }
3
4 interface I2 extends I { }
5
6 @:autoBuild(AutoBuildingMacro.fromBaseClass())
7 class Base { }
8
9 class Main extends Base implements I2 {
10     static public function main() { }
11 }

```

This outputs during compilation:

```

1 AutoBuildingMacro.hx:6:
2   fromInterface: TInst(I2, [])
3 AutoBuildingMacro.hx:6:
4   fromInterface: TInst(Main, [])
5 AutoBuildingMacro.hx:11:
6   fromBaseClass: TInst(Main, [])

```

It is important to keep in mind that the order of these macro executions is undefined, which is detailed in [Build Order](#) (Section 9.6.3).

### 9.5.3 @:genericBuild

Since Haxe 3.1.0

Normal build-macros (9.5) are run per-type and are already very powerful. In some cases it is useful to run a build macro per type *usage* instead, i.e. whenever it actually appears in the code. Among other things this allows accessing the concrete type parameters in the macro.

@:genericBuild is used just like @:build by adding it to a type with the argument being a macro call:

```

1 import haxe.macro.Expr;
2 import haxe.macro.Context;
3 import haxe.macro.Type;
4
5 class GenericBuildMacro1 {
6     static public function build() {
7         switch (Context.getLocalType()) {
8             case TInst(_, [t1]):
9                 trace(t1);
10             case t:
11                 Context.error("Class expected", Context.currentPos());
12         }
13         return null;
14     }
15 }

```

```

1 @:genericBuild(GenericBuildMacro1.build())
2 class MyType<T> { }
3
4 class Main {
5     static function main() {
6         var x:MyType<Int>;
7         var x:MyType<String>;
8     }
9 }

```

When running this example the compiler outputs `TAbstract(Int, [])` and `TInst(String, [])`, indicating that it is indeed aware of the concrete type parameters of `MyType`. The macro logic could use this information to generate a custom type (using `haxe.macro.Context.defineType`) or refer to an existing one. For brevity we return `null` here which asks the compiler to infer (3.6) the type.

In Haxe 3.1 the return type of a `@:genericBuild` macro has to be a `haxe.macro.Type`. Haxe 3.2 allows (and prefers) returning a `haxe.macro.ComplexType` instead, which is the syntactic representation of a type. This is easier to work with in many cases because types can simply be referenced by their paths.

**Const type parameter** Haxe allows passing constant expression (5.2) as a type parameter if the type parameter name is `Const`. This can be utilized in the context of `@:genericBuild` macros to pass information from the syntax directly to the macro:

```

1 import haxe.macro.Expr;
2 import haxe.macro.Context;
3 import haxe.macro.Type;
4
5 class GenericBuildMacro2 {
6     static public function build() {
7         switch (Context.getLocalType()) {
8             case TInst(_, [TInst(_.get() => { kind: KExpr(macro $v{(s:String)
9                 }) }, _)]):
10                 trace(s);
11             case t:
12                 Context.error("Class expected", Context.currentPos());
13         }
14         return null;
15     }
16 }

```

```

1 @:genericBuild(GenericBuildMacro2.build())
2 class MyType<Const> { }
3
4 class Main {
5     static function main() {
6         var x:MyType<"myfile.txt">;
7     }
8 }

```

Here the macro logic could load a file and use its contents to generate a custom type.

## 9.6 Limitations

### 9.6.1 Macro-in-Macro

### 9.6.2 Static extension

The concepts of static extensions (6.3) and macros are somewhat conflicting: While the former requires a known type in order to determine used functions, macros execute before typing on plain syntax. It is thus not surprising that combining these two features can lead to issues. Haxe 3.0 would try to convert the typed expression back to a syntax expression, which is not always possible and may lose important information. We recommend that it is used with caution.

Since Haxe 3.1.0

The combination of static extensions and macros was reworked for the 3.1.0 release. The Haxe Compiler does not even try to find the original expression for the macro argument and instead passes a special `@:this this` expression. While the structure of this expression conveys no information, the expression can still be typed correctly:

```
1 import haxe.macro.Context;
2 import haxe.macro.Expr;
3
4 using Main;
5 using haxe.macro.Tools;
6
7 class Main {
8     static public function main() {
9         #if !macro
10             var a = "foo";
11             a.test();
12         #end
13     }
14
15     macro static function test(e:ExprOf<String>) {
16         trace(e.toString()); // @:this this
17         // TInst(String, [])
18         trace(Context.typeof(e));
19         return e;
20     }
21 }
```

### 9.6.3 Build Order

The build order of types is unspecified and this extends to the execution order of build-macros (9.5). While certain rules can be determined, we strongly recommend to not rely on the execution order of build-macros. If type building requires multiple passes, this should be reflected directly in the macro code. In order to avoid multiple build-macro execution on the same type, state can be stored in static variables or added as metadata (6.9) to the type in question:

```
1 import haxe.macro.Context;
2 import haxe.macro.Expr;
3
4 #if !macro
5 @:autoBuild(MyMacro.build())
```

```

6 #end
7 interface I1 { }
8
9 #if !macro
10 @:autoBuild(MyMacro.build())
11 #end
12 interface I2 { }
13
14 class C implements I1 implements I2 { }
15
16 class MyMacro {
17     macro static public function build():Array<Field> {
18         var c = Context.getLocalClass().get();
19         if (c.meta.has(":processed")) return null;
20         c.meta.add(":processed", [], c.pos);
21         // process here
22         return null;
23     }
24 }
25
26 class Main {
27     static public function main() { }
28 }

```

With both interfaces `I1` and `I2` having `:autoBuild` metadata, the build macro is executed twice for class `C`. We guard against duplicate processing by adding a custom `:processed` metadata to the class, which can be checked during the second macro execution.

## 9.6.4 Type Parameters

## 9.7 Initialization macros

Initialization macros are invoked from command line by using the `--macro callExpr(args)` command. This registers a callback which the compiler invokes after creating its context, but before typing what was argument to `-main`. This then allows configuring the compiler in some ways.

If the argument to `--macro` is a call to a plain identifier, that identifier is looked up in the class `haxe.macro.Compiler` which is part of the Haxe Standard Library. It comes with several useful initialization macros which are detailed in its [API](#).

As an example, the `include` macro allows inclusion of an entire package for compilation, recursively if necessary. The command line argument for this would then be `--macro include('some.pack true)`.

Of course it is also possible to define custom initialization macros to perform various tasks before the real compilation. A macro like this would then be invoked via `--macro some.Class.theMacro(a`. For instance, as all macros share the same context ([9.1](#)), an initialization macro could set the value of a static field which other macros use as configuration.

**Part III**

**Standard Library**

# Chapter 10

## Standard Library

### 10.1 String

#### Type: String

A String is a sequence of characters.

**Character code** Use the `.code` property on a constant single-char string in order to compile its ASCII character code:

```
1 "#".code // will compile as 35
```

See the [String API](#) for details about its methods.

### 10.2 Data Structures

#### 10.2.1 Array

An `Array` is a collection of elements. It has one type parameter (3.2) which corresponds to the type of these elements. Arrays can be created in three ways:

1. By using their constructor: `new Array()`
2. By using array declaration syntax (5.5): `[1, 2, 3]`
3. By using array comprehension (6.6): `[for (i in 0...10) if (i % 2 == 0) i]`

Arrays come with an [API](#) to cover most use-cases. Additionally they allow read and write array access (5.8):

```
1 class Main {
2   static public function main() {
3     var a = [1, 2, 3];
4     trace(a[1]); // 2
5     a[1] = 1;
6     trace(a[1]); // 1
7   }
8 }
```



Since array access in Haxe is unbounded, i.e. it is guaranteed to not throw an exception, this requires further discussion:

- If a read access is made on a non-existing index, a target-dependent value is returned.
- If a write access is made with a positive index which is out of bounds, `null` (or the default value (2.2) for basic types (2.1) on static targets (2.2)) is inserted at all positions between the last defined index and the newly written one.
- If a write access is made with a negative index, the result is unspecified.

Arrays define an iterator (6.7) over their elements. This iteration is typically optimized by the compiler to use a `while` loop (5.14) with array index:

```
1 class Main {
2   static public function main() {
3     var scores = [110, 170, 35];
4     var sum = 0;
5     for (score in scores) {
6       sum += score;
7     }
8     trace(sum); // 315
9   }
10 }
```

Haxe generates this optimized JavaScript output:

```
1 Main.main = function() {
2   var scores = [110,170,35];
3   var sum = 0;
4   var _g = 0;
5   while(_g < scores.length) {
6     var score = scores[_g];
7     ++_g;
8     sum += score;
9   }
10  console.log(sum);
11 };
```

Haxe does not allow arrays of mixed types unless the parameter type is forced to `Dynamic` (2.7):

```
1 class Main {
2   static public function main() {
3     // Compile Error: Arrays of mixed types are only allowed if the
4     // type is
5     // forced to Array<Dynamic>
6     //var myArray = [10, "Bob", false];
7
8     // Array<Dynamic> with mixed types
9     var myExplicitArray:Array<Dynamic> = [10, "Sally", true];
10  }
```

### Trivia: Dynamic Arrays

In Haxe 2, mixed type array declarations were allowed. In Haxe 3, arrays can have mixed types only if they are explicitly declared as `Array<Dynamic>`.

See the [Array API](#) for details about its methods.

## 10.2.2 Vector

A `Vector` is an optimized fixed-length *collection* of elements. Much like `Array` (10.2.1), it has one type parameter (3.2) and all elements of a vector must be of the specified type, it can be *iterated over* using a for loop (5.13) and accessed using array access syntax (2.8.3). However, unlike `Array` and `List`, vector length is specified on creation and cannot be changed later.

```
1 class Main {
2     static function main() {
3         var vec = new haxe.ds.Vector(10);
4
5         for (i in 0...vec.length) {
6             vec[i] = i;
7         }
8
9         trace(vec[0]); // 0
10        trace(vec[5]); // 5
11        trace(vec[9]); // 9
12    }
13 }
```

`haxe.ds.Vector` is implemented as an abstract type (2.8) over a native array implementation for given target and can be faster for fixed-size collections, because the memory for storing its elements is pre-allocated.

See the [Vector API](#) for details about the vector methods.

## 10.2.3 List

A `List` is a *collection* for storing elements. On the surface, a list is similar to an `Array` (Section 10.2.1). However, the underlying implementation is very different. This results in several functional differences:

1. A list can not be indexed using square brackets, i.e. `[0]`.
2. A list can not be initialized.
3. There are no list comprehensions.
4. A list can freely modify/add/remove elements while iterating over them.

A simple example for working with lists:

```
1 class Main {
2     static public function main() {
3         var myList = new List<Int>();
4         for (ii in 0...5)
5             myList.add(ii);
6     }
7 }
```

```

6     trace(myList); //{0, 1, 2, 3, 4}
7 }
8 }

```

See the [List API](#) for details about the list methods.

### 10.2.4 GenericStack

A `GenericStack`, like `Array` and `List` is a container for storing elements. It has one type parameter (3.2) and all elements of the stack must be of the specified type. Here is a small example program for initializing and working with a `GenericStack`.

```

1 import haxe.ds.GenericStack;
2
3 class Main {
4     static public function main() {
5         var myStack = new GenericStack<Int>();
6         for (ii in 0...5)
7             myStack.add(ii);
8         trace(myStack); //{4, 3, 2, 1, 0}
9         trace(myStack.pop()); //4
10    }
11 }

```

#### Trivia: FastList

In Haxe 2, the `GenericStack` class was known as `FastList`. Since its behavior more closely resembled a typical stack, the name was changed for Haxe 3.

The *Generic* in `GenericStack` is literal. It is attributed with the `:generic` metadata. Depending on the target, this can lead to improved performance on static targets. See [Generic \(Section 3.3\)](#) for more details.

See the [GenericStack API](#) for details about its methods.

### 10.2.5 Map

A `Map` is a container composed of *key, value* pairs. A `Map` is also commonly referred to as an associative array, dictionary, or symbol table. The following code gives a short example of working with maps:

```

1 class Main {
2     static public function main() {
3         // Maps are initialized like arrays, but
4         // use the map literal syntax with the
5         // '=>' operator. Maps can have their
6         // key value types defined explicitly
7         var map1:Map<Int, String> =
8             [1 => "one", 2=>"two"];
9
10        // Or they can infer their key value types
11        var map2 = [
12            "one"=>1,
13            "two"=>2,

```

```

14     "three"=>3
15 ];
16 $type(map2); // Map<String, Int>
17
18 // Keys must be unique
19 // Error: Duplicate Key
20 //var map3 = [1=>"dog", 1=>"cat"];
21
22 // Maps values can be accessed using array
23 // accessors "[]"
24 var map4 = ["M"=>"Monday", "T"=>"Tuesday"];
25 trace(map4["M"]); //Monday
26
27 // Maps iterate over their values by
28 // default
29 var valueSum;
30 for (value in map4) {
31     trace(value); // Monday \n Tuesday
32 }
33
34 // Can iterate over keys by using the
35 // keys() method
36 for (key in map4.keys()) {
37     trace(key); // M \n T
38 }
39
40 // Like arrays, a new Map can be made using
41 // comprehension
42 var map5 = [
43     for (key in map4.keys())
44         key => "FRIDAY!!"
45 ];
46 // {M => FRIDAY!!, T => FRIDAY!!}
47 trace(map5);
48 }
49 }

```

Under the hood, a Map is an abstract (2.8) type. At compile time, it gets converted to one of several specialized types depending on the *key* type:

- String: `haxe.ds.StringMap`
- Int: `haxe.ds.IntMap`
- EnumValue: `haxe.ds.EnumValueMap`
- {}: `haxe.ds.ObjectMap`

The Map type does not exist at runtime and has been replaced with one of the above objects. Map defines array access (2.8.3) using its key type. See the [Map API](#) for details of its methods.

### 10.2.6 Option

An [Option](#) is an enum (2.4) in the Haxe Standard Library which is defined like so:

```

1 enum Option<T> {
2     Some(v:T);
3     None;
4 }

```

It can be used in various situations, such as communicating whether or not a method had a valid return and if so, what value it returned:

```

1 import haxe.ds.Option;
2
3 class Main {
4     static public function main() {
5         var result = trySomething();
6         switch (result) {
7             case None:
8                 trace("Got None");
9             case Some(s):
10                 trace("Got a value: " +s);
11         }
12     }
13
14     static function trySomething():Option<String> {
15         if (Math.random() > 0.5) {
16             return None;
17         } else {
18             return Some("Success");
19         }
20     }
21 }

```

## 10.3 Regular Expressions

Haxe has built-in support for *regular expressions*<sup>1</sup>. They can be used to verify the format of a string, transform a string or extract some regular data from a given text.

Haxe has special syntax for creating regular expressions. We can create a regular expression object by typing it between the `~/` combination and a single `/` character:

```

1 var r = ~/haxe/i;

```

Alternatively, we can create regular expression with regular syntax:

```

1 var r = new EReg("haxe", "i");

```

First argument is a string with regular expression pattern, second one is a string with *flags* (see below).

We can use standard regular expression patterns such as:

- `.` any character
- `*` repeat zero-or-more
- `+` repeat one-or-more

---

<sup>1</sup>[http://en.wikipedia.org/wiki/Regular\\_expression](http://en.wikipedia.org/wiki/Regular_expression)

- `?` optional zero-or-one
- `[A-Z0-9]` character ranges
- `[\r\n\t]` character not-in-range
- `(...)` parenthesis to match groups of characters
- `^` beginning of the string (beginning of a line in multiline matching mode)
- `$` end of the string (end of a line in multiline matching mode)
- `|` "OR" statement.

For example, the following regular expression matches valid email addresses:

```
1 ~/ [A-Z0-9._\%-]+@[A-Z0-9.-]+\.[A-Z][A-Z][A-Z]?/i;
```

Please notice that the `i` at the end of the regular expression is a *flag* that enables case-insensitive matching.

The possible flags are the following:

- `i` case insensitive matching
- `g` global replace or split, see below
- `m` multiline matching, `^` and `$` represent the beginning and end of a line
- `s` the dot `.` will also match newlines (*Neko, C++, PHP, Flash and Java targets only*)
- `u` use UTF-8 matching (*Neko and C++ targets only*)

See the [EReg API](#) for details about its methods.

### 10.3.1 Matching

Probably one of the most common uses for regular expressions is checking whether a string matches the specific pattern. The `match` method of a regular expression object can be used to do that:

```
1 class Main {
2     static function main() {
3         var r = ~/world/;
4         var str = "hello world";
5         // true : 'world' was found in the string
6         trace(r.match(str));
7         trace(r.match("hello !")); // false
8     }
9 }
```

### 10.3.2 Groups

Specific information can be extracted from a matched string by using *groups*. If `match()` returns true, we can get groups using the `matched(X)` method, where `X` is the number of a group defined by regular expression pattern:

```

1 class Main {
2     static function main() {
3         var str = "Nicolas is 26 years old";
4         var r =
5             ~/[A-Za-z]+) is ([0-9]+) years old/;
6         r.match(str);
7         trace(r.matched(1)); // "Nicolas"
8         trace(r.matched(2)); // "26"
9     }
10 }

```

Note that group numbers start with 1 and `r.matched(0)` will always return the whole matched substring.

The `r.matchedPos()` will return the position of this substring in the original string:

```

1 class Main {
2     static function main() {
3         var str = "abcdeeeefghi";
4         var r = ~/e+/;
5         r.match(str);
6         trace(r.matched(0)); // "eeeeee"
7         // { pos : 4, len : 5 }
8         trace(r.matchedPos());
9     }
10 }

```

Additionally, `r.matchedLeft()` and `r.matchedRight()` can be used to get substrings to the left and to the right of the matched substring:

```

1 class Main {
2     static function main() {
3         var r = ~/b/;
4         r.match("abc");
5         trace(r.matchedLeft()); // a
6         trace(r.matched(0)); // b
7         trace(r.matchedRight()); // c
8     }
9 }

```

### 10.3.3 Replace

A regular expression can also be used to replace a part of the string:

```

1 class Main {
2     static function main() {
3         var str = "aaabcbcbcbz";
4         // g : replace all instances
5         var r = ~/b[^c]/g;
6         // "aaabcbcbcbxx"
7         trace(r.replace(str, "xx"));
8     }
9 }

```

We can use `$X` to reuse a matched group in the replacement:

```

1 class Main {
2     static function main() {
3         var str = "{hello} {0} {again}";
4         var r = ~/{([a-z]+)}/g;
5         // "*hello* {0} *again*"
6         trace(r.replace(str, "*$1*"));
7     }
8 }

```

### 10.3.4 Split

A regular expression can also be used to split a string into several substrings:

```

1 class Main {
2     static function main() {
3         var str = "XaaaYababZbbbW";
4         var r = ~/[ab]+/g;
5         // ["X", "Y", "Z", "W"]
6         trace(r.split(str));
7     }
8 }

```

### 10.3.5 Map

The `map` method of a regular expression object can be used to replace matched substrings using a custom function. This function takes a regular expression object as its first argument so we may use it to get additional information about the match being done and do conditional replacement. For example:

```

1 class Main {
2     static function main() {
3         var r = ~/ (dog|fox) /g;
4         var s = "The quick brown fox jumped over the lazy dog.";
5         var s2 = r.map(s, function(r) {
6             var match = r.matched(0);
7             switch (match) {
8                 case 'dog': return 'fox';
9                 case 'fox': return 'dog';
10                default: throw 'Unknown animal: $match';
11            };
12        });
13        trace(s2); // The quick brown dog jumped over the lazy fox.
14    }
15 }

```

### 10.3.6 Implementation Details

Regular Expressions are implemented:

- in JavaScript, the runtime is providing the implementation with the object `RegExp`.
- in Neko and C++, the PCRE library is used



- in Flash, PHP, C# and Java, native implementations are used
- in Flash 6/8, the implementation is not available

## 10.4 Math

Haxe includes a floating point math library for some common mathematical operations. Most of the functions operate on and return `Floats`. However, an `Int` can be used where a `Float` is expected, and Haxe also converts `Int` to `Float` during most numeric operations (see [Numeric Operators \(Section 2.1.3\)](#) for more details).

Here are some example uses of the math library:

```

1 class Main {
2     static public function main() {
3         var x = 1/2;
4         var y = 20.2;
5         var z = -2;
6
7         trace(Math.abs(z)); //2
8         trace(Math.sin(x*Math.PI)); //1
9         trace(Math.ceil(y)); //21
10
11         // log is the natural logarithm
12         trace(Math.log(Math.exp(5))); //5
13
14         // Output for neko target, may vary
15         // depending on platform
16         trace(1/0); //inf
17         trace(-1/0); //-inf
18         trace(Math.sqrt(-1)); //nan
19     }
20 }
```

See the [Math API](#) for all available functions.

### 10.4.1 Special Numbers

The math library has definitions for several special numbers:

- NaN (Not a Number): returned when a mathematically incorrect operation is executed, e.g. `Math.sqrt(-1)`
- `POSITIVE_INFINITY`: e.g. divide a positive number by zero
- `NEGATIVE_INFINITY`: e.g. divide a negative number by zero
- `PI` : 3.1415...

### 10.4.2 Mathematical Errors

Although neko can fluidly handle mathematical errors, like division by zero, this is not true for all targets. Depending on the target, mathematical errors may produce exceptions and ultimately errors.

### 10.4.3 Integer Math

If you are targeting a platform that can utilize integer operations, e.g. integer division, it should be wrapped in `Std.int()` for improved performance. The Haxe Compiler can then optimize for integer operations. An example:

```
1 var intDivision = Std.int(6.2/4.7);
```

I think C++ can use integer operations, but I don't know about any other targets. Only saw this mentioned in an old discussion thread, still true?

### 10.4.4 Extensions

It is common to see [Static Extension](#) ([Section 6.3](#)) used with the math library. This code shows a simple example:

```
1 class MathStaticExtension {
2     /* Converts an angle in radians to degrees */
3     inline public static function toDegrees(radians:Float):Float {
4         return radians * 180 / Math.PI;
5     }
6 }
```

```
1 using MathStaticExtension;
2
3 class Main {
4     public static function main() {
5         var ang = 1/2*Math.PI;
6         trace(ang.toDegrees()); //90
7     }
8 }
```

## 10.5 Lambda

### Definition: Lambda

Lambda is a functional language concept within Haxe that allows you to apply a function to a list or iterators ([6.7](#)). The Lambda class is a collection of functional methods in order to use functional-style programming with Haxe.

It is ideally used with `using Lambda` (see [Static Extension \(6.3\)](#)) and then acts as an extension to `Iterable` types.

On static platforms, working with the `Iterable` structure might be slower than performing the operations directly on known types, such as `Array` and `List`.

**Lambda Functions** The Lambda class allows us to operate on an entire `Iterable` at once. This is often preferable to looping routines since it is less error prone and easier to read. For convenience, the `Array` and `List` class contains some of the frequently used methods from the Lambda class.

It is helpful to look at an example. The `exists` function is specified as:

```
1 static function exists<A>( it : Iterable<A>, f : A -> Bool ) : Bool
```

Most Lambda functions are called in similar ways. The first argument for all of the Lambda functions is the `Iterable` on which to operate. Many also take a function as an argument.

**Lambda.array, Lambda.list** Convert `Iterable` to `Array` or `List`. It always returns a new instance.

**Lambda.count** Count the number of elements. If the `Iterable` is a `Array` or `List` it is faster to use its `length` property.

**Lambda.empty** Determine if the `Iterable` is empty. For all `Iterables` it is best to use the `this` function; it's also faster than compare the `length` (or result of `Lambda.count`) to zero.

**Lambda.has** Determine if the specified element is in the `Iterable`.

**Lambda.exists** Determine if criteria is satisfied by an element.

**Lambda.indexOf** Find out the index of the specified element.

**Lambda.find** Find first element of given search function.

**Lambda.foreach** Determine if every element satisfies a criteria.

**Lambda.iter** Call a function for each element.

**Lambda.concat** Merge two `Iterables`, returning a new `List`.

**Lambda.filter** Find the elements that satisfy a criteria, returning a new `List`.

**Lambda.map, Lambda.mapI** Apply a conversion to each element, returning a new `List`.

**Lambda.fold** Functional fold, which is also known as `reduce`, `accumulate`, `compress` or `inject`.

This example demonstrates the `Lambda.filter` and `map` on a set of strings:

```
1 using Lambda;
2 class Main {
3     static function main() {
4         var words = ['car', 'boat', 'cat', 'frog'];
5
6         var isThreeLetters = function(word) return word.length == 3;
7         var capitalize = function(word) return word.toUpperCase();
8
9         // Three letter words and capitalized.
10        trace(words.filter(isThreeLetters).map(capitalize)); // [CAR,
11                       CAT]
12    }
13 }
```

This example demonstrates the `Lambda.count`, `has`, `foreach` and `fold` function on a set of ints.

```
1 using Lambda;
2 class Main {
3     static function main() {
4         var numbers = [1, 3, 5, 6, 7, 8];
5
6         trace(numbers.count()); // 6
7         trace(numbers.has(4)); // false
8     }
9 }
```

```

9      // test if all numbers are greater/smaller than 20
10     trace(numbers.foreach(function(v) return v < 20)); // true
11     trace(numbers.foreach(function(v) return v > 20)); // false
12
13     // sum all the numbers
14     var sum = function(num, total) return total += num;
15     trace(numbers.fold(sum, 0)); // 30
16 }
17 }

```

See the [Lambda API](#) for all available functions.

## 10.6 Template

Haxe comes with a standard template system with an easy to use syntax which is interpreted by a lightweight class called [haxe.Template](#).

A template is a string or a file that is used to produce any kind of string output depending on the input. Here is a small template example:

```

1 class Main {
2     static function main() {
3         var sample = "My name is <strong>::name::</strong>, <em>::age::</em> years old";
4         var user = {name:"Mark", age:30};
5         var template = new haxe.Template(sample);
6         var output = template.execute(user);
7         trace(output);
8     }
9 }

```

The console will trace `My name is Mark, 30 years old`.

**Expressions** An expression can be put between the `::`, the syntax allows the current possibilities:

- `::name::` the variable name
- `::expr.field::` field access
- `::(expr)::` the expression `expr` is evaluated
- `::(e1 op e2)::` the operation `op` is applied to `e1` and `e2`
- `::(135)::` the integer 135. Float constants are not allowed

**Conditions** It is possible to test conditions using `::if flag1::`. Optionally, the condition may be followed by `::elseif flag2::` or `::else::`. Close the condition with `::end::`.

```

1 ::if isValid:: valid ::else:: invalid ::end::

```

Operators can be used but they don't deal with operator precedence. Therefore it is required to enclose each operation in parentheses `()`. Currently, the following operators are allowed: `+`, `-`, `*`, `/`, `>`, `<`, `>=`, `<=`, `==`, `!=`, `&&` and `||`.

For example `::((1 + 3) == (2 + 2))::` will display true.

```
1 ::if (points == 10):: Great! ::end::
```

To compare to a string, use double quotes " in the template.

```
1 ::if (name == "Mark"):: Hi Mark ::end::
```

**Iterating** Iterate on a structure by using `::foreach::`. End the loop with `::end::`.

```
1 <table>
2   <tr>
3     <th>Name</th>
4     <th>Age</th>
5   </tr>
6   ::foreach users::
7     <tr>
8       <td>::name::</td>
9       <td>::age::</td>
10    </tr>
11  ::end::
12 </table>
```

**Sub-templates** To include templates in other templates, pass the sub-template result string as a parameter.

```
1 var users = [{name:"Mark", age:30}, {name:"John", age:45}];
2
3 var userTemplate = new haxe.Template("::foreach users:: ::name::(:age
4   ::) ::end::");
5
6 var userOutput = userTemplate.execute({users: users});
7
8 var template = new haxe.Template("The users are ::users::");
9 var output = template.execute({users: userOutput});
10 trace(output);
```

The console will trace `The users are Mark(30) John(45)`.

**Template macros** To call custom functions while parts of the template are being rendered, provide a `macros` object to the argument of `Template.execute`. The key will act as the template variable name, the value refers to a callback function that should return a `String`. The first argument of this macro function is always a `resolve()` method, followed by the given arguments. The resolve function can be called to retrieve values from the template context. If `macros` has no such field, the result is unspecified.

The following example passes itself as macro function context and executes `display` from the template.

```
1 class Main {
2   static function main() {
3     new Main();
4   }
5
6   public function new() {
7     var user = {name:"Mark", distance:3500};
8     var sample = "The results: $$display(::user::,::time::)";
```

```

9     var template = new haxe.Template(sample);
10    var output = template.execute({user:user, time: 15}, this);
11    trace(output);
12  }
13
14  function display(resolve:String->Dynamic, user:User, time:Int) {
15    return user.name + " ran " + (user.distance/1000) + " kilometers
        in " + time + " minutes";
16  }
17 }
18 typedef User = {name:String, distance:Int}

```

The console will trace The results: Mark ran 3.5 kilometers in 15 minutes.

**Globals** Use the `Template.globals` object to store values that should be applied across all `haxe.Template` instances. This has lower priority than the context argument of `Template.execute`.

**Using resources** To separate the content from the code, consider using the resource embedding system (8.4). Place the template-content in a new file called `sample.mtt`, add `-resource sample.mtt@my_sample` to the compiler arguments and retrieve the content using `haxe.Resource.getString`.

```

1 class Main {
2   static function main() {
3     var sample = haxe.Resource.getString("my_sample");
4     var user = {name:"Mark", age:30};
5     var template = new haxe.Template(sample);
6     var output = template.execute(user);
7     trace(output);
8   }
9 }

```

When running the template system on the server side, you can simply use `neko.Lib.print` or `php.Lib.print` instead of `trace` to display the HTML template to the user.

See the [Template API](#) for details about its methods.

## 10.7 Reflection

Haxe supports runtime reflection of types and fields. Special care has to be taken here because runtime representation generally varies between targets. In order to use reflection correctly it is necessary to understand what kind of operations are supported and what is not. Given the dynamic nature of reflection, this can not always be determined at compile-time.

The reflection API consists of two classes:

**Reflect:** A lightweight API which work best on anonymous structures (2.5), with limited support for classes (2.3).

**Type:** A more robust API for working with classes and enums (2.4).

The available methods are detailed in the API for [Reflect](#) and [Type](#).

Reflection can be a powerful tool, but it is important to understand why it can also cause problems. As an example, several functions expect a `String` (10.1) argument and try to resolve it to a type or field. This is vulnerable to typing errors:

```

1 class Main {
2   static function main() {
3     trace(Type.resolveClass("Mian")); // null
4   }
5 }

```

However, even if there are no typing errors it is easy to come across unexpected behavior:

```

1 class Main {
2   static function main() {
3     // null
4     trace(Type.resolveClass("haxe.Template"));
5   }
6 }

```

The problem here is that the compiler never actually “sees” the type `haxe.Template`, so it does not compile it into the output. Furthermore, even if it were to see the type there could be issues arising from dead code elimination (8.2) eliminating types or fields which are only used via reflection.

Another set of problems comes from the fact that, by design, several reflection functions expect arguments of type `Dynamic` (2.7), meaning the compiler cannot check if the passed in arguments are correct. The following example demonstrates a common mistake when working with `callMethod`:

```

1 class Main {
2   static function main() {
3     // wrong
4     // Reflect.callMethod(Main, "f", []);
5     // right
6     Reflect.callMethod(Main,
7       Reflect.field(Main, "f"), []);
8   }
9
10  static function f() {
11    trace('Called');
12  }
13 }

```

The commented out call would be accepted by the compiler because it assigns the string `"f"` to the function argument `func` which is specified to be `Dynamic`.

A good advice when working with reflection is to wrap it in a few functions within an application or API which are called by otherwise type-safe code. An example could look like this:

```

1 typedef MyStructure = {
2   name: String,
3   score: Int
4 }
5
6 class Main {
7   static function main() {
8     var data = reflective();
9     // At this point data is nicely typed as MyStructure
10  }
11
12  static function reflective():MyStructure {

```

```

13     // Work with reflection here to get some values we want to return.
14     return {
15         name: "Reflection",
16         score: 0
17     }
18 }
19 }

```

While the method `reflective` could internally work with reflection (and `Dynamic` for that matter) a lot, its return value is a typed structure which the callers can use in a type-safe manner.

## 10.8 Serialization

Many runtime values can be serialized and deserialized using the `haxe.Serializer` and `haxe.Unserializer` classes. Both support two usages:

1. Create an instance and continuously call the `serialize/unserialize` method to handle multiple values.
2. Call their static `run` method to serialize/deserialize a single value.

The following example demonstrates the first usage:

```

1 import haxe.Serializer;
2 import haxe.Unserializer;
3
4 class Main {
5     static function main() {
6         var serializer = new Serializer();
7         serializer.serialize("foo");
8         serializer.serialize(12);
9         var s = serializer.toString();
10        trace(s); // y3:foo12
11
12        var unserializer = new Unserializer(s);
13        trace(unserializer.unserialize()); // foo
14        trace(unserializer.unserialize()); // 12
15    }
16 }

```

The result of the serialization (here stored in local variable `s`) is a String (10.1) and can be passed around at will, even remotely. Its format is described in [Serialization format](#) (Section 10.8.1).

### Supported values

- `null`
- `Bool`, `Int` and `Float` (including infinities and NaN)
- `String`
- `Date`
- `haxe.io.Bytes` (encoded as base64)



- Array (10.2.1) and List (10.2.3)
- `haxe.ds.StringMap`, `haxe.ds.IntMap` and `haxe.ds.ObjectMap`
- anonymous structures (2.5)
- Haxe class instances (2.3) (not native ones)
- enum instances (2.4)

**Serialization configuration** Serialization can be configured in two ways. For both a static variable can be set to influence all `haxe.Serializer` instances, and a member variable can be set to only influence a specific instance:

**USE\_CACHE, useCache:** If true, repeated structures or class/enum instances are serialized by reference. This can avoid infinite loops for recursive data at the expense of longer serialization time. By default, object caching is disabled; strings however are always cached.

**USE\_ENUM\_INDEX, useEnumIndex:** If true, enum constructors are serialized by their index instead of their name. This can make the resulting string shorter, but breaks if enum constructors are inserted into the type before deserialization. This behavior is disabled by default.

**Deserialization behavior** If the serialization result is stored and later used for deserialization, care has to be taken to maintain compatibility when working with class and enum instances. It is then important to understand exactly how unserialization is implemented.

- The type has to be available in the runtime where the deserialization is made. If dead code elimination (8.2) is active, a type which is used only through serialization might be removed.
- Each `Unserializer` has a member variable `resolver` which is used to resolve classes and enums by name. Upon creation of the `Unserializer` this is set to `Unserializer.DEFAULT_RESOLVER`. Both that and the instance member can be set to a custom resolver.
- Classes are resolved by name using `resolver.resolveClass(name)`. The instance is then created using `Type.createEmptyInstance`, which means that the class constructor is not called. Finally, the instance fields are set according to the serialized value.
- Enums are resolved by name using `resolver.resolveEnum(name)`. The enum instance is then created using `Type.createEnum`, using the serialized argument values if available. If the constructor arguments were changed since serialization, the result is unspecified.

**Custom (de)serialization** If a class defines the member method `hxSerialize`, that method is called by the serializer and allows custom serialization of the class. Likewise, if a class defines the member method `hxUnserialize` it is called by the deserializer:

```
1 import haxe.Serializer;
2 import haxe.Unserializer;
3
4 class Main {
5
6     var x:Int;
7     var y:Int;
8 }
```

```

9  static function main() {
10     var s = Serializer.run(new Main(1, 2));
11     var c:Main = Unserializer.run(s);
12     trace(c.x); // 1
13     trace(c.y); // -1
14 }
15
16 function new(x, y) {
17     this.x = x;
18     this.y = y;
19 }
20
21 @:keep
22 function hxSerialize(s:Serializer) {
23     s.serialize(x);
24 }
25
26 @:keep
27 function hxUnserialize(u:Unserializer) {
28     x = u.unserialize();
29     y = -1;
30 }
31 }

```

In this example we decide that we want to ignore the value of member variable `y` and do not serialize it. Instead we default it to `-1` in `hxUnserialize`. Both methods are annotated with the `@:keep` metadata to prevent dead code elimination (8.2) from removing them as they are never properly referenced in the code.

See [Serializer](#) and [Unserializer](#) API documentation for details.

### 10.8.1 Serialization format

Each supported value is translated to a distinct prefix character, followed by the necessary data.

**null:** `n`

**Int:** `z` for zero, or `i` followed by the integer display (e.g. `i456`)

**Float:**

**NaN:** `k`

**negative infinity:** `m`

**positive infinity:** `p`

**finite floats:** `d` followed by the float display (e.g. `d1.45e-8`)

**Bool:** `t` for true, `f` for false

**String:** `y` followed by the url encoded string length, then `:` and the url encoded string (e.g. `y10:hi%20there` for "hi there").

**name-value pairs:** a serialized string representing the name followed by the serialized value

**structure:** `o` followed by the list of name-value pairs and terminated by `g` (e.g. `oy1:xi2y1:kng` for `{x:2, k:null}`)

**List:** l followed by the list of serialized items, followed by h (e.g. lnnh for a list of two null values)

**Array:** a followed by the list of serialized items, followed by h. For multiple consecutive null values, u followed by the number of null values is used (e.g. aili2u4i7ni9h for [1, 2, null, null, null, null, 7, null, 9])

**Date:** v followed by the date itself (e.g. v2010-01-01 12:45:10)

**haxe.ds.StringMap:** b followed by the name-value pairs, followed by h (e.g. by1:xi2y1:knh for {"x" => 2, "k" => null})

**haxe.ds.IntMap:** q followed by the key-value pairs, followed by h. Each key is represented as :<int> (e.g. q:4n:5i45:6i7h for {4 => null, 5 => 45, 6 => 7})

**haxe.ds.ObjectMap:** M followed by serialized value pairs representing the key and value, followed by h

**haxe.io.Bytes:** s followed by the length of the base64 encoded bytes, then : and the byte representation using the codes A-Za-z0-9% (e.g. s3:AAA for 2 bytes equal to 0, and s10:SGVsbG8gIQ for haxe.io.Bytes.ofString("Hello !"))

**exception:** x followed by the exception value

**class instance:** c followed by the serialized class name, followed by the name-value pairs of the fields, followed by g (e.g. cy5:Pointy1:xzy1:yzg for new Point(0, 0) (having two integer fields x and y)

**enum instance (by name):** w followed by the serialized enum name, followed by the serialized constructor name, followed by :, followed by the number of arguments, followed by the argument values (e.g. wy3:Fooy1:A:0 for Foo.A (with no arguments), wy3:Fooy1:B:2i4n for Foo.B(4, null))

**enum instance (by index):** j followed by the serialized enum name, followed by :, followed by the constructor index (starting from 0), followed by :, followed by the number of arguments, followed by the argument values (e.g. jy3:Foo:0:0 for Foo.A (with no arguments), jy3:Foo:1:2i4n for Foo.B(4, null))

**cache references:**

**String:** R followed by the corresponding index in the string cache (e.g. R456)

**class, enum or structure** r followed by the corresponding index in the object cache (e.g. r42)

**custom:** C followed by the class name, followed by the custom serialized data, followed by g

Cached elements and enum constructors are indexed from zero.

## 10.9 Xml

Haxe provides built-in support for working with XML<sup>2</sup> data via the `haxe.Xml` class.

---

<sup>2</sup><http://en.wikipedia.org/wiki/XML>

## 10.9.1 Getting started with Xml

**Creating a root element** A Xml root element can be created using the `Xml.createElement` method.

```
1 var root = Xml.createElement('root');
2 trace(root); // <root />
```

An root node element can also be created by parsing a String containing the XML data.

```
1 var root = Xml.parse('<root />').firstElement();
2 trace(root); // <root />
```

**Creating child elements** Adding child elements to the root can be done using the `addChild` method.

```
1 var child:Xml = Xml.createElement('child');
2 root.addChild(child);
3 trace(root); // <root><child/></root>
```

Adding attributes to an element can be done by using the `set()` method.

```
1 child.set('name', 'John');
2 trace(root); // <root><child name="John"/></root>
```

**Accessing elements and values** This code parses an XML string into an object structure `Xml` and then accesses properties of the object.

```
1 var xmlString = '<hello name="world!">Haxe is great!</hello>';
2 var xml:Xml = Xml.parse(xmlString).firstElement();
3
4 trace(xml.nodeName); // hello
5 trace(xml.get('name')); // world!
6 trace(xml.firstChild().nodeValue); // Haxe is great!
```

The difference between `firstChild` and `firstElement` is that the second function will return the first child with the type `Xml.Element`.

**Iterate on Xml elements** We can as well use other methods to iterate either over children or elements.

```
1 for (child in xml) {
2     // iterate on all children.
3 }
4 for (elt in xml.elements()) {
5     // iterate on all elements.
6 }
7 for (user in xml.elementsNamed("user")) {
8     // iterate on all elements with a nodeName "user".
9 }
10 for (att in xml.attributes()) {
11     // iterator on all attributes.
12 }
```

See [Xml](#) API documentation for details about its methods.

## 10.9.2 Parsing Xml

The static method `Xml.parse` can be used to parse XML data and obtain a Haxe value from it.

```
1 var xml = Xml.parse('<root>Haxe is great!</root>').firstElement();
2 trace(xml.firstChild().nodeValue);
```

## 10.9.3 Encoding Xml

The method `xml.toString()` can be used to obtain the String representation.

```
1 var xml = Xml.createElement('root');
2 xml.addChild(Xml.createElement('child1'));
3 xml.addChild(Xml.createElement('child2'));
4
5 trace(xml.toString()); // <root><child1/><child2/></root>
```

## 10.10 Json

Haxe provides built-in support for (de-)serializing JSON<sup>3</sup> data via the `haxe.Json` class.

### 10.10.1 Parsing JSON

Use the `haxe.Json.parse` static method to parse JSON data and obtain a Haxe value from it:

```
1 class Main {
2     static function main() {
3         var s = '{"rating": 5}';
4         var o = haxe.Json.parse(s);
5         trace(o); // { rating: 5 }
6     }
7 }
```

Note that the type of the object returned by `haxe.Json.parse` is `Dynamic`, so if the structure of our data is well-known, we may want to specify a type using anonymous structures (2.5). This way we provide compile-time checks for accessing our data and most likely more optimal code generation, because compiler knows about types in a structure:

```
1 typedef MyData = {
2     var name:String;
3     var tags:Array<String>;
4 }
5
6 class Main {
7     static function main() {
8         var s = '{
9             "name": "Haxe",
10            "tags": ["awesome"]
11        }';
12         var o:MyData = haxe.Json.parse(s);
13         trace(o.name); // Haxe (a string)
```

---

<sup>3</sup><http://en.wikipedia.org/wiki/JSON>

```

14     // awesome (a string in an array)
15     trace(o.tags[0]);
16 }
17 }

```

### 10.10.2 Encoding JSON

Use the `haxe.Json.stringify` static method to encode a Haxe value into a *JSON* string:

```

1 class Main {
2     static function main() {
3         var o = {rating: 5};
4         var s = haxe.Json.stringify(o);
5         trace(s); // {"rating":5}
6     }
7 }

```

### 10.10.3 Implementation details

The `haxe.Json` API automatically uses native implementation on targets where it is available, i.e. *JavaScript*, *Flash* and *PHP* and provides its own implementation for other targets.

Usage of Haxe own implementation can be forced with `-D haxeJSON` compiler argument. This will also provide serialization of enums (2.4) by their index, maps (10.2.5) with string keys and class instances.

Older browsers (Internet Explorer 7, for instance) may not have native *JSON* implementation. In case it's required to support them, we can include one of the *JSON* implementations available on the internet in the HTML page. Alternatively, a `-D old_browser` compiler argument that will make `haxe.Json` try to use native *JSON* and, in case it's not available, fallback to its own implementation.

## 10.11 Input/Output

### 10.12 Sys/sys

### 10.13 Remoting

Haxe remoting is a way to communicate between different platforms. With Haxe remoting, applications can transmit data transparently, send data and call methods between server and client side.

See the [remoting package](#) on the API documentation for more details on its classes.

#### 10.13.1 Remoting Connection

In order to use remoting, there must be a connection established. There are two kinds of Haxe Remoting connections:

**`haxe.remoting.Connection`** is used for *synchronous connections*, where the results can be directly obtained when calling a method.

**`haxe.remoting.AsyncConnection`** is used for *asynchronous connections*, where the results are events that will happen later in the execution process.

**Start a connection** There are some target-specific constructors with different purposes that can be used to set up a connection:

**All targets: `HttpAsyncConnection.urlConnect(url:String)`** Returns an asynchronous connection to the given URL which should link to a Haxe server application.

**Flash: `ExternalConnection.jsConnect(name:String, ctx:Context)`** Allows a connection to the local JavaScript Haxe code. The JS Haxe code must be compiled with the class `ExternalConnection` included. This only works with Flash Player 8 and higher.

**`AMFConnection.urlConnect(url:String)` and `AMFConnection.connect(cnx : NetConn)`** Allows a connection to an **AMF Remoting server** such as **Flash Media Server** or **AMF-PHP**.

**`SocketConnection.create(sock:flash.XMLSocket)`** Allows remoting communications over an `XMLSocket`

**`LocalConnection.connect(name:String)`** Allows remoting communications over a **Flash LocalConnection**

**JavaScript: `ExternalConnection.flashConnect(name:String, obj:String, ctx:Context)`** Allows a connection to a given Flash Object. The Haxe Flash content must be loaded and it must include the `haxe.remoting.Connection` class. This only works with Flash 8 and higher.

**Neko: `HttpConnection.urlConnect(url:String)`** Will work like the asynchronous version but in synchronous mode.

**`SocketConnection.create(...)`** Allows real-time communications with a Flash client which is using an `XMLSocket` to connect to the server.

**Remoting context** Before communicating between platforms, a remoting context has to be defined. This is a shared API that can be called on the connection at the client code.

This server code example creates and shares an API:

```
1 class Server {
2     function new() { }
3     function foo(x, y) { return x + y; }
4
5     static function main() {
6         var ctx = new haxe.remoting.Context();
7         ctx.addObject("Server", new Server());
8
9         if(haxe.remoting.HttpConnection.handleRequest(ctx))
10            {
11                return;
12            }
13
14            // handle normal request
15            trace("This is a remoting server !");
16        }
17    }
```

**Using the connection** Using a connection is pretty convenient. Once the connection is obtained, use classic dot-access to evaluate a path and then use `call()` to call the method in the remoting context and get the result. The asynchronous connection takes an additional function parameter that will be called when the result is available.

This client code example connects to the server remoting context and calls a function `foo()` on its API.

```
1 class Client {
2   static function main() {
3     var cnx = haxe.remoting.HttpAsyncConnection.urlConnect("http://
4       localhost/");
5     cnx.setErrorHandler( function(err) trace('Error: $err'); } );
6     cnx.Server.foo.call([1,2], function(data) trace('Result: $data');)
7   }
8 }
```

To make this work for the Neko target, setup a Neko Web Server, point the url in the Client to "http://localhost2000/remoting.n" and compile the Server using `-main Server -neko remoting.n`.

### Error handling

- When an error occurs in a asynchronous call, the error handler is called as seen in the example above.
- When an error occurs in a synchronous call, an exception is raised on the caller-side as if we were calling a local method.

**Data serialization** Haxe Remoting can send a lot of different kinds of data. See [Serialization \(10.8\)](#).

See the [remoting package](#) on the API documentation for more details on its classes.

## 10.13.2 Implementation details

**JavaScript security specifics** The html-page wrapping the js client must be served from the same domain as the one where the server is running. The same-origin policy restricts how a document or script loaded from one origin can interact with a resource from another origin. The same-origin policy is used as a means to prevent some of the cross-site request forgery attacks.

To use the remoting across domain boundaries, CORS (cross-origin resource sharing) needs to be enabled by defining the header `X-Haxe-Remoting` in the `.htaccess`:

```
1 # Enable CORS
2 Header set Access-Control-Allow-Origin "*"
3 Header set Access-Control-Allow-Methods: "GET, POST, OPTIONS, DELETE, PUT"
4 Header set Access-Control-Allow-Headers: X-Haxe-Remoting
```

See [same-origin policy](#) for more information on this topic.

Also note that this means that the page can't be served directly from the file system "file:///C:/example1

**Flash security specifics** When Flash accesses a server from a different domain, set up a `crossdomain.xml` file on the server, enabling the `X-Haxe` headers.



```

1 <cross-domain-policy>
2   <allow-access-from domain="*" /> <!-- or the appropriate domains
   -->
3   <allow-http-request-headers-from domain="*" headers="X-Haxe*" />
4 </cross-domain-policy>

```

**Arguments types are not ensured** There is no guarantee of any kind that the arguments types will be respected when a method is called using remoting. That means even if the arguments of function `foo` are typed to `Int`, the client will still be able to use strings while calling the method. This can lead to security issues in some cases. When in doubt, check the argument type when the function is called by using the `Std.is` method.

## 10.14 Unit testing

The Haxe Standard Library provides basic unit testing classes from the `haxe.unit` package.

**Creating new test cases** First, create a new class extending `haxe.unit.TestCase` and add own test methods. Every test method name must start with "test".

```

1 class MyTestCase extends haxe.unit.TestCase {
2   public function testBasic() {
3     assertEquals("A", "A");
4   }
5 }

```

**Running unit tests** To run the test, an instance of `haxe.unit.TestRunner` has to be created. Add the `TestCase` using the `add` method and call `run` to start the test.

```

1 class Main {
2   static function main() {
3     var r = new haxe.unit.TestRunner();
4     r.add(new MyTestCase());
5     // add other TestCases here
6
7     // finally, run the tests
8     r.run();
9   }
10 }

```

The result of the test looks like this:

```

1 Class: MyTestCase
2 .
3 OK 1 tests, 0 failed, 1 success

```

**Test functions** The `haxe.unit.TestCase` class comes with three test functions.

**assertEquals(a, b)** Succeeds if `a` and `b` are equal, where `a` is value tested and `b` is the expected value.

**assertTrue(a)** Succeeds if `a` is `true`

**assertFalse(a)** Succeeds if `a` is `false`

**Setup and tear down** To run code before or after the test, override the functions `setup` and `tearDown` in the `TestCase`.

**setup** is called before each test runs.

**tearDown** is called once after all tests are run.

```
1 class MyTestCase extends haxe.unit.TestCase {
2     var value:String;
3
4     override public function setup() {
5         value = "foo";
6     }
7
8     public function testSetup() {
9         assertEquals("foo", value);
10    }
11 }
```

**Comparing Complex Objects** With complex objects it can be difficult to generate expected values to compare to the actual ones. It can also be a problem that `assertEquals` doesn't do a deep comparison. One way around these issues is to use a string as the expected value and compare it to the actual value converted to a string using `Std.string`. Below is a trivial example using an array.

```
1 public function testArray() {
2     var actual = [1,2,3];
3     assertEquals("[1, 2, 3]", Std.string(actual));
4 }
```

See the [haxe.unit](#) package on the API documentation for more details.

**Part IV**

**Miscellaneous**

# Chapter 11

## Haxelib

Haxelib is the library manager that comes with any Haxe distribution. Connected to a central repository, it allows submitting and retrieving libraries and has multiple features beyond that. Available libraries can be found at <http://lib.haxe.org>.

A basic Haxe library is a collection of `.hx` files. That is, libraries are distributed by source code by default, making it easy to inspect and modify their behavior. Each library is identified by a unique name, which is utilized when telling the Haxe Compiler which libraries to use for a given compilation.

### 11.1 Using a Haxe library with the Haxe Compiler

Any installed Haxe library can be made available to the compiler through the `-lib <library-name>` argument. This is very similar to the `-cp <path>` argument, but expects a library name instead of a directory path. These commands are explained thoroughly in [Compiler Usage \(Chapter 7\)](#).

For our exemplary usage we chose a very simple Haxe library called “random”. It provides a set of static convenience methods to achieve various random effects, such as picking a random element from an array.

```
1 class Main {
2     static public function main() {
3         var elt = Random.fromArray([1, 2, 3]);
4         trace(elt);
5     }
6 }
```

Compiling this without any `-lib` argument causes an error message along the lines of `Unknown identifier : Random`. This shows that installed Haxe libraries are not available to the compiler by default unless they are explicitly added. A working command line for above program is `haxe -lib random -main Main --interp`.

If the compiler emits an error `Error: Library random is not installed : run 'haxelib install random' the library has to be installed via the haxelib command first`. As the error message suggests, this is achieved through `haxelib install random`. We will learn more about the `haxelib` command in [Using Haxelib \(Section 11.4\)](#).

### 11.2 haxelib.json

Each Haxe library requires a `haxelib.json` file in which the following attributes are defined:

**name:** The name of the library. It must contain at least 3 characters among the following:

$A - Za - z0 - 9_.$

. In particular, no spaces are allowed.

**url:** The URL of the library, i.e. where more information can be found.

**license:** The license under which the library is released. Can be GPL, LGPL, BSD, Public (for Public Domain) or MIT.

**tags:** An array of tag-strings which are used on the repository website to sort libraries.

**description:** The description of what the library is doing.

**version:** The version string of the library. This is detailed in [Versioning \(Section 11.2.1\)](#).

**classPath:** The path string to the source files.

**releasenote:** The release notes of the current version.

**contributors:** An array of user names which identify contributors to the library.

**dependencies:** An object describing the dependencies of the library. This is detailed in [Dependencies \(Section 11.2.2\)](#).

The following JSON is a simple example of a haxelib.json:

```
1 {
2   "name": "useless_lib",
3   "url" : "https://github.com/jasononeil/useless/",
4   "license": "MIT",
5   "tags": ["cross", "useless"],
6   "description": "This library is useless in the same way on every
7     platform.",
8   "version": "1.0.0",
9   "releasenote": "Initial release, everything is working correctly.",
10  "contributors": ["Juraj", "Jason", "Nicolas"],
11  "dependencies": {
12    "tink_macro": "",
13    "nme": "3.5.5"
14  }
```

### 11.2.1 Versioning

Haxelib uses a simplified version of [SemVer](#). The basic format is this:

```
1 major.minor.patch
```

These are the basic rules:

- Major versions are incremented when you break backwards compatibility - so old code will not work with the new version of the library.
- Minor versions are incremented when new features are added.

- Patch versions are for small fixes that do not change the public API, so no existing code should break.
- When a minor version increments, the patch number is reset to 0. When a major version increments, both the minor and patch are reset to 0.

Examples:

**"0.0.1"**: A first release. Anything with a "0" for the major version is subject to change in the next release - no promises about API stability!

**"0.1.0"**: Added a new feature! Increment the minor version, reset the patch version

**"0.1.1"**: Realised the new feature was broken. Fixed it now, so increment the patch version

**"1.0.0"**: New major version, so increment the major version, reset the minor and patch versions. You promise your users not to break this API until you bump to 2.0.0

**"1.0.1"**: A minor fix

**"1.1.0"**: A new feature

**"1.2.0"**: Another new feature

**"2.0.0"**: A new version, which might break compatibility with 1.0. Users are to upgrade cautiously.

If this release is a preview (Alpha, Beta or Release Candidate), you can also include that, with an optional release number:

```
1 major.minor.patch-(alpha/beta/rc).release
```

Examples:

**"1.0.0-alpha"**: The alpha of 1.0.0 - use with care, things are changing!

**"1.0.0-alpha.2"**: The 2nd alpha

**"1.0.0-beta"**: Beta - things are settling down, but still subject to change.

**"1.0.0-rc.1"**: The 1st release candidate for 1.0.0 - you shouldn't be adding any more features now

**"1.0.0-rc.2"**: The 2nd release candidate for 1.0.0

**"1.0.0"**: The final release!

## 11.2.2 Dependencies

As of Haxe 3.1.0, haxelib supports only exact version matching for dependencies. Dependencies are defined as part of the haxelib.json (11.2), with the library name serving as key and the expected version (if required) as value in the format described in [Versioning](#) (Section 11.2.1).

We have seen an example of this when introducing haxelib.json:

```
1 "dependencies": {
2   "tink_macros": "",
3   "nme": "3.5.5"
4 }
```

This adds two dependencies to the given Haxe library:

1. The library “tink.macros” can be used in any version. Haxelib will then always try to use the latest version.
2. The library “nme” is required in version “3.5.5”. Haxelib will make sure that this exact version is used, avoiding potential breaking changes with future versions.

## 11.3 extraParams.xml

If you add a file named `extraParams.xml` to your library root (at the same level as `haxelib.json`), these parameters will be automatically added to the compilation parameters when someone use your library with `-lib`.

## 11.4 Using Haxelib

If the `haxelib` command is executed without any arguments, it prints an exhaustive list of all available arguments. Access the <http://lib.haxe.org> website to view all the libraries available.

The following commands are available:

**Basic** `haxelib install [project-name] [version]` installs the given project. You can optionally specify a specific version to be installed. By default, latest released version will be installed.

`haxelib update [project-name]` updates a single library to their latest version.

`haxelib upgrade` upgrades all the installed projects to their latest version. This command prompts a confirmation for each upgradeable project.

`haxelib remove project-name [version]` removes complete project or only a specified version if specified.

`haxelib list` lists all the installed projects and their versions. For each project, the version surrounded by brackets is the current one.

`haxelib set [project-name] [version]` changes the current version for a given project. The version must be already installed.

**Information** `haxelib search [word]` lists the projects which have either a name or description matching specified word.

`haxelib info [project-name]` gives you information about a given project.

`haxelib user [user-name]` lists information on a given Haxelib user.

`haxelib config` prints the Haxelib repository path. This is where Haxelib get installed by default.

`haxelib path [project-name]` prints paths to libraries and its dependencies (defined in `haxelib.xml`).

**Development** `haxelib submit [project.zip]` submits a package to Haxelib. If the user name is unknown, you’ll be first asked to register an account. If the user already exists, you will be prompted for your password. If the project does not exist yet, it will be created, but no version will be added. You will have to submit it a second time to add the first released version. If you want to modify the project url or description, simply modify your `haxelib.xml` (keeping version information unchanged) and submit it again.

**haxelib register [project-name]** submits or update a library package.

**haxelib local [project-name]** tests the library package. Make sure everything (both installation and usage) is working correctly before submitting, since once submitted, a given version cannot be updated.

**haxelib dev [project-name] [directory]** sets a development directory for the given project. To set project directory back to global location, run command and omit directory.

**haxelib git [project-name] [git-clone-path] [branch] [subdirectory]** uses git repository as library. This is useful for using a more up-to-date development version, a fork of the original project, or for having a private library that you do not wish to post to Haxelib. When you use **haxelib upgrade** any libraries that are installed using GIT will automatically pull the latest version.

**Miscellaneous haxelib setup** sets the Haxelib repository path. To print current path use **haxelib config**.

**haxelib selfupdate** updates Haxelib itself. It will ask to run `haxe update.hxml` after this update.

**haxelib convertxml** converts `haxelib.xml` file to `haxelib.json`.

**haxelib run [project-name] [parameters]** runs the specified library with parameters. Requires a precompiled Haxe/Neko `run.n` file in the library package. This is useful if you want users to be able to do some post-install script that will configure some additional things on the system. Be careful to trust the project you are running since the script can damage your system.

**haxelib proxy** setup the Http proxy.



# Chapter 12

## Target Details

### 12.1 JavaScript

#### 12.1.1 Getting started with Haxe/JavaScript

Haxe can be a powerful tool for developing JavaScript applications. Let's look at our first sample. This is a very simple example showing the toolchain.

Create a new folder and save this class as `Main.hx`.

```
1 import js.Browser;
2 class Main {
3     static function main() {
4         var button = Browser.document.createElement();
5         button.textContent = "Click me!";
6         button.onclick = function(event) {
7             Browser.alert("Haxe is great");
8         }
9         Browser.document.body.appendChild(button);
10    }
11 }
```

To compile, either run the following from the command line:

```
1 haxe -js main-javascript.js -main Main -D js-flatten -dce full
```

Another possibility is to create and run (double-click) a file called `compile.hxml`. In this example the `hxml`-file should be in the same directory as the example class.

```
1 -js main-javascript.js
2 -main Main
3 -D js-flatten
4 -dce full
```

The output will be a `main-javascript.js`, which creates and adds a clickable button to the document body.

**Run the JavaScript** To display the output in a browser, create an HTML-document called `index.html` and open it.

```
1 <!DOCTYPE html>
2 <html>
```

```

3     <body>
4         <script src="main-javascript.js"></script>
5     </body>
6 </html>

```

### More information

- [Haxe JavaScript API docs](#)
- [MDN JavaScript Reference](#)

## 12.1.2 Using external JavaScript libraries

The externs mechanism (6.2) provides access to the native APIs in a type-safe manner. It assumes that the defined types exist at run-time but assumes nothing about how and where those types are defined.

An example of an extern class is the [jQuery class](#) of the Haxe Standard Library. To illustrate, here is a simplified version of this extern class:

```

1 package js.jquery;
2 @:native("$") extern class JQuery {
3     /**
4         Creates DOM elements on the fly from the provided string of
5         raw HTML.
6         OR
7         Accepts a string containing a CSS selector which is then used
8         to match a set of elements.
9         OR
10        Binds a function to be executed when the DOM has finished
11        loading.
12    **/
13    @:selfCall
14    @:overload(function(element:js.html.Element):Void { })
15    @:overload(function(selection:js.jquery.JQuery):Void { })
16    @:overload(function(callback:haxe.Constraints.Function):Void { })
17    @:overload(function(selector:String, ?context:haxe.extern.
18        EitherType<js.html.Element, js.jquery.JQuery>):Void { })
19    public function new():Void;
20
21    /**
22        Adds the specified class(es) to each element in the set of
23        matched elements.
24    **/
25    @:overload(function(_function:Int -> String -> String):js.jquery.
26        JQuery { })
27    public function addClass(className:String):js.jquery.JQuery;
28
29    /**
30        Get the HTML contents of the first element in the set of
31        matched elements.
32        OR

```

```

26         Set the HTML contents of each element in the set of matched
           elements.
27     **/
28     @:overload(function(htmlString:String):js.jquery.JQuery { })
29     @:overload(function(_function:Int -> String -> String):js.jquery.
        JQuery { })
30     public function html():String;
31 }

```

Note that functions can be overloaded to accept different types of arguments and return values, using the `@:overload` metadata. Function overloading works only in externs.

Using this extern, we can use jQuery like this:

```

1 import js.jquery.*;
2 ..
3 new JQuery("#my-div").addClass("brand-success").html("haxe is great!");
4 ..

```

The package and class name of the extern class should be the same as defined in the external library. If that is not the case, rewrite the path of a class using `@:native`.

```

1 package my.application.media;
2
3 @:native('external.library.media.video')
4 extern class Video {
5 ..

```

Some JavaScript libraries favor instantiating classes without using the `new` keyword. To prevent the Haxe compiler outputting the `new` keyword when using a class, we can attach a `@:selfCall` metadata to its constructor. For example, when we instantiate the jQuery extern class above, `new JQuery()` will be outputted as `$()` instead of `new $()`. The `@:selfCall` metadata can also be attached to a method. In this case, the method will be interpreted as a direct call to the object, illustrated as follows:

```

1 extern class Functor {
2     public function new():Void;
3     @:selfCall function call():Void;
4 }
5
6 class Test {
7     static function main() {
8         var f = new Functor();
9         f.call(); // will be outputted as `f();`
10    }
11 }

```

Beside externs, Typedefs (3.1) can be another great way to name (or alias) a JavaScript type. The major difference between typedefs and externs is that, typedefs are duck-typed but externs are not. Typedefs are suitable for common data structures, e.g. `point ({x:Float, y:Float})`. Use of a point structure typedef for function arguments allows external JavaScript functions to accept point class instances from Haxe or from another JavaScript library. It is also useful for typing JSON objects.

The Haxe Standard Library comes with externs of [jQuery](#) and [SWFObject](#). Their version compatibility is summarized as follows:

Haxe version	Library	Externs location
3.3	jQuery 1.11.3 / 2.1.4	<code>js.jquery.*</code>
3.2-	jQuery 1.6.4	<code>js.JQuery</code>
3.3	SWFObject 2.3	<code>js.swfobject.*</code>
3.2-	SWFObject 1.5	<code>js.SWFObject</code>

There are many externs for other popular native libraries available on Haxelib library (11). To view a list of them, check out the [extern tag](#).

### 12.1.3 Inject raw JavaScript

In Haxe, it is possible to call an exposed function thanks to the `untyped` keyword. This can be useful in some cases if we don't want to write externs. Anything untyped that is valid syntax will be generated as it is.

```
1 untyped window.trackEvent("page1");
```

### 12.1.4 JavaScript untyped functions

These functions allow to access specific JavaScript platform features. It works only when the Haxe compiler is targeting JavaScript and should always be prefixed with `untyped`.

*Important note:* Before using these functions, make sure there is no alternative available in the Haxe Standard Library. The resulting syntax can not be validated by the Haxe compiler, which may result in invalid or error-prone code in the output.

**`untyped __js__(expr, params)`** Injects raw JavaScript expressions (12.1.3). It's allowed to use `{0}`, `{1}`, `{2}` etc in the expression and use the rest arguments to feed Haxe fields. The Haxe compiler will take care of the surrounding quotes if needed. The function can also return values.

```
1 untyped __js__('alert("Haxe is great!")');
2 // output: alert("Haxe is great!");
3
4 var myMessage = "Haxe is great!";
5 untyped __js__('alert({0})', myMessage);
6 // output:
7 //   var myMessage = "Haxe is great!";
8 //   alert(myMessage);
9
10 var myVar:Bool = untyped __js__('confirm({0})', "Are you sure?");
11 // output: var myVar = confirm("Are you sure?");
12
13 var hexString:String = untyped __js__('({0}).toString({1})', 255, 16);
14 // output: var hexString = (255).toString(16);
```

**`untyped __instanceof__(o, cl)`** Same as `o instanceof cl` in JavaScript.

```
1 var myString = new String("Haxe is great");
2 var isString = untyped __instanceof__(myString, String);
3 output: var isString = (myString instanceof String);
```

**untyped \_\_typeof\_\_(o)** Same as `typeof o` in JavaScript.

```
1 var isNodeJS = untyped __typeof__(window) == null;
2 output: var isNodeJS = typeof(window) == null;
```

**untyped \_\_strict\_eq\_\_(a,b)** Same as `a === b` in JavaScript, tests on **strict equality** (or "triple equals" or "identity").

```
1 var a = "0";
2 var b = 0;
3 var isEqual = untyped __strict_eq__(a, b);
4 output: var isEqual = ((a) === b);
```

**untyped \_\_strict\_neq\_\_(a,b)** Same as `a !== b` in JavaScript, tests on negative strict equality.

```
1 var a = "0";
2 var b = 0;
3 var isntEqual = untyped __strict_neq__(a, b);
4 output: var isntEqual = ((a) !== b);
```

**Expression injection** In some cases it may be needed to inject raw JavaScript code into Haxe-generated code. With the `__js__` function we can inject pure JavaScript code fragments into the output. This code is always untyped and can not be validated, so it accepts invalid code in the output, which is error-prone. This could, for example, write a JavaScript comment in the output.

```
1 untyped __js__('// haxe is great!');
```

A more useful demonstration would be to call a function and pass arguments using the `__js__` function. This example illustrates how to call this function and how to pass parameters. Note that the *code interpolation* will wrap the quotes around strings in the generated output.

```
1 // Haxe code:
2 var myVar = untyped __js__('myObject.myJavaScriptFunction({0}, {1})',
    "Mark", 31);
```

This will generate the following JavaScript code:

```
1 // JavaScript Code
2 var myVar = myObject.myJavaScriptFunction("Mark", 31);
```

### 12.1.5 Debugging JavaScript

Haxe is able to generate **source maps**, allowing Javascript debuggers to map from generated JavaScript back to the original Haxe source. This makes reading error stack traces, debugging with breakpoints, and profiling much easier.

Compiling with the `-debug` flag will create a `.map` alongside the `.js` file. Enable it in Chrome by clicking on the cog settings button in the bottom right of the Developer Tools window, and checking "Enable source maps". The pause button on the bottom left can be toggled to pause on uncaught exceptions.

### 12.1.6 JavaScript target Metadata

This is the list of JavaScript specific metadata. For more information, see also the complete list of all Haxe built-in metadata ([8.1](#)).

JavaScript metadata	
Metadata	Description
@:expose _(?Name=Class path)-	Makes the class available on the <code>window</code> object or <code>exports</code> for node.js
@:jsRequire	Generate javascript module require expression for given extern
@:selfCall	Translates method calls into calling object directly

### 12.1.7 Exposing Haxe classes for JavaScript

It is possible to make Haxe classes or static fields available for usage in plain JavaScript. To expose, add the `@:expose` metadata to the desired class or static fields.

This example exposes the Haxe class `MyClass`.

```
1 @:expose
2 class MyClass {
3     var name:String;
4     function new(name:String) {
5         this.name = name;
6     }
7     public function foo() {
8         return 'Greetings from $name!';
9     }
10 }
```

It generates the following JavaScript output:

```
1 (function ($hx_exports) { "use strict";
2 var MyClass = $hx_exports.MyClass = function(name) {
3     this.name = name;
4 };
5 MyClass.prototype = {
6     foo: function() {
7         return "Greetings from " + this.name + "!";
8     }
9 };
10 })(typeof window !== "undefined" ? window : exports);
```

By passing globals (like `window` or `exports`) as parameters to our anonymous function in the JavaScript module, it becomes available which allows to expose the Haxe generated module.

In plain JavaScript it is now possible to create an instance of the class and call its public functions.

```
1 // JavaScript code
2 var instance = new MyClass('Mark');
3 console.log(instance.foo()); // logs a message in the console
```

The package path of the Haxe class will be completely exposed. To rename the class or define a different package for the exposed class, use `@:expose("my.package.MyExternalClass")`

**Shallow expose** When the code generated by Haxe is part of a larger JavaScript project and wrapped in a large closure it is not always necessary to expose the Haxe types to global variables. Compiling the project using `-D shallow-expose` allows the types or static fields to be available for the surrounding scope of the generated closure only.

When the code is compiled using `-D shallow-expose`, the generated output will look like this:

```
1 var $hx_exports = $hx_exports || {};  
2 (function () { "use strict";  
3 var MyClass = $hx_exports.MyClass = function(name) {  
4     this.name = name;  
5 };  
6 MyClass.prototype = {  
7     foo: function() {  
8         return "Greetings from " + this.name + "!";  
9     }  
10 };  
11 }) ();  
12 var MyClass = $hx_exports.MyClass;
```

In this pattern, a `var` statement is used to expose the module; it doesn't write to the `window` or `exports` object.

### 12.1.8 Loading extern classes using "require" function

Since Haxe 3.2.0

Modern JavaScript platforms, such as Node.js provide a way of loading objects from external modules using the "require" function. Haxe supports automatic generation of "require" statements for extern classes.

This feature can be enabled by specifying `@:jsRequire` metadata for the extern class. If our extern class represents a whole module, we pass a single argument to the `@:jsRequire` metadata specifying the name of the module to load:

```
1 @:jsRequire("fs")  
2 extern class FS {  
3     static function readFileSync(path:String, encoding:String):String;  
4 }
```

In case our extern class represents an object within a module, second `@:jsRequire` argument specifies an object to load from a module:

```
1 @:jsRequire("http", "Server")  
2 extern class HTTPServer {  
3     function new();  
4 }
```

The second argument is a dotted-path, so we can load sub-objects in any hierarchy.

If we need to load custom JavaScript objects in runtime, a `js.Lib.require` function can be used. It takes `String` as its only argument and returns `Dynamic`, so it is advised to be assigned to a strictly typed variable.

## 12.2 Flash

### 12.2.1 Getting started with Haxe/Flash

Developing Flash applications is really easy with Haxe. Let's look at our first code sample. This is a basic example showing most of the toolchain.

Create a new folder and save this class as `Main.hx`.

```
1 import flash.Lib;
2 import flash.display.Shape;
3 class Main {
4     static function main() {
5         var stage = Lib.current.stage;
6
7         // create a center aligned rounded gray square
8         var shape = new Shape();
9         shape.graphics.beginFill(0x333333);
10        shape.graphics.drawRoundRect(0, 0, 100, 100, 10);
11        shape.x = (stage.stageWidth - 100) / 2;
12        shape.y = (stage.stageHeight - 100) / 2;
13
14        stage.addChild(shape);
15    }
16 }
```

To compile this, either run the following from the command line:

```
1 haxe -swf main-flash.swf -main Main -swf-version 15 -swf-header
   960:640:60:f68712
```

Another possibility is to create and run (double-click) a file called `compile.hxml`. In this example the `hxml`-file should be in the same directory as the example class.

```
1 -swf main-flash.swf
2 -main Main
3 -swf-version 15
4 -swf-header 960:640:60:f68712
```

The output will be a `main-flash.swf` with size 960x640 pixels at 60 FPS with an orange background color and a gray square in the center.

**Display the Flash** Run the SWF standalone using the [Standalone Debugger FlashPlayer](#).

To display the output in a browser using the Flash-plugin, create an HTML-document called `index.html` and open it.

```
1 <!DOCTYPE html>
2 <html>
3     <body>
4         <embed src="main-flash.swf" width="960" height="640">
5     </body>
6 </html>
```

#### More information

- [Haxe Flash API docs](#)
- [Adobe Livedocs](#)



## 12.2.2 Embedding resources

Embedding resources is different in Haxe compared to ActionScript 3. Instead of using

*embed*

(AS3-metadata) use Flash specific compiler metadata ([12.2.4](#)) like `@:bitmap`, `@:font`, `@:sound` or `@:file`.

```
1 import flash.Lib;
2 import flash.display.BitmapData;
3 import flash.display.Bitmap;
4
5 class Main {
6     public static function main() {
7         var img = new Bitmap( new MyBitmapData(0, 0) );
8         Lib.current.addChild(img);
9     }
10 }
11
12 @:bitmap("relative/path/to/myfile.png")
13 class MyBitmapData extends BitmapData { }
```

## 12.2.3 Using external Flash libraries

To embed external `.swf` or `.swc` libraries, use the following **compilation options**:

**-swf-lib <file>** Embeds the SWF library in the compiled SWF.

**-swf-lib-extern <file>** Adds the SWF library for type checking but doesn't include it in the compiled SWF.

The standard compilation options also provide more Haxe sources to be added to the project:

- To add another class path use `-cp <directory>`.
- To add a Haxelib library ([11](#)) use `-lib <library-name>`.
- To force a whole package to be included in the project, use `--macro include('mypackage')` which will include all the classes declared in the given package and subpackages.

## 12.2.4 Flash target Metadata

This is the list of Flash specific metadata. For a complete list see Haxe built-in metadata ([8.1](#)).

Flash metadata	
Metadata	Description
@:bind	Override Swf class declaration
@:bitmap <i>_(Bitmap file path)_</i>	<i>_</i> Embeds given bitmap data into the class (must extend <code>flash.display.BitmapData</code> )
@:debug	Forces debug information to be generated into the Swf even without <code>-debug</code>
@:file(File path)	Includes a given binary file into the target Swf and associates it with the class
@:font <i>_(TTF path Range String)_</i>	<i>_</i> Embeds the given TrueType font into the class (must extend <code>flash.text.TextField</code> )
@:getter <i>_(Class field name)_</i>	<i>_</i> Generates a native getter function on the given field
@:noDebug	Does not generate debug information into the Swf even if <code>-debug</code> is set
@:ns	Internally used by the Swf generator to handle namespaces
@:setter <i>_(Class field name)_</i>	<i>_</i> Generates a native setter function on the given field
@:sound <i>_(File path)_</i>	<i>_</i> Includes a given <code>_.wav_</code> or <code>_.mp3_</code> file into the target Swf and associates it with the class

## 12.3 Neko

## 12.4 PHP

### 12.4.1 Getting started with Haxe/PHP

To get started with Haxe/PHP, create a new folder and save this class as `Main.hx`.

```

1 import php.Lib;
2
3 class Main {
4     static function main() {
5         Lib.println('Haxe is great!');
6     }
7 }
```

To compile, either run the following from the command line:

```
1 haxe -php bin -main Main
```

Another possibility is to create and run (double-click) a file called `compile.html`. In this example the `html`-file should be in the same directory as the example class.

```

1 -php bin
2 -main Main
```

The compiler outputs in the given `bin`-folder, which contains the generated PHP classes that prints the traced message when you run it. The generated PHP-code runs for version 5.1.0 and later.

#### More information

- [Haxe PHP API docs](#)
- [PHP.net Documentation](#)
- [PHP to Haxe tool](#)

## 12.4.2 PHP untyped functions

These functions allow to access specific PHP platform features. It works only when the Haxe compiler is targeting PHP and should always be prefixed with `untyped`.

*Important note:* Before using these functions, make sure there is no alternative available in the Haxe Standard Library. The resulting syntax can not be validated by the Haxe compiler, which may result in invalid or error-prone code in the output.

**`untyped __php__(expr)`** Injects raw PHP code expressions. It's possible to pass fields from Haxe source code using String Interpolation (6.5).

```
1 var value:String = "test";
2 untyped __php__("echo '<pre>'; print_r($value); echo '</pre>';");
3 // output: echo '<pre>'; print_r('test'); echo '</pre>';
```

**`untyped __call__(function, arg, arg, arg...)`** Calls a PHP function with the desired number of arguments and returns what the PHP function returns.

```
1 var value = untyped __call__("array", 1,2,3);
2 // output returns a NativeArray with values [1,2,3]
```

**`untyped __var__(global, paramName)`** Get the values from global vars. Note that the dollar sign in the Haxe code is omitted.

```
1 var value : String = untyped __var__('_SERVER', 'REQUEST_METHOD')
2 // output: $value = $_SERVER['REQUEST_METHOD']
```

**`untyped __physeq__(val1, val2)`** Strict equals test between the two values. Returns a Bool.

```
1 var isFalse = untyped __physeq__(false, value);
2 // output: $isFalse = false == $value;
```

## 12.5 C++

### 12.5.1 Using C++ Defines

- `ANDROID_HOST`
- `ANDROID_NDK_DIR`
- `ANDROID_NDK_ROOT`
- `BINDIR`
- `DEVELOPER_DIR`
- `HXCPP`
- `HXCPP_32`
- `HXCPP_COMPILE_CACHE`

- HXCPP\_COMPILE\_THREADS
- HXCPP\_CONFIG
- HXCPP\_CYGWIN
- HXCPP\_DEPENDS\_OK
- HXCPP\_EXIT\_ON\_ERROR
- HXCPP\_FORCE\_PDB\_SERVER
- HXCPP\_M32
- HXCPP\_M64
- HXCPP\_ARMV6
- HXCPP\_ARMV7
- HXCPP\_ARMV7S
- HXCPP\_ARM64
- HXCPP\_X86
- HXCPP\_MINGW
- HXCPP\_MSVC
- HXCPP\_MSVC\_CUSTOM
- HXCPP\_MSVC\_VER
- HXCPP\_NO\_COLOR
- HXCPP\_NO\_COLOUR
- HXCPP\_VERBOSE
- HXCPP\_WINXP\_COMPAT
- HXCPP\_API\_LEVEL
- HXCPP\_DEBUG\_LINK
- DHXCPP\_STACK\_TRACE
- HXCPP\_STACK\_LINE
- HXCPP\_CHECK\_POINTER
- HXCPP\_DEBUGGER
- HXCPP\_CPP11
- HXCPP\_STRICT\_CASTS
- HXCPP\_VISIT\_ALLOCS
- HXCPP\_WINXP\_COMPAT

- HXCPP\_OPTIMIZE\_FOR\_SIZE
- IPHONE\_VER
- LEGACY\_MACOSX\_SDK
- LEGACY\_XCODE\_LOCATION
- MACOSX\_VER
- MSVC\_VER
- NDKV
- NO\_AUTO\_MSVC
- PLATFORM
- QNX\_HOST
- QNX\_TARGET
- TOOLCHAIN\_VERSION
- USE\_GCC\_FILETYPES
- USE\_PRECOMPILED\_HEADERS
- android
- apple
- blackberry
- cygwin
- dll\_import
- dll\_import\_include
- dll\_import\_link
- emcc
- emscripten
- gph
- hardfp
- haxe\_ver
- ios
- iphone
- iphoneos
- iphonesim
- linux

- linux\_host
- mac\_host
- macos
- mingw
- rpi
- simulator
- tizen
- toolchain
- webos
- windows
- windows\_host
- winrt
- xcompile

### **12.5.2 Using C++ Pointers**

## **12.6 Java**

## **12.7 C#**

## **12.8 Python**